United States
Department of
Agriculture

**Agricultural
Research
Service**

ARS–113

September 1995

RHIZOS 1991: A Simulator of Row Crop Rhizospheres

RHIZOS 1991: A Simulator of Row Crop Rhizospheres

M.Y.L. Boone, D.O. Porter, and J.M. McKinion

Boone is assistant research soil physicist, Mississippi State University, Mississippi State. Porter was general engineer and McKinion is electronics engineer/research leader, U.S. Department of Agriculture, Agricultural Research Service, Crop Simulation Research Unit, Mississippi State, MS.

ABSTRACT

Boone, M.Y.L., D.O. Porter, and J.M McKinion. 1995. RHIZOS 1991: A Simulator of Row Crop Rhizospheres. U.S. Department of Agriculture, Agricultural Research Service, ARS–113, 180 pp.

This work constitutes the first detailed publication of RHIZOS, a computer model that simulates the processes occurring in the soil under a growing row crop. Comprising 22 subroutines coded in FORTRAN 77, the model is intended to aid decision making about irrigation, cultivation, and other farm processes. It offers the following parameters: (1) effective soil water potential, (2) estimates of metabolite sink strength in roots, and (3) rate of nitrogen uptake. This publication is also intended to facilitate work with the RHIZOS source code, which was incorporated in the mid-1980's into the widely used GOSSYM/COMAX system.

Keywords: Fertilizers, irrigation, nitrification, rhizosphere, soil bulk density, soil moisture, soil physics, soil temperature.

The "Source Code" portions of this manual are printed essentially as they appear in the computer model. They were not edited for style, consistency, or conventions of usage.

While supplies last, single copies of this publication and copies of the source code may be obtained free of charge from the authors at USDA-ARS, Crop Simulation Research Unit, P.O. Box 5367, Mississippi State, MS 39762–5367.

Copies of this publication may be purchased from the National Technical Information Service, 5285 Port Royal Road, Springfield, VA 22161.

Issued September 1995

CONTENTS

ACKNOWLEDGMENTS

The work of many individuals went into reviewing and revising the RHIZOS model and then preparing the manuscript so this book could be published. In particular, the authors wish to express their appreciation to the following people: Gary Theseira for his assistance in preparing, reviewing, and revising the subroutines; Susan Bridges for her technical advice, assistance in unit analyses, and proofreading; Sam Turner for his programming support; and Martha Barnes and Wendell Ladner for preparing drawings and assisting in the compilation of the manuscript.

To acknowledge their significant contributions, we list here the participants of the 1991 RHIZOS Working Group.

Soil Scientists

M.Y.L. Boone, Mississippi State University, Mississippi State, MS
F. Khorsandi, Mississippi State University
G. Stevens, GOSSYM-COMAX Information Unit, Starkville, MS
G. Theseira, Mississippi State University
J. Varco, Mississippi State University
F.D. Whisler, Mississippi State University

Plant Physiologists

D. Albers, University of Missouri-Columbia, Portageville
N. Bhattacharya, USDA, ARS, Water Conservation Laboratory, Phoenix, AZ
H.F. Hodges, Mississippi State University
G. Wall, USDA, ARS, Water Conservation Laboratory, Phoenix, AZ

Agricultural Engineer

D.O. Porter, USDA, ARS, Crop Simulation Research Unit, Mississippi State

Computer Scientists

S.M. Bridges, Mississippi State University
S.B. Turner, USDA, ARS, Crop Simulation Research Unit, Mississippi State

Electronics Engineer/Research Leader

J.M. McKinion, USDA, ARS, Crop Simulation Research Unit, Mississippi State

ABOUT RHIZOS

Mechanistic modeling of crop growth and development requires simulation of intricate processes in the soil. To that end, Lambert and Baker (1984) developed the soil-systems model RHIZOS in 1973. They had the following objectives:

1. development of a whole-crop simulation model;
2. simulation of phenomena not included in other whole-crop simulation models (for example, nutritional drought, subirrigation, aeration, compaction, and nitrogen transformations);
3. ongoing analysis of research needs in the soil-crop system;
4. creation of a framework to accommodate knowledge and research information; and
5. adaptation of whole-crop simulation models to be used when making decisions about irrigation, cultivation, and other farming processes.

RHIZOS is designed to provide a general rhizosphere model for all crops. For aboveground crops, the model offers three parameters: (1) effective soil water potential, used in calculating plant water potential, (2) estimates of metabolite sink strength in roots, and (3) the rate of nitrogen uptake (Baker et al. 1983).

BACKGROUND

RHIZOS was first documented in 1984 by Lambert and Baker, but not published. By that time, it had been incorporated into GOSSYM, a cotton simulation model. Since the time GOSSYM was field tested in 1984 and 1985, it and its companion expert management system, COMAX, have been employed by more than 300 users including extension agents, researchers, farmers, and consultants. These users provided feedback to the scientists involved in developing and maintaining the model by reporting problems encountered in the field. Generally, the problems were related to water and nitrogen movement, evapotranspiration, mineralization, and nitrification processes.

As new information became available, RHIZOS was continuously updated. Unfortunately, many of the underlying, simplifying assumptions were lost, never documented, or never stated. The resulting ambiguities caused problems in verifying the soundness of assumptions, as well as in following the logic of the code.

Over the years, researchers at the U.S. Department of Agriculture, Agricultural Research Service, Crop Simulation Research Unit (in Mississippi State, MS) and across the Cotton Belt expressed their difficulty in working with the RHIZOS source code (FORTRAN program). Major complaints concerned the lack of code documentation, reference materials, and standard definitions and units for variables. Portions of the code documentation were erroneous because sections of code had been added or changed without including an explanation, the date, the reason for change, or the names of the people responsible.

As a result, a multidisciplinary review team (the 1991 RHIZOS Working Group) was assembled, comprising soil scientists, agricultural engineers, plant physiologists, and computer scientists. This book, the result of their half-year effort, documents RHIZOS as used in the 1991 GOSSYM-COMAX model. It is intended to aid research scientists in their work with RHIZOS and GOSSYM.

Scientists not previously active in the activities of the Crop Simulation Research Unit (CSRU) were included as members of the review team and contributed valuable insights into the validity of assumptions and the behavior of factors under field conditions.

In the past some of the scientists had expressed interest in contributing to the work of CSRU but were discouraged or intimidated by the computer language and terminology commonly used at the laboratory. To accommodate these individuals, each subroutine was carefully translated into easy-to-read pseudocode, outlines, and diagrams. In some cases, background information was helpful; lectures on nitrogen transformation, soil physics, numerical analysis, and microclimatology preceded or accompanied discussions of some subroutines. With such "language barriers" removed, individuals openly asked questions and shared laboratory research and field experiences. The resulting multidisciplinary interaction was one of the most valuable accomplishments of the project.

The team reviewed the source code, located the references used to develop the code, listed the requirements for input and output, prepared the glossary of terms, and presented alternative courses of action when using the various subroutines that make up the RHIZOS model. They also developed research plans to improve the logic of the code and obtain data sets for validation.

The team also found and corrected some errors in the source code. Assumptions were questioned, and where applicable, they were confirmed, updated, or replaced. Needs for further research and for better data were identified.

New methods of modeling and numerical analysis, as well as data from current and recent research, were evaluated as possible alternatives to modify or replace some RHIZOS operations. The review team recommended that some routines be replaced as new models become available and suggested possible replacement of some equations or curves.

RECOMMENDATIONS

Evaluation of the various components or subroutines of RHIZOS also exposed some shortcomings and the need for additional data in some areas, briefly discussed next.

Water movement in the CAPFLO subroutine is based on vertical and horizontal gradients in volumetric water content. Evaluation revealed that this process may produce errors when adjacent cells contain soils of different properties, as in an interface between soil horizons. It was discovered that, under certain conditions, water moving in response to a water content gradient was actually moving against an energy potential gradient. A new soil water flux model designed to respond to energy gradients will soon be available for use with RHIZOS.

Water movement and distribution problems seem to be prevalent in sandy soils. A possible cause for the model's difficulty with these soils involves the calculation or slope of the diffusivity term. Other methods will be evaluated as potential replacements for diffusivity calculations in the current model.

RHIZOS cannot adequately simulate an alternate furrow system or a drip irrigation system, because the assumption of symmetry within the profile is not valid when these systems are used. An alternate furrow model is currently under development, and such a change to the model will increase its applicability.

Evapotranspiration (subroutine ET), based on the Penman equation (Ritchie 1972), was modified with data collected from one particular soil. It is recommended that other evapotranspiration models be considered and the current ET model be modified or replaced.

RIMPED, like the ET model, is based on data from one soil. Because data are available for other soils, it is recommended that they be compiled and incorporated in the model in a format similar to the current root impedance tables.

RRUNOFF is a generalized routine to estimate water loss due to surface runoff. It is based upon a Soil Conservation Service method for estimating storm runoff in flood control structures. RRUNOFF will be replaced with a new infiltration-runoff model.

TMPSOL is a routine currently used for estimating soil temperature by soil horizon. It is based on empirical experiments done with one soil in one location. The model cannot accommodate variations in soil characteristics, such as texture, color, thermal properties, or soil moisture content. A new soil temperature model is currently under development.

Adequate data sets are available to verify and validate some parts of the RHIZOS model, but adequate data for validation of the entire RHIZOS model are not available. There is a need for both data and models describing denitrification, uptake of the nutrients potassium and phosphorus, soil oxygen, the effects of cultivation, and fluctuations in the water table.

RHIZOS is a model that continues to develop as new data and models become available, and it is hoped that improved versions will be issued in the future. Optimal development of RHIZOS requires the cooperation of a multidisciplinary team to conduct research, evaluate new models and other resources, develop the computer source code, and document all related activities. All persons involved in model development must be familiar with the RHIZOS model as a whole and understand its assumptions, methods, and limitations. Efforts to modify or improve the model without such understanding often produce confusion, model inefficiency, and errors, as evidenced by past experiences.

Documentation of the model was a very important part of this review project, and as the model is updated, the documentation must also be updated. Those of us responsible for the content of this book strongly urge individuals working with the model to document their work by recording information similar to that included here. Changes or recommendations for changes in the model should be discussed with other individuals involved in the development of RHIZOS. Available contact people are identified in the individual header of the pseudocodes.

RHIZOS BASICS

THE MODEL

RHIZOS is a dynamic computer simulation of processes occurring in the soil under a growing row crop. These processes include root growth, water and nitrogen uptake and redistribution, and microbiological processes involving nitrogen. A spatial description of roots, water, nitrogen, and temperature is achieved through the use of two-dimensional geometry.

The model treats a slab of soil 1 cm thick that lies perpendicular to the row. Extending from the center of one row to the center of the adjacent row (see fig. 1), the slab is represented as a matrix of cells that is NL cells wide and NK cells deep. Each individual cell has a width of WCELL (cm) and a depth of DCELL (cm). These cells fill the plane between the rows of NK*WCELL spacing and to the bottom of the root zone NL*DCELL deep. Roots are assumed to grow symmetrically with respect to the row and, thus, the slab.



Figure 1. Geometry of the soil slab used in simulating root growth and other soil processes. Row spacing is NK*WCELL (cm); profile depth is NL*DCELL (cm); slab is 1 cm thick.

The governing equation for the model is the two-dimensional continuity equation (in partial differential form):

$$\frac{\delta\Theta(x,z,t)}{\delta t} = -\frac{\delta q_z}{\delta z} - \frac{\delta q_x}{\delta x} - S(x,z,t)$$

where

Θ = volumetric water content (ml cm^{-3})
t = time (sec)
z = vertical distance from the reference planes (cm)
x = horizontal distance from the reference planes (cm)
q_z = fluxes in the soil in the z direction (ml cm^{-1} sec^{-1})
q_x = fluxes in the soil in the x direction (ml cm^{-1} sec^{-1})
S = the volumetric sink term, which specifies the positive rate of water uptake from the soil by roots (ml cm^{-3} soil sec^{-1}).

UNDERLYING ASSUMPTIONS

The model assumes the following initial and boundary conditions:

1. For any cell (located at a depth of z and a width of x) within the profile at a time of 0, the soil water potential of the cell, h(x,z,0) will be the same as the cell's initial soil water potential, $h_0(x,z)$. This condition also implies that the moisture content of the cell, $\Theta(x,z,0)$ is equal to the cell's initial moisture content, $\Theta_0(x,z)$. In addition, the roots capable of uptake (expressed in dry weight) that are present in the cell, RTWTCU(x,z,0), are similar to the cell's initial root weight capable of uptake, $RTWTCU_0(x,z)$. Expressed mathematically, at t = 0, $0 \leq z \leq NL*DCELL$, and $0 \leq x \leq NK*WCELL$:

   $h(x,z,0) = h_0(x,z)$
   $\Theta(x,z,0) = \Theta_0(x,z)$
   $RTWTCU(x,z,0) = RTWTCU_0(x,z)$.

2. At the soil surface, z = 0, and as time passes, t > 0, the flux of water, q(x,0,t) can either be less than or equal to the prescribed flux, P(x,0,t), (cm^3 cm^{-2} day^{-1}). Furthermore, the moisture content, $\Theta(x,0,t)$ can have values ranging from saturated, Θ_s, to dry, Θ_d, (dry equaling the permanent wilting point or 15,000 cm).

   The prescribed flux is positive during rainfall or irrigation, P(x,0,t) > 0, and negative during evaporation, P(x,0,t) < 0. Since the amount of rainfall or irrigation is used as input instead of the rate of rainfall or irrigation, $\Theta(x,0,t) = \alpha_1\Theta_s$ is assumed until the total amount of applied water has infiltrated the soil or until a new portion of water is applied. The proportionality constant α_1 can be less than or equal to 1, ($\alpha_1 \leq 1$). The uninfiltrated amount of water is assumed to be lost as runoff. Thus, where z = 0 and t > 0,

   $|q(x,0,t)| \leq |P(x,0,t)|$
   $\Theta_d \leq \Theta(x,0,t) \leq \Theta_s$.

3. At the lateral boundaries, x = 0, x = NK*WCELL at t > 0, and q(0,z,t) = 0. These boundaries are located under the row and midway between the rows.

4. At the bottom, b, of the soil profile, z = NL*DCELL, and at $t \geq 0$, $\Theta(x,z,t) = \Theta_b$. Roots are not allowed to extend beyond the defined boundaries of the slab.

STRUCTURE OF THE MODEL

RHIZOS is a collection of 22 subroutines, coded in standard FORTRAN 77, which describes related sets of physical processes. Eleven subroutines, simulating major soil processes, are called from SOIL, the control subroutine (see fig. 2 for depiction of SOIL as a flow chart). The name and a brief description of each subroutine are given following:

GBLOCK: Initializes the values of state variables and variable arrays. State variables are the components of the system that change over time and are usually expressed in units of mass or energy. DATA statements in this subroutine give numerical value to variables that are not dynamically initialized before use.

SOILHYDR: Reads the soil hydrology data file.

SOILIMPD: Reads the soil impedance data file.

INITSOIL: Reads the initial soil fertility data file.

AGINPUTS: Reads the irrigation and fertilizer data file.

INITIALIZE: Initializes soil cell dimensions, saves initial soil hydrology data, calculates maximum and minimum water fluxes, initializes weights of roots in cells, and initializes volumetric nitrogen content of the soil.

WEATHER: Reads actual and future weather data files.

CLYMAT: Renames weather variables and converts them to metric units; calculates daylength, mean day and night air temperatures, and canopy light interception; and calls the following subroutines:

> RRUNOFF: Estimates the amount of runoff for the day from rain or irrigation.
>
> JULANTOCAL: Keeps track of the Julian day number and the calendar date being simulated.
>
> TMPSOL: Calculates the soil temperature by soil layer.

SOIL: Is the control subroutine for RHIZOS.

FRTLIZ: Initializes the nitrogen and organic matter content of the profile at planting and distributes ammonium, nitrate, and urea fertilizers in the profile.

GRAFLO: Moves rain and irrigation water into the soil profile by gravitational flow and moves nitrogen in solution by mass flow.

ET: Estimates evaporation from the soil surface and transpiration from the plant.

EVPSOIL: Adjusts soil evaporation to account for canopy growth.

UPTAKE: Calculates the amount of water that roots take up from the surrounding soil.

CAPFLO: Estimates the movement of water within the profile in response to soil moisture gradients and moves nitrogen by mass flow.

NITRIF: Mineralizes organic matter and urea and calculates the conversion of ammonium to nitrate by bacterial action in the soil.

RUTGRO: Calculates potential and actual growth (in terms of dry matter) of roots and average soil water potential. It also calls RIMPED.

RIMPED: Calculates the effect of soil bulk density on root elongation.



Figure 2. Flowchart of the SOIL subroutine



Figure 2. Flowchart of the SOIL subroutine (continued)



Figure 2. Flowchart of the SOIL subroutine (continued)

OUT: Plots the soil slab and the densities of the array elements in each cell.

All of the subroutines contain the FORTRAN statement INCLUDE GOSCOM.FOR, which inserts the entire contents of the GOSCOM.FOR file at the point occupied by the statement. The GOSCOM.FOR file contains all of the REAL, INTEGER, and COMMON statements. The COMMON statements are lists of all variables accessible by all subroutines.

The majority of the book constitutes a detailed treatment of each subroutine, including:

- a brief description of the process or processes simulated,
- a list of assumptions,
- inputs and outputs,
- general pseudocode, that is, a listing of major processes in the subroutine (not included for every subroutine),
- pseudocode, which is an almost line-by-line interpretation of the source code,
- the source code, and
- a glossary of terms.

GENERAL INPUT REQUIREMENTS

Five data files, stored in ASCII format, are needed to run RHIZOS.

Soil Hydrology

Data include bulk density, soil moisture retention values (soil water potentials and actual water contents), diffusivity, and the percentage of sand and the percentage of clay in each soil horizon (table 1).

Soil Impedance

Data are included in the discussion of the RIMPED subroutine (table 2).

Initial Soil Fertility

Data include residual ammonia and nitrate concentrations, percentage of organic matter content, and moisture contents at 15-cm intervals in the upper meter of soil as a percentage of field capacity values (table 3).

Agricultural Inputs

Data include the date, amount, method, and cost of each irrigation application and the date, type (ammonium, nitrate, urea), cost, method, and location of each addition of fertilizer (table 4).

Weather

Data include daily values for maximum and minimum temperatures, wind run, rainfall, and solar radiation (table 5).

Table 1. Sample soil hydrology data file

Line*		Column* 1	2	3	4	5	6	7
1	'VICMIX2.HYD'							
2	'Victoria-mix clay, King Ranch, TX, TAMU (FDW)'							
3	2							
4		0.2999E−04	0.1890E+00	0.7163E+02	0.4190E+00			
5		0.3420E+00	0.1840E+00	0.1700E+00	0.1530E+01			
6		0.2999E−04	0.1890E+00	0.7163E+02	0.4190E+00			
7		0.3420E+00	0.1840E+00	0.1700E+00	0.1530E+01			
8	47.	-0.3170E+00	0.0000E+00	0.1000E+01	-0.3300E+00	999	1000	
9	25	0.2999E−04	0.1890E+00	0.7163E+02	0.4190E+00			
10		0.3420E+00	0.1840E+00	0.1700E+00	0.1530E+01	20	60	1000
11	201	0.3303E−05	0.1810E+00	0.1119E+03	0.3170E+00			
12		0.2890E+00	0.1790E+00	0.1770E+00	0.1720E+01	00	00	1000

*Line and column numbers were added to help explain the file content.

Line

1 Soil hydrology data file name (FILNAM)

2 Soil file description (DSCRIP)

3 Number of soil horizons (LYRSOL); in this instance, 2

4 Data for the cultivated condition (used in CULVAT routine)

Column

1 Diffusivity at –15,000 cm potential (DIFF0C) (cm^2 day^{-1})

2 Volumetric water content at –15,000 cm potential THTA0C (cm^3H_2O $cm^{-3}soil$)

3 Hydraulic conductance (BETAC) (cm^3H_2O $cm^{-3}soil$)

4 Saturated volumetric water content (THETASC) (cm^3H_2O $cm^{-3}soil$)

5 Data for the cultivated condition (continued)

Column

1 Volumetric water content at field capacity (FCINIC) (cm^3H_2O $cm^{-3}soil$)

2 Residual volumetric water content (THTARC) (cm^3H_2O $cm^{-3}soil$)

3 Volumetric water content at air dry (AIRDRC) (cm^3H_2O $cm^{-3}soil$)

4 Bulk density (BDC) (mg m^{-3})

6 Data for the wheel traffic condition (same as line 4)

7 Data for the wheel traffic condition (same as line 5)

8 Data for handling the presence of the water table in the profile

Column

1 "Time-days" factor for calculating movement of the water table (TD)

2 Volumetric water content at field capacity (THETAI) (cm^3H_2O $cm^{-3}soil$)

3 Boundary theta slope for calculating movement of the water table (BDSLOP)

4 Boundary ratio for calculating movement of the water table (BDRATO)

5 Water potential at field capacity (PSISFC) (cm^3H_2O $cm^{-3}soil$)

6 Depth from the soil surface to the water table (WATTBL) (cm)

7 Pointer (POINTR)

Line

9 Data for the first soil horizon

Column

1 Depth of the layer (LDEPTH) (cm)

2 Diffusivity at –15,000 cm potential (DIFF0) (cm^2 day^{-1})

3 Volumetric water content at –15,000 cm potential (THETA0) (cm^3H_2O $cm^{-3}soil$)

4 Hydraulic conductance (BETA) (cm^3H_2O $cm^{-3}soil$)

5 Saturated volumetric water content (THETAS) (cm^3H_2O $cm^{-3}soil$)

10 Data for the first soil horizon (continued)

Column

1 Volumetric water content at field capacity (FCININ) (cm^3H_2O $cm^{-3}soil$)

2 Residual volumetric water content (THETAR) (cm^3H_2O $cm^{-3}soil$)

3 Volumetric water content at "air dry" (AIRDR) (cm^3H_2O $cm^{-3}soil$)

4 Bulk density (BD) (mg m^{-3})

5 Percent sand (IPSAND)

6 Percent clay (IPCLAY)

7 Pointer (POINTR)

11 Data for the second soil horizon (same as line 9)

12 Data for the second soil horizon (same as line 10)

Table 2. Sample soil impedance data file

Line*	Column* 1	2
1	NORFOLK SANDY LOAM	
2	7	
3	6	0.05
4	0.9	0.10
5	1.1	5.40
6	1.3	16.20
7	1.5	36.00
8	1.7	62.00
9	1.9	93.00

*Line and column numbers were added to help explain the file content.

Line

1 Soil type (SNAME)

2 Number of curves (NCURVE)

3 **Column**
- 1 Number of points on the curve (INRIM)
- 2 Gravimetric water content (GH2OC) (mgH_2O cm^{-3}soil)

4–9 **Column**
- 1 Input bulk density (TSTBD) (mg m^{-3})
- 2 Impedance value (TSTIMP)

Table 3. Sample initial soil fertility data file

Line*	Column* 1	2	3	4
1	'DUNDEE1.INT'			
2	"			
3	5	5	1.390000	100
4	3	7	1.110000	100
5	3	12	0.840000	100
6	2	18	0.630000	100
7	7	16	0.580000	100
8	8	20	0.580000	100
9	0	0	0.00E+00	100
10	0	0	0.00E+00	100
11	0	0	0.00E+00	100
12	0	0	0.00E+00	100
13	0	0	0.00E+00	100
14	0	0	0.00E+00	100

*Line and column numbers were added to help explain the file content.

Line

1 Initial soil fertility data file name (FILNAM)

2 Comment or description (DSCRIP)

3–14 **Column**
- 1 Residual ammonium content (RNNH4) (lbs acre^{-1})
- 2 Residual nitrate content (RNNO3) (lbs acre^{-1})
- 3 Percent organic matter (OMA)
- 4 Water content as a percentage of field capacity (ISOILW)

Table 4. Sample agricultural inputs (irrigation and fertilizer) data file

Line*	Column* 1	2	3	4	5	6	7	8
1	'COLE90.IRR'							
2	'Inputs for Coleman 1990 4th cultivation'							
3	'07/02/90'	0.000	0.000	0	2	0.750	0	
4	0.000	30.000	0.000	0.000	1	8	6	0

*Line and column numbers were added to help explain the file content.

Line

1 Agricultural inputs data file name (FILNAM)

2 Description (DSCRIP)

3 **Column**

1 Calendar date (CDATE)
2 Amount of irrigation water applied (H2OAMT) (inches)
3 Cost of irrigation water applied (CSTIRR)
4 Method of irrigation used (MTHIRR)
5 Depth of cultivation (ICDPTH) (inches)
6 Cost of cultivation (CSTCUL)
7 Tire width (ITRWTH) (inches)

4 **Column**

1 Amount of NH_4 fertilizer applied (AMTAMM) (lbs acre^{-1})
2 Amount of NO_3 fertilizer applied (AMTNIT) (lbs acre^{-1})
3 Amount of urea fertilizer applied (AMTURA) (lbs acre^{-1})
4 Total cost of fertilizer applied (CSTFRT)
5 Method of fertilizer application used (MTHFRT)
6 Horizontal placement of side-dressed fertilizer (ISDHRZ) (inches)
7 Vertical placement of side-dressed fertilizer (ISDDPH) (inches)
8 Pointer (POINTR)

Table 5. Sample weather data file

Julian day (days)	Calendar day (days)	Solar radiation (langleys)	Temperature (°F) Maximum	Temperature (°F) Minimum	Daily rainfall (inches)	Wind run (miles)
187	'07/06'	513	90	73	0.00	72.30
188	'07/07'	313	83	73	0.02	58.66
189	'07/08'	580	94	73	0.00	58.47
190	'07/09'	642	97	76	0.00	95.90

SUBROUTINE SOIL

Revised by S.B. Turner

The subroutine SOIL calls FRTLIZ, GRAFLO, ET, EVPSOIL, UPTAKE, CAPFLO, and NITRIF. SOIL controls the order in which the subroutines are called and the frequency with which they are called. It initializes certain variables, calculates several intermediate variables, and converts the units on some of them. It checks values for reasonableness and flags those which are out of bounds. It also adds variables to provide sums, such as total soil nitrogen in the soil profile; total soil and plant evaporation; total nitrate-nitrogen (nitrate-N or NO_3-N) and ammonium-nitrogen (ammonium-N or NH_4-N) taken up by roots; total volumetric nitrate-N, ammonium-N, and water content of the soil profile; and total water draining from the bottom of the soil profile or into the water table. SOIL takes daily time steps from crop emergence to the end of a season or some earlier specified termination in the simulation period.

ASSUMPTIONS

1. The number of times SOIL calls GRAFLO is determined by a ratio of the amount of rainfall over 12.7 mm. GRAFLO is called at least once whenever rainfall or irrigation occurs.

2. Normally, CAPFLO, EVPSOIL, and UPTAKE have 5 iterations daily, the fewest required for convergence in CAPFLO.

INPUTS

In addition to the general required inputs, SOIL requires the following:

ACELLDW
DAYNUM
DCELL
EMERGE
NEWEP
NEWES
POPFAC
RAIN
SUMSUB
SUPNH4
SUPNO3
TCELL
UPNH4
UPNO3
VH20C(L,K)
VNH4C(L,K)
VNO3C(L,K)
WCELL

OUTPUTS

CUMEP
CUMES
CUMRAN
SOILN
SUBIRR
SUPNH4
SUPNO3
TH20
TNNH4
TNNO3
UPTAKEN

Note: Array variables are designated by K and L, where K represents the soil cell column number and L, the soil cell row number.

PSEUDOCODE

Call FRTLIZ

Initialize to 0.0 the water uptake of all the NL*NK cells [ZUPT(L,K)] in the soil profile

Initialize to 0.0 the adjusted soil evaporation (NEWES), the excess water in the water table (SUMSUB), and the irrigation amount for COMAX (CMXIRR)

If there is rainfall or irrigation or both today (RAIN), then

> cumulate RAIN (CUMRAN)
>
> the number of iterations (NOITR) is a ratio of RAIN over 12.7 mm with a minimum value of 1 (**Note:** The 12.7 mm divisor is an arbitrary number to make GRAFLO drain the water faster.)
>
> redefine RAIN as RAIN divided by NOITR
>
> for each iteration, do:
>
>> call GRAFLO
>>
>> set the logical variable (FULPRO) to TRUE
>>
>> call CAPFLO
>
> end do
>
> convert RAIN to its original value

Endif

Call ET

Initialize to 0.0 the supply of ammonium-N (SUPNH4) and nitrate-N (SUPNO3) to plants from the soil

Initialize the total plant evaporation to 0.0

For each cell in the 40 × 21 array of the soil profile, initialize:

> the flux of nitrogen to the left [FNL(L,K)]
>
> the flux of nitrogen upward [FNU(L,K)]
>
> the flux of water to the left [FWL(L,K)]
>
> the flux of water upward [FWU(L,K)]

End do

Set the number of iterations (NOITR) to 5 (**Note:** In case there is no rainfall or irrigation, the number of iterations is normally limited to 5.)

For each of the iterations, do:

> if this is the first iteration, set FULPRO to TRUE, else set it to FALSE—the whole profile is not going to be considered
>
> call EVPSOIL
>
> if the simulation date (DAYNUM) is equal to or past the emergence date (EMERGE), then

call UPTAKE

if the plant's uptake of ammonium-N (UPNH4) and nitrate-N (UPNO3) from the cells is greater than 0.0, then accumulate the total supply of ammonium-N (SUPNH4) and nitrate-N (SUPNO3) to the plant from the soil

end if

call CAPFLO

End do

Call NITRIF

Accumulate the total nitrogen uptake of the plant (UPTAKEN) in lb acre^{-1}

Convert SUPNH4 and SUPNO3 from mg N plant^{-1} to g N dm^{-2}

Accumulate the total soil (CUMES) and plant (CUMEP) evaporation

Initialize to 0.0 the total volumetric water (TH2O), ammonium-N (TNNH4), and nitrate-N (TNNO3) content of the soil profile

For each cell in the NL $\times$ NK array of the soil profile, accumulate the total volumetric water (TH2O), ammonium-N (TNNH4), and nitrate-N (TNNO3) content of the soil profile

Convert TNNH4 and TNNO3 from mg N cm^{-3} to lb N acre^{-1}

Total soil nitrogen in the profile (SOILN) is the sum of TNNH4 and TNNO3

Convert TH2O from cm^3 H_2O cm^{-3} soil to depth of water in cm

Accumulate the drainage through the bottom of the profile or into the water table (SUBIRR)

Return to GOSSYM

End SOIL

SOURCE CODE

```
      SUBROUTINE SOIL
C  *************************************************************
C  *                                                           *
C  *     SOIL SUBROUTINE.  CALLS FRTLIZ, GRAFLO, ET,           *
C  *     UPTAKE, CAPFLO, AND NITRIF.                           *
C  *                                                           *
C  *************************************************************

      INCLUDE 'GOSCOM.FOR'

      CALL FRTLIZ

      DO 10 L=1,NL
        DO 10 K=1,NK
          ZUPT(L,K)=0.0
   10 CONTINUE

      NEWES = 0.
      SUMSUB = 0.
      CMXIRR=0.
      IF(RAIN.GT.0.0) THEN
         CUMRAN = CUMRAN + RAIN
         NOITR = IFIX(AMAX1(1.,RAIN/12.7)+0.5)
         RAIN=RAIN/FLOAT(NOITR)
         DO 20 ITER=1,NOITR
            CALL GRAFLO
            FULPRO=.TRUE.
            CALL CAPFLO
  20     CONTINUE
         RAIN = RAIN*FLOAT(NOITR)
      ENDIF
      CALL ET
C
      SUPNH4 = 0.
      SUPNO3 = 0.
      SUMEP = 0.
      DO 860 L=1,40
         DO 860 K=1,21
            FNL(L,K) = 0.
            IF(K.LE.20) FNU(L,K) = 0.
            IF(K.LE.11) FWL(L,K) = 0.
            IF(K.LE.10) FWU(L,K) = 0.
 860  CONTINUE

      NOITR = 5
      DO 60 ITER=1,NOITR
         IF(ITER.EQ.1) THEN
            FULPRO=.TRUE.
         ELSE
            FULPRO=.FALSE.
         ENDIF
         CALL EVPSOIL
         IF(DAYNUM.GE.EMERGE) THEN
            CALL UPTAKE
            IF(UPNO3.GT.0.) SUPNO3 = SUPNO3 + UPNO3
            IF(UPNH4.GT.0.) SUPNH4 = SUPNH4 + UPNH4
         ENDIF
         CALL CAPFLO
   60 CONTINUE

      CALL NITRIF
```

```
      UPTAKEN = UPTAKEN + ((SUPNO3 + SUPNH4)/ACELLDW)*(0.891*ACELLDW)

C .001 CONVERTS MG OF N TO GRAMS

      SUPNO3 = SUPNO3 * POPFAC * .001
      SUPNH4 = SUPNH4 * POPFAC * .001
      CUMES = CUMES + NEWES
      CUMEP = CUMEP + NEWEP

C TOTAL WATER PROFILE

      TH2O = 0.
      TNNO3 = 0.
      TNNH4 = 0.
      DO 14 L=1,NL
         DO 14 K=1,NK
            TNNO3 = TNNO3 + VNO3C(L,K)
            TNNH4 = TNNH4 + VNH4C(L,K)
            TH2O  = TH2O  + VH2OC(L,K)
 14   CONTINUE

C 0.891*D*W (0.891*ACELLDW) CONVERTS MG N/CM**3 ADDED THE WHOLE SOIL
C PROFILE TO LBS N /ACRE. (MG/.01 M**2) * (10000 M**2/2.475 ACRES)
C * (0.001 G/MG) * (LBS/454 G) = 0.891 LBS/ACRE

      TNNO3 = TNNO3 * 0.891*ACELLDW
      TNNH4 = TNNH4 * 0.891*ACELLDW
      SOILN = TNNO3 + TNNH4
      TH2O = (TH2O * DCELL *10.)/NK
      SUBIRR = SUBIRR + (SUMSUB * 10.)/(NK * WCELL * TCELL)

      RETURN
      END
```

GLOSSARY

ACELLDW	Cross-sectional area of a soil cell DCELL*WCELL (cm^2).
CMXIRR	Irrigation amount, use in COMAX (mm H_2O).
CUMEP	Cumulative plant transpiration (mm H_2O day^{-1}).
CUMES	Cumulative soil evaporation (mm H_2O day^{-1}).
CUMRAN	Cumulative amount of rainfall (mm H_2O).
DAYNUM	Day number of the year (Julian days).
DCELL	Depth of soil cell (5 cm).
EMERGE	Emergence date (Julian days).
FNL(L,K)	Flux of nitrogen from the cell to the left (mg N).
FNU(L,K)	Flux of nitrogen from the cell upward (mg N).
FULPRO	Flag indicating whether all parts of CAPFLO will operate during an iteration; set to TRUE immediately after GRAFLO and during the first iteration (logical variable).
FWL(L,K)	Flux of water from the cell to the left (cm^3 H_2O).
FWU(L,K)	Flux of water from the cell upward (cm^3 H_2O).
NEWEP	Adjusted value of transpiration from lack of soil moisture (mm H_2O day^{-1}).
NEWES	Adjusted value of evaporative losses from the soil surface (mm H_2O day^{-1}).
NK	Number of vertical columns of soil cells in the profile.
NL	Number of layers (horizontal rows) of soil cells in the profile.
NOITR	Number of iterations or times a subroutine is called per day.
POPFAC	Population factor (dm^2 $plant^{-1}$).
RAIN	Total water application for the day (IDAY), including rainfall or irrigation or both (mm H_2O).
SOILN	Total nitrogen in the soil profile (lb $acre^{-1}$).
SUBIRR	Cumulative drainage into the water table (cm^3 H_2O).
SUMEP	Cumulative transpiration (mm H_2O).
SUMSUB	Cumulative excess water in the water table (cm^3 H_2O).
SUPNH4	Supply of ammonium-N to plants from the soil (mg N day^{-1}).
SUPNO3	Supply of nitrate-N to plants from the soil (mg N day^{-1}).
TCELL	Thickness of a soil cell (1 cm).
TH2O	Total volumetric water content of the soil profile (cm^3 H_2O cm^{-3} soil).
TNNH4	Total ammonium-N in the soil profile (mg N).

TNNO3	Total nitrate-N in the soil profile (mg N).
UPNH4	Uptake of ammonium-N from the cell (mg N day^{-1}).
UPNO3	Uptake of nitrate-N from the cell (mg N day^{-1}).
UPTAKEN	Total nitrogen uptake (lb $acre^{-1}$).
VH2OC(L,K)	Volumetric water content of the cell (cm^3 H_2O cm^{-3} soil).
VNH4C(L,K)	Volumetric ammonium-N content of the cell (mg N cm^{-3} soil).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N cm^{-3} soil).
WCELL	Width of soil cell; 5 cm or row spacing/NK (cm).
ZUPT(L,K)	Water uptake of the cell (cm^3 H_2O cm^{-3} soil).

SUBROUTINE GRAFLO

Revised by S.B. Turner
Review chaired by G. Theseira

Called only on days with rain or irrigation, GRAFLO moves rain water and irrigation water into the soil profile by gravitational flow. It also moves nitrogen in solution by mass flow.

ASSUMPTIONS

1. Water moves vertically only.
2. As long as infiltrating water is available, water moves rapidly into the profile layer by layer, filling the soil to saturation.
3. If the available water is insufficient to move into the next layer, then water moves from the cells at the center of the profile to the outermost cells.
4. Nitrate moves in complete solution by mass flow.
5. There is horizontal uniformity of effective rainfall or irrigation water.

INPUTS

NK
NL
RAIN
THTS(L)
VCELL
VH2OC(L,K)
VNO3C(L,K)
WCELL

OUTPUTS

CUMNSOK
CUMSOK
SOAKN(K)
VH2OC(L,K)
VNO3C(L,K)

Note: Array variables are designated by K and L, where K represents the soil cell column number and L, the soil cell row number.

GENERAL PSEUDOCODE

For all the columns in the soil profile, initialize the amount of nitrate in solution moving by mass flow

Calculate the amount of water that will move through the profile

Start with the first layer

While the current layer is still not the bottom layer and more water is available to move, do:

> Calculate the water deficit of the layer
>
> If the available water to move in the layer is greater than or equal to the water deficit, then (uniformly move the water and nitrogen in solution on this layer)

For all the cells in this layer, do:

Calculate the amount of nitrate-N remaining in the cell and the amount soaking through to the cell below

Update the volumetric nitrate-N content of the cell

Set the water content of the cell to its saturated value

End do

Calculate the amount of water leaving this layer

Else—there is insufficient water to move into the next layer—then (move the water starting from the cells at the center of the profile and proceeding to the outermost cells)

Set the column indices to the left and to the right of the center of the profile

If there is enough water to move in the cells, then

Calculate the water deficit for the cells

If there is insufficient water to fill the cells, then update the water status of the left and right cells (assuming each cell will receive equal amounts); else set the water status of the left and right cells to saturated water content

Update the volumetric nitrate-N content of the left and right cells

Else, if the layer number is greater than 1, then add the amount of nitrate-N from the layer above to the volumetric nitrate-N content of the left and right cells

Endif

If this is the last layer and there is water to move through the profile, then

Initialize the cumulative nitrogen soaking through the free drainage boundary

Add all the nitrate-N in solution in all the cells in this last layer

Endif

Increment the layer number

End do

Update cumulative water percolation

Return to SOIL

End GRAFLO

PSEUDOCODE

For all the columns in the soil profile, initialize the amount of nitrate-N in solution moving by mass flow [SOAKN(K)]

Calculate the amount of water that will move through the profile (TH2OADD)

Start with the first layer

while the current layer is still not the bottom layer and more water is available to move, do:

set to 0.0 the sum of the current water status of the layer (SUMH2O)

accumulate the present water content of all the cells in the current layer

calculate the water deficit of the layer (H2ODEF)

if the available water to move is greater than or equal to the water deficit, then

*** Move the water uniformly on this layer. ***

calculate the maximum water content a cell can hold (H2OMAX)

for each cell in this layer, do:

calculate the amount of nitrate-N in the cell, including the amount it receives from the cell above (TOTNO3)

calculate the water content of the cell, including the amount added to it (TH2OCELL)

calculate the ratio of the maximum water content of the cell to its updated water content (PRCENT)

if the ratio is greater than 1, set the ratio to 1—no water or nitrogen will leave the cell

calculate the amount of nitrate-N remaining in the cell (NO3STAY)

calculate the amount of nitrate-N soaking through to the cell below [SOAKN(K)]

update the volumetric nitrate-N content of the cell [VNO3C(L,K)]

set the water content of the cell to its saturated value [VH2OC(L,K)]

end do

calculate the amount of water leaving this layer

else—there is insufficient water to move into the next layer—then

*** Distribute the water from the middle cells of the profile toward the rightmost and leftmost (boundary) cells. ***

set the values of K1 and K2 (indices of the middle cells of the profile)

if there is an odd number of columns, then

calculate the water deficit of the middle cell

if there is insufficient water to fill the middle cell, then—the water deficit is greater than the water received

update the water status of the cell

set to 0.0 the amount of water that will move out of this cell

else—there is enough water to move into the cell

update the water status of this cell to its saturated water content

calculate the amount of water left to move into adjacent cells

endif

update the volumetric nitrate-N content of the cell

redefine to 0.0 the amount of nitrate-N soaking through to the cell below [SOAKN(K1+1)]

endif

for all the cells in half of the profile, do:

set column indices to the left (KL) and right (KR) of the center of the profile

*** Move the water from the centermost cells of the profile to the outermost cells. ***

if there is enough water to move, then

calculate the water deficit for cells on columns KR and KL

if there is insufficient water to fill the cells, then

update the water status of the left and right cells, assuming each cell will receive equal amounts

set the remaining amount of water to move to 0.0

else

set the water status of the left and right cells to saturated water content

update the amount of water left to move to adjacent cells

endif

update the volumetric nitrate-N content of the left and right cells

redefine to 0.0 the amount of nitrate-N soaking into these cells below [SOAKN(KL) and SOAKN(KR)]

else

if the layer number is greater than 1, then

update the volumetric nitrate-N content of the left and right cells

redefine to 0.0 the amount of nitrate-N soaking through into these cells below [SOAKN(KL) and SOAKN(KR)]

endif

endif

end do

endif

if this is the last layer and there is water to move through the profile, then

initialize to 0.0 the cumulative nitrogen soaking through the free drainage boundary (CUMNSOK)

add together all the nitrate-N in solution in all the cells in this last layer

endif

increment the layer number

End do

Update the cumulative water percolation (CUMSOK)

Return to SOIL

End GRAFLO

```
      SUBROUTINE GRAFLO
C   ****************************************************************
C   *                                                              *
C   *   GRAVITY FLOW OF NO3 AND H2O, AFTER RAIN OR IRRIGATION.     *
C   *                                                              *
C   ****************************************************************
C RAIN OR IRRIGATION IS IN MM.

      REAL*4 NO3STAY

      INCLUDE 'GOSCOM.FOR'

      DO K=1,NK
         SOAKN(K) = 0.
      END DO
      TH2OADD = RAIN*.1*NK*WCELL*1
      L = 1
      DO WHILE ((L.LE.NL).AND.(TH2OADD.GT.0))

         SUMH2O = 0.
         DO K=1,NK
            SUMH2O = SUMH2O+VH2OC(L,K)*VCELL
         END DO

         H2ODEF = THTS(L)*VCELL*NK-SUMH2O
         IF(TH2OADD.GE.H2ODEF) THEN
            H2OMAX = THTS(L)*VCELL
            DO K=1,NK
               TOTNO3 = VNO3C(L,K)*VCELL+SOAKN(K)
               TH2OCELL = (TH2OADD-H2ODEF)/NK+H2OMAX
               PRCENT = H2OMAX/TH2OCELL
               IF(PRCENT.GT.1.) PRCENT = 1.
               NO3STAY = PRCENT*TOTNO3
               SOAKN(K) = TOTNO3-NO3STAY
               VNO3C(L,K) = NO3STAY/VCELL
               VH2OC(L,K) = THTS(L)
            END DO
            TH2OADD = TH2OADD-H2ODEF
         ELSE
            K1 = NK/2
            K2 = NK-2*K1
            IF(K2.NE.0) THEN
               H2ODEF = (THTS(L)-VH2OC(L,K1+1))*VCELL
               IF(TH2OADD.LT.H2ODEF) THEN
                  VH2OC(L,K1+1) = VH2OC(L,K1+1)+TH2OADD/VCELL
                  TH2OADD = 0.
               ELSE
                  VH2OC(L,K1+1) = THTS(L)
                  TH2OADD = TH2OADD-H2ODEF
               ENDIF
                 VNO3C(L,K1+1) = VNO3C(L,K1+1)+SOAKN(K1+1)/VCELL
                 SOAKN(K1+1)=0
            ENDIF
            DO K=1,K1
               KL = K1+1-K
               KR = K1+K+K2
               IF(TH2OADD.GT.0) THEN
                  H2ODEF = (THTS(L)-(VH2OC(L,KR)+VH2OC(L,KL))/2)*VCELL
                  IF(TH2OADD.LT.H2ODEF*2) THEN
                     VH2OC(L,KL) = VH2OC(L,KL)+0.5*TH2OADD/VCELL
                     VH2OC(L,KR) = VH2OC(L,KR)+0.5*TH2OADD/VCELL
                     TH2OADD = 0.
```

```
            ELSE
               VH2OC(L,KL) = THTS(L)
               VH2OC(L,KR) = THTS(L)
               TH2OADD = TH2OADD-H2ODEF*2
            ENDIF
            VNO3C(L,KL) = VNO3C(L,KL)+SOAKN(KL)/VCELL
            VNO3C(L,KR) = VNO3C(L,KR)+SOAKN(KR)/VCELL
            SOAKN(KL) = 0.
            SOAKN(KR) = 0.
         ELSE
            IF(L.GT.1) THEN
               VNO3C(L-1,KL) = VNO3C(L-1,KL)+SOAKN(KL)/VCELL
               VNO3C(L-1,KR) = VNO3C(L-1,KR)+SOAKN(KR)/VCELL
               SOAKN(KL) = 0.
               SOAKN(KR) = 0.
            ENDIF
         ENDIF
      END DO
   ENDIF
   IF((L.EQ.NL).AND.(TH2OADD.GT.0.)) THEN
      CUMNSOK=0.
      DO I=1,NK
         CUMNSOK=CUMNSOK+SOAKN(I)
      END DO
   ENDIF
   L=L+1
END DO
CUMSOK = CUMSOK+TH2OADD/.10/(NK*WCELL*1)
RETURN
END
```

GLOSSARY

CUMNSOK	Cumulative nitrate-N leached from the profile (mg N).
CUMSOK	Cumulative water percolated from the profile (mm H_2O).
H2ODEF	Amount of water the layer can accept; unfilled pore volume (cm^3 H_2O).
H2OMAX	Maximum amount of water that a soil cell can hold (cm^3 H_2O).
K1, K2	Indices of the middle cells of the profile.
KL	Soil cell column marker, located in the left half of the profile.
KR	Soil cell column marker, located in right half of the profile.
NK	Number of vertical columns of soil cells in the profile.
NL	Number of layers (horizontal rows) of soil cells in the profile.
NO3STAY	Amount of nitrate-N that can remain in the cell (mg N).
PRCENT	Fraction of total water that the cell can contain.
RAIN	Total water application for the day (IDAY), including rainfall or irrigation or both (mm H_2O).
SOAKN(K)	Amount of nitrate-N leaching from the profile (mg N).
SUMH2O	Sum of the volumetric water content of the cells (cm^3 H_2O).
TH2OADD	Total water added to the soil (cm H_2O).
TH2OCELL	Total water to be taken into account for any given cell (cm^3 H_2O).
THTS(L)	Saturated volumetric water content for the layer (cm^3 H_2O cm^{-3} soil).
TOTNO3	Total nitrate-N in the soil profile (mg N).
VCELL	Volume of a soil cell, DCELL*WCELL*TCELL (cm^3).
VH2OC(L,K)	Volumetric water content of the cell (cm^3 H_2O cm^{-3} soil).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N cm^{-3} soil).
WCELL	Width of a soil cell (cm).

SUBROUTINE CAPFLO

Revised by S.B. Turner
Review chaired by D.O. Porter

CAPFLO simulates the horizontal and vertical "capillary" fluxes of water and nitrate nitrogen in the soil profile. As water is applied to or removed from the soil, gradients in the volumetric water content develop. CAPFLO redistributes water and nitrogen in response to these cell-to-cell gradients.

The SOIL subroutine calls CAPFLO after each call to the gravitational flow routine, GRAFLO. Calls to GRAFLO occur on days in which water is applied through rainfall or irrigation. GRAFLO and CAPFLO are called once for each 1/2 inch of rainfall or irrigation water applied per day. CAPFLO is also called up to 5 additional times a day, as long as water flux within the profile is significant.

Some important functions in CAPFLO include the Gardner-Mayhugh diffusivity function, the Richards equation (Hillel 1980) for water flux, and the Marani equation for the soil-moisture-release curve.

The Gardner-Mayhugh function (Gardner and Mayhugh 1958) follows the form:

$$D(\Theta) = ae^{b\Theta}$$

where

D = diffusivity of water in the soil medium ($cm^2\ day^{-1}$)
Θ = volumetric soil water content (cm^3 soil $cm^{-3}\ H_2O$)
a = interception of the diffusivity-water content curve
b = slope of the diffusivity-water content curve
e = natural base of the logarithm.

The Richards water flux equation (one dimensional) (Prasad and Romkens 1982) may be written as

$$\frac{\delta\Theta}{\delta t} = \frac{\delta}{\delta z} D \frac{\delta\Theta}{\delta z} - \frac{\delta K}{\delta z} \qquad z > 0,\ t > 0$$

where

Θ = soil water content (cm^3 soil $cm^{-3}\ H_2O$)
z = vertical depth (cm)
K = hydraulic conductivity (cm day^{-1})
D = diffusivity of water in soil medium ($cm^2\ day^{-1}$)
t = time (days).

The Marani soil-moisture-release equation (Marani, personal communication, 1983) follows the form

$$\Theta_i = \Theta_{AD} + (\Theta_{FC} - \delta_{AD})\,(h_{FC}/h_i)^{TEMP1}$$

where

$$TEMP = \frac{Ln(-15/h_{FC})}{Ln((\Theta_r - \Theta_{AD})/(\Theta_{FC} - \Theta_{AD}))}$$

Θ_r = residual (15 bar) water content (cm^3 soil $cm^{-3}\ H_2O$)
Θ_{FC} = water content at field capacity (cm^3 soil $cm^{-3}\ H_2O$)
Θ_{AD} = air-dry water content (cm^3 soil $cm^{-3}\ H_2O$)
Θ_i = current water content (cm^3 soil $cm^{-3}\ H_2O$)

h_i = current soil water potential (bar)
h_{FC} = soil water potential at field capacity (bar)
Ln = natural log.

The major processes in CAPFLO perform the following functions:

- Initialize the soil profile and determine the potential volume of water that may be moved.
- Calculate diffusivity for each soil cell by application of the Gardner-Mayhugh diffusivity function.
- Set the lower boundary condition.
- Calculate horizontal water flux and horizontal nitrate-N flux. Water flux is determined through application of a simplified Richards equation.
- Calculate vertical water flux and vertical nitrate-N flux. The Richards equation is applied again.
- Calculate total water and nitrate-N fluxes from each cell. These totals must be within established limits.
- Update the water potential value for each cell. This is accomplished by applying the Marani equation of the soil-moisture-retention curve to locate water potential for the "new" volumetric water content of each cell.

ASSUMPTIONS

1. Due to symmetry, potential horizontal and vertical water flux may be calculated for one-half of the soil profile; the results may be imposed on the other half by reflecting the "mirror image" of the first half.
2. The potential horizontal and vertical nitrogen flux must be calculated for the entire profile. Symmetry may not be assumed.
3. The horizontal movement of water and nitrogen may be calculated separately from the vertical movement. Horizontal movement is calculated before vertical movement is determined.
4. The flux of nitrogen is directly proportional to the flux of water.
5. Water flux into a cell cannot exceed 25 percent of the cell's receiving capacity (air-filled voids) from any direction.

INPUTS

AIRDR(J)
ARDRCN(L)
BDRATO
BDSLOP
BETAI
BETA(J)
CUMNSOK
DCELL
DIFF0I
DIFF0(J)
FBLOOM
FCININ(J)
FLNMIN
FLXMAX(J)
FLXMIN(J)

FULPRO
IDAY
IPCLAY(J)
LYRDPH(L)
NK
NL
PSISFC
PSIS(L,K)
RTEXNT(L)
SUMSUB
TCELL
TD
THETA0(J)
THETAI
THETAR(J)
THTA0I
THTS(L)
TTUPF(L,K)
VCELL
VH2OC(L,K)
VNO3C(L,K)
WATTBL
WCELL

OUTPUTS

CUMNSOK
DIFF(L,K)
PSIS(L,K)
SKPFLG
SUMSUB

Note: Array variables are designated by K and L, where K represents the soil cell column number and L, the soil cell row number.

GENERAL PSEUDOCODE

Initialize the skip flag

Calculate the layer and column indices

Calculate the available water (**Note**: This water may be moved to other cells.)

Calculate the diffusivity for each soil cell in the left half of the profile

Recalculate the available water and diffusivity for the first three cells in the top layer

If there are new roots in the profile or the full profile flag is true or both, then

  Set the water table if it falls within the profile or at the lower boundary

  If there was uptake on the last iteration or if the full profile flag is true, then

    Calculate the horizontal water flux from each cell

    If the water flux is significant, set the skip flag to true

    If flux is very close to 0, set the water and nitrogen flux to 0

    If the water flux is greater than the maximum permitted value, limit the flux to the

positive or negative maximum value

If the horizontal water flux is greater than 0, calculate the horizontal nitrate-N flux for the right and left halves of the profile

Calculate the vertical water flux from each cell

If the flux is very close to 0, set the water and nitrogen flux to 0

If the water flux is greater than the maximum permitted value, limit the flux to the positive or negative maximum value

If the vertical water flux is greater than 0, calculate the vertical nitrate-N flux for the right and left halves of the profile

If there is no water table within the profile, calculate the lower boundary condition

If the depth to the water table is not greater than the depth of the profile, set the lower boundary condition to the saturated water content of the lowest cell layer (cell row NL)

Calculate the maximum water flux that may move in each direction from each cell

If the flux is very close to 0, set the water and nitrogen flux to 0 to prevent underflow

If the water flux is greater than the maximum permitted value, limit the flux to the positive or negative maximum value

Calculate the total water and nitrogen fluxes for each cell

Update the value of psi (soil water potential) for each cell

Return to SOIL

End CAPFLO

PSEUDOCODE

*** SKPFLG controls the number of times CAPFLO will be run per day. If there is a substantial amount of water movement into or out of any cell, SKPFLG will be set to TRUE. If SKPFLG remains FALSE, CAPFLO will not be called again. ***

Initialize SKPFLG to FALSE.

Calculate the necessary indices:

NLM1 = number of soil cell layers minus 1
NKH = number of half the columns
NKMM1 = number of columns minus 1
NKHP1 = number of half the columns plus 1
NKHP2 = number of half the columns plus 2

Calculate the fraction of the day that will be processed during each iteration (DELT)
[**Note:** The number of iterations per day (NOITR) has a maximum value of 5.]

*** Calculation of the diffusivity (Gardner and Mayhugh 1958) of each soil cell is necessary only for the left half of the profile because the profile is assumed to be symmetrical. ***

For each of the soil cells in the left half of the profile, do:

set J as the index for the soil horizon number

calculate the amount of water available for movement from the cell (DUMY02)

if the amount of water available for movement is negligible, then set the diffusivity of the cell [DIFF(L,K)] to the diffusivity of the horizon at −15,000 cm potential [DIFF0(J)]; else calculate the diffusivity of the soil cell

End do

*** This section of code is for the first 3 cells in layer 1 (the surface layer of cells) and is required when CULVAT, the cultivation routine, is applied. It is assumed that the values will be unaffected by cultivation. ***

For the first 3 cells of layer 1, do:

calculate the amount of water available for movement from the cell (DUMY02)

if the amount of water available for movement is negligible, then set the diffusivity of the cell [DIFF(L,K)] to the diffusivity of the horizon at −15,000 cm potential [DIFF0(J)]; else calculate the diffusivity of the soil cell

End do

*** Begin calculation of water and nitrogen flux. ***

For each soil cell layer, do:

if there are no roots in the layer and it is not time to consider the full profile, skip to *[A]* (page 39)

set J as the index for the soil horizon number

for each cell in the left-hand half of the layer, do:

*** Set the water table by redefining the volumetric water content of the cells below the water table (Whisler, personal communication, 1983). ***

initialize the time to adjust the water table (IT0)

if first bloom has occurred, set IT0 to the time of first bloom (FBLOOM), plus the time-day factor (TD), read from the soil hydrology file

if the current layer (L) is not the bottom layer and the depth to the water table (WATTBL) has been reached, then

set the intermediate variable (SAVEH2O) to the volumetric water content of the soil layer L+1 and column K [VH2OC(L+1,K)]

if today is past IT0, then

if VH2OC(L+1,K) is greater than the product of the volumetric water content at field capacity (THETAI) and the boundary slope ratio (BDRATO), then

VH2OC(L+1,K) is a function of THETAI, boundary slope (BDSLOP), and time

else

VH2OC(L+1,K) is the product of THETAI and BDRATO

endif

else

VH2OC(L+1,K) equals THETAI

endif

update the cumulative value of water added to the water table (SUMSUB)

endif

define the column index of the mirror-image cell, NKK (**Note:** Doing this locates the cell onto which the water conditions of the current cell will be reflected.)

if the subroutine UPTAKE removed water from the cell on the previous iteration or if this iteration covers the entire profile, then

*** Begin calculations of horizontal water and nitrogen fluxes. ***

if this soil cell is not in the first column, then

if the sum of the diffusivities of two adjacent cells is greater than 0.0, then

calculate the flux of water to the left [FWL(L,K)] out of the cell (cm^3 H_2O) as a function of the arithmetic mean of the diffusivities and the difference between volumetric water contents of the two cells

else

set FWL(L,K) to 0.0

endif

calculate the intermediate variable (WOUT) and the dimensionless flow value (FLOW) (**Note**: WOUT is the difference between the volumetric water contents of the two adjacent cells.)

*** The following code prevents a cell from receiving more water from other cells than it can hold. It does not allow more than 25 percent of the cell's receiving capacity to enter from any direction (Siefker, personal communication, 1986). ***

if WOUT is less than 0.0 (the current cell has a higher volumetric water content than the cell on its right), then

*** Calculate the flux of water to the right. ***

calculate the ratio of the volumetric water contents of the two adjacent cells (RATIO) (**Note:** This term prevents instabilities that arise when too much water is moved during one iteration.)

calculate the maximum amount of water that a soil cell can receive from all four directions (right, left, up, and down) by capillary action per iteration (H2OMAX)

if FLOW is greater than 25 percent of H2OMAX, then FWL(L,K) is 25 percent of H2OMAX (**Note**: A negative value indicates water flux to the right into the soil cell.)

else

*** Calculate the flux of water to the left ***

calculate the ratio of the volumetric water contents of the two adjacent cells (RATIO)

calculate the maximum amount of water that a soil cell can receive from all four directions (right, left, up, and down) by capillary action per iteration (H2OMAX)

if FLOW is greater than 25 percent of H2OMAX, then limit FWL(L,K) to 25 percent of H2OMAX

endif

if the flux is more than the minimum (FLXMIN, calculated in the INITIALIZE subroutine with a minimum value of 0.0001), set SKPFLG to TRUE—water movement is appreciable

if the flux is less than the minimum, then

*** The following code prevents floating-point underflow. ***

set FWL(L,K) to 0.0

set to 0.0 the flux of nitrogen to the left from the cell [FNL(L,K)] and the flux in its mirror-image cell [FNL(L,NKK)]

else

if the water flux is greater than the maximum allowable (FLXMAX, calculated in INITIALIZE), set the water flux to FLXMAX if greater than 0.0 or -FLXMAX if negative

*** Calculate the flux of nitrogen to the left out of the cell, FNL(L,K). The amount of nitrogen leaving the cell is proportional to the percentage of water leaving the cell. The calculation of nitrogen flux is based on VNO3C and VH2OC of the cell from which it is removed. ***

if the water flux is greater than 0.0, then calculate the nitrogen flux of the current soil cell (layer L and column K) as a function of VNO3C and VH2OC of the current soil cell (L, K); else the nitrogen flux of the current soil cell (L,K) is a function of VNO3 and VH2OC of the previous soil cell (L,K-1)

if the nitrogen flux is less than the minimum value of 0.000001, set the nitrogen flux to 0.0

*** The flux of nitrogen to the right must be treated separately to accommodate sidedress fertilizer placements. ***

if the water flux is greater than 0.0, then calculate the nitrogen flux of the mirror-image soil cell (L,NKK) as a function of VNO3C of the current soil cell (L,K); else the nitrogen flux of cell (L,NKK) is a function of VNO3C of the previous cell (L,K-1)

if the nitrogen flux is less than the minimum (0.000001), set the nitrogen flux to 0.0

endif

endif

*** End calculations of horizontal water and nitrogen fluxes. Begin calculations of vertical water and nitrogen fluxes. ***

if this layer is not the last layer, then

if the sum of the diffusivity of two adjacent cells is greater than 0.0, then calculate the flux of water upward [FWU(L+1,K)] out of the cell; else set FWU(L+1,K) to 0.0

calculate the intermediate variable (WOUT)

if WOUT is less than 0.0, the current soil cell has more water than the cell below it, then

*** Calculate the downward water flow into the soil cell (L+1,K). ***

calculate the ratio of the volumetric water contents of the two adjacent cells (RATIO)

calculate the maximum amount of water that a soil cell can receive from surrounding cells (H2OMAX)

if FLOW is greater than 25 percent of H2OMAX, then set FWU(L+1,K) to 25 percent of FWU(L+1,K) [**Note:** A negative value indicates water flux downward into soil cell (L+1,K).]

else

*** Calculate the flow of water upward into the soil cell (L,K). ***

calculate the ratio (RATIO) of the volumetric water contents of the adjacent cells (L,K) and (L+1,K)

calculate the maximum amount of water that a cell can receive from surrounding cells (H2OMAX)

if FLOW is greater than 25 percent of H2OMAX, then limit FLOW to 25 percent of H2OMAX [**Note:** A positive value indicates movement from cell (L+1,K) into cell (L,K).]

endif

if the flux is less than the minimum allowable, then

set to 0.0 FWU(L+1,K) and also the flux of nitrogen upward from soil cell (L+1,K) and from mirror-image cell (L+1,NKK)

else

set SKPFLG to TRUE—water movement is appreciable

if the water flux is greater than the maximum (FLXMAX), set the flux to FLXMAX if positive or −FLXMAX if negative

*** Calculate the flux of nitrogen upward from soil cell FNU(L+1,K) into soil cell FNU(L,K). The calculation must be based on the values of VNO3C and VH2OC of the cell from which nitrogen is transported. ***

if the water flux is greater than 0.0, then calculate the nitrogen flux of soil cell (L+1,K) as a function of VNO3C(L+1,K) and

VH2OC(L+1,K); else the nitrogen flux is a function of VNO3C(L,K) and VH2OC(L,K)

if the nitrogen flux is less than the minimum value of 0.000001, set the nitrogen flux to 0.0

*** The right side of the profile must be treated separately to accommodate sidedress fertilizer placements; symmetry cannot be assumed. ***

if the water flux is greater than 0.0, then calculate the nitrogen flux from soil cell (L+1,NKK) as a function of VNO3C(L+1,NKK) and VH2OC(L+1,K); else calculate the nitrogen flux from soil cell (L,NKK) as a function of VNO3C(L,NKK) and VH2OC(L,K)

if the nitrogen flux is less than the minimum value of 0.000001, set it to 0.0

endif

*** End calculations of vertical water and nitrogen fluxes. ***

else

*** Begin free drainage calculations. ***

if the water table is below the lower boundary, then

calculate the water content (H2OCNT) of the soil below the boundary (Whisler, personal communication, 1990)

if H2OCNT is greater than field capacity of the horizon at the bottom of the profile [FCININ(J)], then set H2OCNT to FCININ(J)

if H2OCNT is less than the volumetric water content at −15,000 cm (H2O) of the soil horizon at the bottom of the profile (horizon J) THETA0(J), then set H2OCNT to THETA0(J)

else

set H2OCNT to the saturated volumetric water content of the bottom soil cell layer—boundary layer [THTS(NL)]

endif

if the diffusivities of the soil cells in the bottom cell layer are greater than 0, then calculate the flux of water upward from the soil that is immediately below the boundary layer at 200 cm (NL+1 cell layer); else set the water flux to 0.0

calculate the intermediate variable (WOUT) and the dimensionless flow (FLOW)

if WOUT is less than 0.0, cell (L,K) has more water than cell (L+1,K), then

*** Calculate the downward water flux from cell (L,K). ***

calculate the ratio (RATIO) of the volumetric water contents of cells (L,K) and (L+1,K)

calculate the maximum amount of water (H2OMAX) that cell (L+1,K) can receive from surrounding cells

if FLOW is greater than 25 percent of H2OMAX, then FWU(L+1,K) is 25 percent of H2OMAX **[Note:** A negative value indicates flow direction from cell (L,K) to cell (L+1,K).]

else if the water table is within the modeled profile (within 200 cm of the soil surface)

*** Calculate the upward water flux from cell (L+1,K) to cell (L,K). ***

calculate the ratio (RATIO) of the volumetric water contents of cells (L,K) and(L+1,K)

calculate the maximum amount of water (H2OMAX) that cell (L+1,K) can receive from surrounding cells

if FLOW is greater than 25 percent of H2OMAX, then FWU(L+1,K) is 25 percent of H2OMAX **[Note:** A positive value indicates flow direction from cell (L+1,K) to cell (L,K).]

else

set the water flux to 0.0.

endif

update the cumulative value of water added to the water table (SUMSUB)

if the water flux is less than the minimum water flux, then

*** The following code prevents floating-point underflow. ***

set to 0.0 the flux of water upward [FWU(L+1,K)] and also the flux of nitrogen upward from the cell [FNU(L+1,K)] and from the mirror-image cell [FNU(L+1,NKK)]

else

set SKPFLG to TRUE—water flux is appreciable

if the water flux is greater than the maximum (FLXMAX), then set the flux to FLXMAX if positive or −FLXMAX if negative

*** Calculate the flux of nitrogen upward from cell (L+1,K) to cell (L,K). **[Note**: This directional movement "upward" is labeled to correspond with the directional movement of water flux. A negative value of FNU(L+1,K) indicates downward nitrogen flux. Nitrogen flux is limited to downward movement only. Movement from below the profile upward is not allowed.] ***

calculate the "upward" nitrogen flux from soil cell (L+1,K)

if nitrogen flux from cell (L+1,K) is either less than the minimum allowable value of 0.000001 or has a positive value (flux upward), set the flux to 0.0

calculate the "upward" nitrogen flux from soil cell (L+1,NKK)

if nitrogen flux from cell (L+1,NKK) is either less than the minimum value of 0.000001 or has a positive value (flux upward), set the flux to 0.0

update the value of CUMNSOK, which stores the cumulative nitrogen that has leached downward across the lower boundary

endif

*** End free drainage calculations. ***

endif

endif

end do

End do

[A] Continue

*** Calculate total water and nitrogen fluxes. ***

For each soil layer, do:

if there are no roots in the layer and it is not time to consider the full profile, skip to *[B]* (see page 40)

set J as the index for the soil horizon number

for each soil cell in the left half of the profile, by column, do:

define the column index of the mirror-image cell, NKK

if there was water removed in UPTAKE during the previous iteration or if this iteration is to include the entire profile, then

calculate the net flux of water into the cell (FWICN) by adding the fluxes in all directions

update the volumetric water content of the cell [VH2OC(L,K)]

if VH2OC(L,K) is greater than the saturated water content of the layer, set it to the saturated water content of the layer

if VH2OC(L,K) is less than the air-dry water content of the horizon, then set it to the air-dry water content of the horizon

set the water content of the mirror-image cell to VH2OC(L,K)

if this is the first column for each layer, then

*** Allow for the no-flow vertical boundary condition. ***

calculate the net flux of nitrogen from all directions (FNICN) for the left and right vertical boundaries

update the volumetric nitrogen content of these boundary cells, taking into account the lower limit of FLNMIN

else, if this cell is one of the two middle cells of the layer, then

calculate the net flux of nitrogen from all directions (FNICN) for this cell and its mirror-image cell

update the volumetric nitrogen content of these center cells, taking into account the lower limit of FLNMIN

else

calculate the net flux of nitrogen from all directions (FNICN) for the rest of the soil cells

update the volumetric nitrogen content of these cells, taking into account the lower limit of FLNMIN

endif

endif

end do

End do

[B] Continue

*** Update the value of the soil water potential of each soil cell. ***

For each soil layer, do:

set J as the index for the soil horizon number

for each of the soil columns, do:

define the column index of the mirror-image cell

initialize the soil water potential to −15 bars (−15,000 cm)

if VH2OC(L,K) is greater than the residual water content of the soil horizon of this cell, then calculate the soil water potential of this cell using the Marani equation

set the upper limit to −0.00001 and the lower limit to −15 bars (−15,000 cm)

set the soil water potential and diffusivity values of the mirror-image cells

end do

End do

Return to SOIL

End CAPFLO

```
      SUBROUTINE CAPFLO
C   ***************************************************************
C   *                                                             *
C   *  CAPILLARY FLOW OF NO3 AND H2O IN ALL DIRECTIONS.           *
C   *                                                             *
C   ***************************************************************
C AFTER ITERATIONS DURING THE DAY HAVE BEEN ADDED.                *
C FLUX OF H2O TO THE LEFT OUT OF THE CELL, CM**3/CELL/DAY.        *
C FLUX OF H2O UPWARD OUT OF THE CELL, CM**3/CELL/DAY.             *
C FLUX OF NITROGEN TO THE LEFT OUT OF THE CELL, MG N/CELL/DAY.*
C FLUX OF NITROGEN UPWARD OUT OF THE CELL, MG N/CELL/DAY.         *
C ****************************************************************

      INCLUDE 'GOSCOM.FOR'

C SKPFLG CONTROLS THE NUMBER OF ITERATIONS OF CAPFLO THAT ARE
C  NECESSARY PER DAY.  IF THERE IS A SUBSTANTIAL AMOUNT OF WATER
C  MOVEMENT INTO OR OUT OF ANY CELL, SKPFLG WILL BE SET TO TRUE.
C  IF IT REMAINS FALSE, THEN CAPFLO WILL NOT BE CALLED AGAIN IN
C  SOIL.

      SKPFLG = .FALSE.

C CALCULATE NECESSARY INDICES
      NLM1 = NL-1
      NKH = NK/2
      NKMM1 = NK-1
      NKHP1 = NKH + 1
      NKHP2 = NKH + 2

C CALCULATE FRACTION OF DAY THAT WILL BE PROCESSED DURING THIS
C  ITERATION.

      DELT = 1.0 / NOITR

C CALCULATE DIFFUSIVITY OF EACH SOIL CELL IN LEFT HALF OF PROFILE.
C ASSUME NO FLOW ACROSS VERTICAL BOUNDARY DUE TO SYMMETRY, UNDER ROW
C AND MIDWAY BETWEEN ROWS. ASSUME NO FLUX INTO ROOT ZONE FROM BENEATH.
C  DIFFUSIVITY FUNCTION FOUND IN :GARDNER AND MAYHUGH. 1966.
C  SSSAP 22:197-201. FDW.

      DO 4 L = 1, NL
         J = LYRDPH(L)
         DO 4 K = 1, NKH

C              CALCULATE AMOUNT OF WATER THAT CAN MOVE. THEN
C              IF AMOUNT NEGLIGIBLE, SET DIFFUSIVITY TO DIFFO FOR LAYER
C                   ELSE CALCULATE IT.

            DUMY02 = VH2OC(L,K)-THETA0(J)
            IF(DUMY02.LT.0.00001)THEN
               DIFF(L,K)=DIFF0(J)
            ELSE
               DIFF(L,K) = DIFF0(J)*EXP(BETA(J)*DUMY02)
            ENDIF
    4 CONTINUE

C  THE FIRST 3 CELLS IN LAYER 1 ARE ASSUMED UNAFFECTED BY
C    CULTIVATION.  CALCULATE DIFF FOR THESE CELLS FROM
C    INITIAL DIFFO VALUE.

      DO 150 K=1,3
```

```
          DUMY02 = VH2OC(1,K)-THTA0I
          IF(DUMY02.LT.0.00001)THEN
             DIFF(1,K)=DIFF0I
          ELSE
             DIFF(1,K) = DIFF0I*EXP(BETAI*DUMY02)
          ENDIF
 150  CONTINUE

C     CALCULATE POTENTIAL FLUX OF WATER LEFT AND UP IN LEFT HALF OF
C     SOIL PROFILE.  CALCULATE FLUX OF NITROGEN LEFT AND UP IN BOTH
C     HALVES OF PROFILE

      DO 5 L  = 1, NL
C        IF NO ROOTS IN PREVIOUS LAYER AND NOT TIME TO CONSIDER
C           THE FULL PROFILE, EXIT THIS LOOP
         IF((.NOT.RTEXNT(L)).AND.(.NOT.FULPRO)) GO TO 4406

         J = LYRDPH(L)
C        ONLY LEFT HALF OF MATRIX IS NECESSARY DUE TO SYMMETRY.
         DO 5 K = 1, NKH

C  SET WATER TABLE.
C  START WITH THE BOTTOM LAYER AND PROCESS THE WATER TABLE FROM THE
C  BOTTOM UPWARD UNTIL A ABOVE THE WATER TABLE.

            ITO = IDAY + 1
            IF(FBLOOM.GT.0.) ITO = FBLOOM + TD
            IF((L.LT.NL).AND.(WATTBL.LT.DCELL*(L+1))) THEN
               SAVEH2O = VH2OC(L+1,K)
               IF(IDAY.GT.ITO) THEN
                  IF(VH2OC(L+1,K).GT.BDRATO*THETAI) THEN
                     VH2OC(L+1,K) = THETAI-BDSLOP*(IDAY-ITO)
                  ELSE
                     VH2OC(L+1,K) = THETAI*BDRATO
                  ENDIF
               ELSE
                  VH2OC(L+1,K) = THETAI
               ENDIF
               SUMSUB = SUMSUB + (VH2OC(L+1,K) - SAVEH2O)*VCELL*2
            ENDIF
C             NKK = COLUMN INDEX OF MIRROR IMAGE CELL
            NKK = NK + 1 - K

C             IF THERE WAS UPTAKE FROM THE CELL ON THE PREVIOUS
C                ITERATION OR IF THIS IS AN ITERATION OVER ENTIRE
C                PROFILE
            IF(TTUPF(L,K).OR.FULPRO) THEN

               IF (K .NE. 1) THEN
C                    USE HARMONIC MEAN OF DIFFUSIVITY TO
C                    CALCULATE H2O FLOW  K = (2*D1*D2)/(D1+D2)
C                    FWL = FLUX OF WATER TO THE LEFT. CM**3

                  IF(DIFF(L,K)+DIFF(L,K-1).GT.0.00001)THEN
                     FWL(L,K)=(DIFF(L,K)+DIFF(L,K-1))/2.
     .                         *((VH2OC(L,K)-VH2OC(L,K-1))/WCELL)
     .                         * (DCELL*TCELL*DELT)
                  ELSE
                     FWL(L,K)=0.0
                  ENDIF

C                 H2OMAX IS THE MAXIMUM AMOUNT OF WATER THAT A CELL CAN
```

```
C                RECEIVE FROM ALL DIRECTIONS (LEFT,RIGHT,DOWN,UP) BY
C             CAPILLARY ACTION PER ITERATION.  THTS(L) IS THETAS BY
C                 LAYER

            WOUT = VH2OC(L,K) - VH2OC(L,K-1)
C                 CALCULATE DIMENSIONLESS FLOW (CC/CC)
            FLOW = FWL(L,K)/VCELL

C                THE FOLLOWING CODE WAS INSERTED BY JIM SIEFKER TO
C              PREVENT MORE WATER FROM FLOWING INTO A CELL THAN THE
C              CELL CAN HOLD. IT WILL NOT ALLOW MORE THAN 25% OF
C                 THE CAPACITY OF THE CELL TO COME IN FROM ANY ONE
C           DIRECTION.  THE RATIO TERM PREVENTS INSTABILITIES
C                 THAT ARISE WHEN TOO MUCH WATER IS MOVED ON ONE
C                 ITERATION.

            IF(WOUT.LT.0.0) THEN
C                 CALCULATE THE FLOW OF WATER TO THE RIGHT
               RATIO = VH2OC(L,K)/VH2OC(L,K-1)
               H2OMAX = (THTS(L) - VH2OC(L,K))
     .                  *(1.0-RATIO)
               IF(ABS(FLOW).GT.(0.25 * H2OMAX)) THEN
                  FWL(L,K) = -0.25*H2OMAX*VCELL
               ENDIF
            ELSE
C                  CALCULATE THE FLOW OF WATER TO THE LEFT
               RATIO = VH2OC(L,K-1)/VH2OC(L,K)
               H2OMAX = (THTS(L) - VH2OC(L,K-1))
     .                  *(1.0-RATIO)
               IF(ABS(FLOW).GT.(0.25*H2OMAX)) THEN
                  FWL(L,K) = 0.25*H2OMAX*VCELL
               ENDIF
            ENDIF

C                  IF MORE THAN MIN WATER FLUX SET SKPFLG TO TRUE
            IF(ABS(FWL(L,K)).GT.FLXMIN(J)) SKPFLG = .TRUE.

C           FOLLOWING CODE PREVENTS FLOATING POINT UNDERFLOW
            IF(ABS(FWL(L,K)).LE.FLXMIN(J)) THEN
               FWL(L,K) = 0.0
               FNL(L,K) = 0.0
               FNL(L,NKK) = 0.0
            ELSE
               IF(ABS(FWL(L,K)).GT.(ABS(FLXMAX(J))))THEN
                  IF(FWL(L,K).LT.0.)FWL(L,K) = -FLXMAX(J)
                  IF(FWL(L,K).GT.0.)FWL(L,K) =  FLXMAX(J)
               ENDIF

C                     FLOW OF NO3 TO THE LEFT, OUT OF CELL,
C                        MG N/CELL/DAY. FLOW OF NITROGEN IS
C                        PROPORTIONAL TO FLOW OF WATER
               IF (FWL(L,K) .GT. 0.0) THEN
                  FNL(L,K) = FWL(L,K)
     .                       * (VNO3C(L,K)/VH2OC(L,K))
               ELSE
                  FNL(L,K) = FWL(L,K)
     .                       * (VNO3C(L,K-1)/VH2OC(L,K-1))
               ENDIF
               IF(ABS(FNL(L,K)).LT.FLNMIN) FNL(L,K)=0.0

C                 MUST LOOK AT RIGHT HAND SIDE OF PLANE SEPARATELY
C                      DUE TO SIDE DRESSING BEING UNSYMMETRICAL
```

```
            IF (FWL(L,K) .GT. 0) THEN
               FNL(L,NKK) = FWL(L,K)
     .                      *(VNO3C(L,NKK)/VH2OC(L,K))
            ELSE
               FNL(L,NKK) = FWL(L,K)
     .                      *(VNO3C(L,NKK+1)/VH2OC(L,K-1))
            ENDIF
            IF(ABS(FNL(L,NKK)).LT.FLNMIN)
     .      FNL(L,NKK)=0.0
         ENDIF
      ENDIF

C           CALCULATE FLUX OF WATER UPWARD INTO EACH CELL FROM
C              NEXT LAYER OF CELLS DOWNWARD.
C           FWU = FLUX OF WATER UPWARD. CM**3

      IF (L .LT. NL) THEN
         IF(DIFF(L+1,K)+DIFF(L,K).GT.0.00001) THEN
            FWU(L+1,K)=(DIFF(L+1,K)+DIFF(L,K))/2.
     .                 *((VH2OC(L+1,K)-VH2OC(L,K))/DCELL)
     .                 *(WCELL*TCELL*DELT)
         ELSE
            FWU(L+1,K)=0.0
         ENDIF

C        H2OMAX*0.25 IS THE MAXIMUM AMOUNT OF WATER THAT A CELL
C        CAN RECEIVE FROM EITHER DIRECTION (LEFT,RIGHT,DOWN,UP)
C        BY CAPILLARY ACTION PER ITERATION.  THTS(L) IS THETAS
C           BY LAYER

         WOUT = VH2OC(L+1,K) - VH2OC(L,K)
C        DIVIDE BY CELL AREA TO CONVERT TO DIMENSIONLESS FLOW
C             (CC/CC)
         FLOW = FWU(L+1,K)/VCELL

         IF(WOUT.LT.0.0) THEN
C             IF WATER MOVEMENT IS DOWNWARD OUT OF THE CELL
            RATIO = VH2OC(L+1,K)/VH2OC(L,K)
            H2OMAX = (THTS(L+1) - VH2OC(L+1,K))
     .               *(1.0 - RATIO)
            IF(ABS(FLOW).GT.(0.25 * H2OMAX)) THEN
               FWU(L+1,K) = -0.25*H2OMAX*VCELL
            ENDIF
         ELSE
C             IF WATER MOVEMENT IS UPWARD INTO THE CELL
            RATIO = VH2OC(L,K)/VH2OC(L+1,K)
            H2OMAX = (THTS(L) - VH2OC(L,K))
     .               *(1.0 - RATIO)
            IF(ABS(FLOW).GT.(0.25 * H2OMAX)) THEN
               FWU(L+1,K) = 0.25*H2OMAX*VCELL
            ENDIF
         ENDIF

         IF(ABS(FWU(L+1,K)).LE.FLXMIN(J)) THEN
            FWU(L+1,K) = 0.0
            FNU(L+1,K) = 0.0
            FNU(L+1,NKK) = 0.0
         ELSE
            SKPFLG = .TRUE.
            IF(ABS(FWU(L+1,K)).GT.FLXMAX(J)) THEN
               IF(FWU(L+1,K).LT.0.) FWU(L+1,K)= -FLXMAX(J)
```

```
         IF(FWU(L+1,K).GT.0.) FWU(L+1,K)=  FLXMAX(J)
      ENDIF

C       CALCULATE FLOW OF NITROGEN UPWARD, BOTH PLANES
C       FLOW OF NO3 UPWARD IN THE WATER, MG N/CELL/DAY.

      IF (FWU(L+1,K) .GT. 0.0) THEN
         FNU(L+1,K) = FWU(L+1,K)
     .                * (VNO3C(L+1,K)/VH2OC(L+1,K))
      ELSE
         FNU(L+1,K) = FWU(L+1,K)
     .                * (VNO3C(L,K)/VH2OC(L,K))
      ENDIF

      IF(ABS(FNU(L+1,K)).LT.FLNMIN) FNU(L+1,K) = 0.0
      IF (FWU(L+1,K) .GT. 0.0) THEN
         FNU(L+1,NKK) = FWU(L+1,K)
     .                  * (VNO3C(L+1,NKK)/VH2OC(L+1,K))
      ELSE
         FNU(L+1,NKK) = FWU(L+1,K)
     .                  * (VNO3C(L,NKK)/VH2OC(L,K))
      ENDIF

      IF(ABS(FNU(L+1,NKK)).LT.FLNMIN)FNU(L+1,NKK) = 0.0
   ENDIF
ELSE
   IF(WATTBL.GT.NL*DCELL) THEN
      H2OCNT=(IPCLAY(J)-5.)/70.*(FCININ(J)-THETA0(J))
     .         +THETA0(J)
      IF(H2OCNT.GT.FCININ(J)) H2OCNT=FCININ(J)
      IF(H2OCNT.LT.THETA0(J)) H2OCNT=THETA0(J)
   ELSE
      H2OCNT=THTS(NL)
   ENDIF
   IF(DIFF(L,K).GT.0.000005) THEN
      FWU(L+1,K)=DIFF(L,K)*(H2OCNT-VH2OC(L,K))/DCELL
     .              *(WCELL*TCELL*DELT)
   ELSE
      FWU(L+1,K)=0.0
   ENDIF

C  H2OMAX*0.25 IS THE MAXIMUM AMOUNT OF WATER THAT A CELL
C  CAN RECEIVE FROM EITHER DIRECTION (LEFT,RIGHT,DOWN,UP)
C  BY CAPILLARY ACTION PER ITERATION.  THTS(L) IS THETAS
C     BY LAYER

   WOUT = H2OCNT - VH2OC(L,K)
C  DIVIDE BY CELL AREA TO CONVERT TO DIMENSIONLESS FLOW
C       (CC/CC)
   FLOW = FWU(L+1,K)/VCELL

   IF(WOUT.LT.0.0) THEN
C       IF WATER MOVEMENT IS DOWNWARD OUT OF A CELL
      RATIO = H2OCNT / VH2OC(L,K)
      H2OMAX = (THTS(L) - H2OCNT)*(1.0 - RATIO)
      IF(ABS(FLOW).GT.(0.25 * H2OMAX)) THEN
         FWU(L+1,K) = -0.25*H2OMAX*VCELL
      ENDIF
   ELSE IF(NL*DCELL.GE.WATTBL) THEN
C     FOR WATER TO MOVE UPWARD THERE MUST BE A H2O TABLE
      RATIO = VH2OC(L,K) / H2OCNT
      H2OMAX = (THTS(L) - VH2OC(L,K))*(1.0 - RATIO)
```

```
                     IF(ABS(FLOW).GT.(0.25 * H2OMAX)) THEN
                        FWU(L+1,K) = 0.25*H2OMAX*VCELL
                     ENDIF
                  ELSE
                     FWU(L+1,K) = 0.
                  ENDIF

C  SUMSUB - Units are cm**3/(nk cells)

                  SUMSUB = SUMSUB+FWU(L+1,K)*2

                  IF(ABS(FWU(L+1,K)).LE.FLXMIN(J)) THEN
                     FWU(L+1,K) = 0.0
                     FNU(L+1,K) = 0.0
                     FNU(L+1,NKK) = 0.0
                  ELSE
                     SKPFLG = .TRUE.
                     IF(ABS(FWU(L+1,K)).GT.FLXMAX(J)) THEN
                        IF(FWU(L+1,K).LT.0.) FWU(L+1,K)= -FLXMAX(J)
                        IF(FWU(L+1,K).GT.0.) FWU(L+1,K)=  FLXMAX(J)
                     ENDIF

C                      CALCULATE FLOW OF NITROGEN UPWARD, BOTH PLANES
C                      FLOW OF NO3 UPWARD IN THE WATER, MG N/CELL/DAY.

                     FNU(L+1,K) = FWU(L+1,K) * (VNO3C(L,K)/H2OCNT)
                     IF((ABS(FNU(L+1,K)).LT.FLNMIN).OR.
     .                           (FNU(L+1,K).GT.0.)) FNU(L+1,K) = 0.0
                     FNU(L+1,NKK) = FWU(L+1,K)*(VNO3C(L,NKK)/H2OCNT)
                     IF((ABS(FNU(L+1,NKK)).LT.FLNMIN).OR.
     .                         (FNU(L+1,NKK).GT.0.)) FNU(L+1,NKK) = 0.0
                     CUMNSOK=CUMNSOK+FNU(L+1,NKK)+FNU(L+1,K)

                  ENDIF
               ENDIF
            ENDIF
 5    CONTINUE
 4406 CONTINUE

C CALCULATE TOTAL FLUX OF WATER NITROGEN FOR EACH CELL
C FWICN = FLUX OF H2O INTO THE CELL, NET, CM**2.
C FWICN/ACELLDW= CM**2/CM**2 (UNIT LESS)
C FNICN IS FLUX OF NO3 INTO THE CELL, NET, MG N/CELL/DAY.
C VNO3C IS VOLUMETRIC CONTENT OF SOIL CELL, MG N/CM**3.

      DO 16 L = 1,NL
         IF((.NOT.RTEXNT(L)).AND.(.NOT.FULPRO)) GO TO 4706
         J = LYRDPH(L)
         DO 16 K = 1,NKH
C         NKK = COLUMN INDEX OF MIRROR IMAGE CELL
            NKK = NK + 1 - K
            IF(TTUPF(L,K).OR.FULPRO) THEN
               FWICN = FWL(L,K+1) - FWL(L,K)
                   + FWU(L+1,K) - FWU(L,K)
              VH2OC(L,K) = VH2OC(L,K) + FWICN/VCELL
              IF(VH2OC(L,K).GT.THTS(L)) VH2OC(L,K) =THTS(L)
              IF(VH2OC(L,K).LT.AIRDR(J)) VH2OC(L,K)=AIRDR(J)
              VH2OC(L,NKK) = VH2OC(L,K)

C             CALCULATE N FLUX FOR EACH CELL.  EXTREME COLUMN IN EACH
C               PLANE MUST BE TREATED SEPARATELY.
```

```
                IF (K .EQ. 1) THEN
C                   CALCULATE FLUX FOR FAR LEFT CELL AND FAR RIGHT CELL
                   FNICN = FNL(L,2) + FNU(L+1,1) - FNU(L,1)
                   VNO3C(L,1) = AMAX1(VNO3C(L,1) + FNICN/VCELL,
     .                                FLNMIN/VCELL)
                   FNICN = FNL(L,NK-1) + FNU(L+1,NK) - FNU(L,NK)
                   VNO3C(L,NK) = AMAX1(VNO3C(L,NK) + FNICN/VCELL,
     .                                 FLNMIN/VCELL)
                ELSE IF (K .EQ. NKH) THEN
                   FNICN = - FNL(L,NKH)+FNU(L+1,NKH)-FNU(L,NKH)
                   VNO3C(L,NKH) = AMAX1(VNO3C(L,NKH) + FNICN/VCELL,
     .                                  FLNMIN/VCELL)
                   FNICN = -FNL(L,NKHP1)+ FNU(L+1,NKHP1)-FNU(L,NKHP1)
                   VNO3C(L,NKHP1) = AMAX1(VNO3C(L,NKHP1)+ FNICN/VCELL,
     .                                    FLNMIN/VCELL)
                ELSE
                   FNICN = FNL(L,K+1)-FNL(L,K)+ FNU(L+1,K)-FNU(L,K)
                   VNO3C(L,K) = AMAX1(VNO3C(L,K) + FNICN/VCELL,
     .                                FLNMIN/VCELL)
                   FNICN = FNL(L,NKK-1) - FNL(L,NKK)
     .                     + FNU(L+1,NKK) - FNU(L,NKK)
                   VNO3C(L,NKK) = AMAX1(VNO3C(L,NKK)+ FNICN/VCELL,
     .                                  FLNMIN/VCELL)
                ENDIF
             ENDIF
  16   CONTINUE
 4706 CONTINUE

C   UPDATE THE VALUE OF PSIS FOR EACH CELL.  UPDATE NOT NECESSARY ON
C   3RD OR 4TH ITERATION BECAUSE THERE WILL NOT BE AN INTERVENING
C   CALL TO UPTAKE ON THESE CALLS. IF SKPFLG IS STILL FALSE AT THIS
C   POINT, THERE WAS NO SIGNIFICANT WATER MOVEMENT SO THIS MAY BE
C   THE LAST ITERATION DONE AND SO VALUES MUST BE UPDATED.

      DO 8 L = 1,NL
         J = LYRDPH(L)
         DO 8 K = 1, NKH
            NKK = NK + 1 - K
            PSIS(L,K) = -15.
            IF(VH2OC(L,K).GT.THETAR(J)) THEN
               TEMP1C = (VH2OC(L,K)-AIRDR(J))/(FCININ(J)-AIRDR(J))
               PSIS(L,K) = PSISFC * TEMP1C**ARDRCN(L)
               IF(PSIS(L,K).GT.-0.00001)PSIS(L,K)=-0.00001
            ENDIF
            IF(PSIS(L,K).LT.-15.) PSIS(L,K)=-15.
            PSIS(L,NKK) = PSIS(L,K)
            DIFF(L,NKK) = DIFF(L,K)
  8   CONTINUE

      RETURN
      END
```

GLOSSARY

ACELLDW	Cross-sectional area of a soil cell, DCELL*WCELL (cm^2).
AIRDR(J)	Volumetric water content at air dry; read from the soil hydrology file (cm^3 H_2O cm^{-3} soil).
ARDRCN(L)	Temporary variable; inverse exponent in Marani equation.

$$ARDRCN = \frac{\ln(-15/(\text{FIELD CAPACITY SOIL H2O POTENTIAL}))}{\ln((THETA(R)-THETA(AD))/(THETA(FC)-THETA(AD)))}$$

where

R = residual
AD = air dry
FC = field capacity.

BDRATO	Boundary slope ratio for calculating movement of the water table; read from the soil hydrology file.
BDSLOP	Boundary theta slope for calculating movement of the water table; read from the soil hydrology file.
BETAI	Hydraulic conductance of the first layer (cm^3 soil cm^{-3} H_2O).
BETA(J)	Hydraulic conductance; read from the soil hydrology file and used in the diffusivity function (cm^3 soil cm^{-3} H_2O).
CUMNSOK	Cumulative nitrate-N leached from the profile (mg N).
DCELL	Depth of a soil cell (5 cm).
DELT	1/NOITR; fraction of daily water application moved in the current iteration. Fraction of the day included in the current iteration.
DIFF0I	DIFF0(1); DIFFO(J) of the top layer of soil (cm^2 day^{-1}).
DIFF0(J)	Diffusivity of the soil layer at −15 bar potential; read from the soil hydrology file (cm^2 day^{-1}).
DIFF(L,K)	Diffusivity; read from the soil hydrology file and a function of DIFFO, BETA, and THETA (cm^2 day^{-1}).
DUMY02	Amount of water that can be taken into the soil cell; defined as THETA(J) −THETAR(J) (cm^3 H_2O cm^{-3} soil).
FBLOOM	Day of first bloom (number of days after emergence).
FCININ(J)	Volumetric water content at field capacity; read from the soil hydrology file (cm^3 H_2O cm^{-3} soil).
FLNMIN	Minimum value of nitrogen flux, 1.0E-6 (mg N).
FLOW	Water movement per unit volume (cm^3 H_2O cm^{-3} soil).
FLXMAX(J)	Maximum water flux per iteration (cm^3 H_2O).
FLXMIN(J)	Minimum water flux per iteration (cm^3 H_2O).
FNICN	Net flux of nitrogen into the cell (mg N).
FNL(L,K)	Flux of nitrogen from the cell to the left (mg N).

FNU(L,K)	Flux of nitrogen from the cell upward (mg N).
FULPRO	Flag indicating whether all parts of CAPFLO will operate during the iteration. Set to TRUE immediately after GRAFLO and during the first iteration (logical variable).
FWICN	Net flux of water into the cell (cm^3 H_2O).
FWL(L,K)	Flux of water from the cell to the left (cm^3 H_2O).
FWU(L,K)	Flux of water from the cell upward (cm^3 H_2O).
H2OCNT	Volumetric water content of soil below the water table or below the lower boundary of the profile (the NL+1 cell layer) (cm^3 H_2O cm^{-3} soil).
H2OMAX	Maximum amount of water that a cell can receive (cm^3 H_2O cm^{-3} soil).
IDAY	Day of simulation (days).
IPCLAY	Percentage clay content of a layer; used in determining the lower boundary condition (%).
IT0	IDAY + 1 until after first bloom, then FBLOOM + TD; used in adjusting the water table.
ITER	Iteration number from 1 to NOITR; represents the number of times CAPFLO is called.
LYRDPH(L)	Integer counter from 1 to the number of soil layers; each element is the soil horizon number for that layer.
NK	Number of vertical columns of soil cells in the profile.
NL	Number of layers (horizontal rows) of soil cells in the profile.
NOITR	Number of iterations; number of times CAPFLO will be called on this day.
PSISFC	Soil water potential at field capacity (bars).
PSIS(L,K)	Soil water potential of the cell at layer (L) and column (K) (bars).
RATIO	Ratio of volumetric water contents of adjacent cells; RATIO = last cell/ current cell (no units).
RTEXNT(L)	A flag to identify whether new roots (less than 15 days old) are contained in the layer; used to determine if CAPFLO operations dealing with uptake factor will be used (logical variable)
SAVEH2O	A temporary variable used to store volumetric water content when VH2OC is updated (cm^3 H_2O cm^{-3} soil).
SKPFLG	A flag to indicate whether a section of CAPFLO will be used; indicates TRUE if there is water movement (logical variable).
SUMSUB	Cumulative excess water in the water table (cm^3 H_2O).
TCELL	Thickness of a soil cell (1 cm).
TD	"Time-day" factor for calculating movement of the water table; read from the soil hydrology file.

THETA	Volumetric water content (cm^3 H_2O cm^{-3} soil).
THETA0	Volumetric water content at psi = −15 bar (cm^3 H_2O cm^{-3} soil).
THETAI	Volumetric water content at field capacity (cm^3 H_2O cm^{-3} soil).
THETAR	Residual volumetric water content (cm^3 H_2O cm^{-3} soil).
THTA0I	THETA0 for the top soil layer (cm^3 H_2O cm^{-3} soil).
THTS(L)	Saturated volumetric water content for the layer (cm^3 H_2O cm^{-3} soil).
TTUPF(L,K)	A flag to indicate if there are roots in the cell that can extract water (logical variable).
VCELL	Volume of a soil cell, DCELL*WCELL*TCELL (cm^3).
VH2OC(L,K)	Volumetric water content of the cell (cm^3 H_2O cm^{-3} soil).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N cm^{-3} soil).
WATTBL	Depth to the water table from the soil surface (cm).
WCELL	Width of a soil cell; 5 cm or row spacing/NK (cm).
WOUT	Water flowing out of a cell (cm^3 H_2O cm^{-3} soil).

SUBROUTINE ET

Review chaired by G. Wall and D.O. Porter

The ET subroutine estimates evaporation from the soil surface and transpiration through plant surfaces. The model is patterned after the Ritchie model (1972) for predicting evaporation from a row crop with incomplete cover. Both the ET model and the Ritchie model apply a modified Penman equation, which takes into account the effects of solar radiation, wind, and vapor pressure deficit. After estimating the potential evapotranspiration using the Penman equation (American Society of Civil Engineers 1990), the ET value is reduced for evaporation from the plant and for the reduced capacity of the soil to meet evaporative demand as it dries.

The Penman equation follows the form

- $$E_T = \frac{DELTA}{DELTA + GAMMA} R_n + \frac{GAMMA}{DELTA + GAMMA} (0.35)(1.0 + 0.01W_2)(e_a - e_d)$$

  where

  E_T = potential transpiration (mm day^{-1})*
  R_n = net radiation (mm day^{-1})*
  DELTA = slope of the saturation vapor-pressure curve at mean air temperature
  GAMMA = constant of the wet- and dry-bulb psychrometric equation
  W_2 = mean wind speed at a height of 2 meters (miles day^{-1})
  e_a = saturation vapor pressure at mean air temperature (mm Hg)
  e_d = saturation vapor pressure at dew point (mm Hg).

Net radiation may be determined by

- $R_n = (1 - alpha) R_s - R_L$

  where

  $(1 - alpha) R_s$ = the net shortwave radiation received by a green crop with a full cover

  alpha = the mean daily shortwave reflectance (albedo); has a value between 0.20 and 0.25 for most green crops

  R_L = the net outgoing longwave radiation.

R_L is determined by the equation

- $R_L = (1.35 R_s/R_{so} - 0.35) R_{LO}$

  where

  R_s = observed solar radiation for a given day (mm day^{-1})
  R_{so} = solar radiation on the same day with no cloud cover (mm day^{-1})
  R_{LO} = net outgoing longwave radiation on a clear day (mm day^{-1}).

R_{LO} can be estimated by the equation

- R_{LO} = sigma$T^4 (0.31 - 0.051 (e_d)^{.5})$

  where

*1 mm day^{-1} is equivalent to an energy flux of 59 cal cm^{-2} day^{-1}.

sigmaT^4 = black-body radiation at air temperature (Pair et al. 1983)
sigma = Stefan-Boltzman constant (4.903 x 10-9 MJ $m^{-2}K^{-4}d^{-1}$)
T = air temperature in Kelvin (273.2 + °C).

ASSUMPTIONS

1. The wet-bulb temperature is very near the dewpoint (that is, the relative humidity in the field is approximately 100 percent at the minimum daily temperature). This assumption is necessary in order to use the vapor pressures at the average daily temperature and the wet-bulb temperature in order to determine the water vapor deficit.

2. The reflectivities and drying curves of all soils are similar to those of the Houston clay described in the Ritchie model (1972).

3. The soil drying curve (that is, the cumulative evaporation versus time) can be approximated in two stages. Stage I represents wet soil conditions in which evaporation is limited by evaporative demand. Stage II represents drying soil conditions in which evaporation is limited by the soil's capacity to supply water for surface evaporation.

4. Heat of vaporization is approximately 585 calories per gram (the value at standard pressure and 22 °C). This value varies with temperature and pressure.

INPUTS

ALPHA
GAMMA
INT
LAI
LAMDAC
LAMDAS
PSIAVG
RAIN
RI
TAVG
TMIN
VP(TAVG)
WIND

OUTPUTS

E
EP
ES
RN

GENERAL PSEUDOCODE

Include the common variables from GOSCOM.FOR

Initialize the upper limit of stage I evaporation, effective rain, and solar radiation

Estimate the slope of the psychrometric curve at the average daily temperature

Estimate the net radiation above the plant canopy

Set the wet- and dry-bulb temperatures and their corresponding saturated water vapor pressures

Estimate the maximum potential evapotranspiration (evaporative demand) by applying the Penman equation

Estimate the net solar radiation on the soil surface and the potential evaporation from the soil surface

Locate the current position on the soil drying curve (**Note:** The position will depend on the previous position on the curve, effective rainfall depth, and evaporation.

Estimate the transpiration as a proportion of light interception

Calculate the reduction factor and apply it to adjust transpiration

Return to SOIL

End ET

PSEUDOCODE

Include the common block of variables from GOSCOM.FOR

Set the upper limit of stage I drying to 6 mm (**Note:** This is the point at which the evaporation from the soil surface becomes dependent upon the soil's ability to transmit water upward to the surface. The value 6 mm was determined by the drying curve of a Houston clay.)

Initialize the effective depth of water added to the soil on this day (P) (**Note**: This value includes rainfall and irrigation.)

Convert daily incident solar radiation (RI) to an equivalent depth of water that would be evaporated by the energy of solar radiation

Set the average daily temperature minus 1 (TAVM1)

Calculate the difference between the saturated water pressures in air at the average daily temperature and at TAVM1 (**Note**: This value is used to represent the slope of the psychrometric curve at the average daily temperature.)

Calculate the reflectance of the field (LAMDA), taking into account the reflectance of the soil and plant surfaces that intercept solar radiation

Calculate net solar radiation above the canopy (**Note**: This is the difference between incident solar radiation and reflected solar radiation.)

Set the dry-bulb temperature (TDRY) to the average daily temperature (average of maximum and minimum temperatures)

Set the saturated water vapor pressure at the dry-bulb temperature (VPO)

Set the wet-bulb temperature to the minimum daily temperature

Set the saturated water vapor pressure at the wet-bulb temperature (VPA)

Calculate the potential evaporation rate above the canopy in mm H_2O per day (**Note**: The Penman equation is used as the model for this estimate. Wet-bulb temperature is used in place of the dewpoint required by Penman. The wet-bulb temperature is acceptable if it can be assumed that humidity is near saturation at the minimum daily temperature.)

Calculate net solar radiation at the soil surface below the crop canopy

Calculate the potential soil water evaporation below the canopy as a function of the net solar radiation below the canopy and the psychrometric curve

If cumulative stage I drying from the soil surface exceeds its upper limit, then

*** Determine stage II drying. ***

if effective rainfall or irrigation is greater than or equal to cumulative stage II evaporation, then

reduce amount of water added to satisfy stage II cumulative evaporation from the soil surface

reduce stage I evaporation by the remaining water applied

cumulative evaporation will be the difference between the upper limit of stage I drying and the net water application

else

increment the day (T) when the drying curve will reach the upper limit of stage I evaporation

determine the evaporation rate from the soil surface as a function of the constant slope (ALPHA) and time (T)

if there is precipitation or irrigation on this day, then

estimate soil evaporation from the soil surface during stage II drying (ESX) as a portion of the precipitation

if ESX is less than the soil evaporation estimated from the curve, let ESX equal the sum of estimated soil evaporation and precipitation

if ESX is greater than the calculated potential soil surface evaporation (ESO), then set ESX to ESO

set soil evaporation (ES) to the estimated value of ESX

else if soil evaporation (ES) is greater than the potential evaporation, set evaporation to the potential value

endif

update the value of cumulative stage II evaporation, including the current day's evaporation and water application

determine a new position on the drying curve to determine the effective time (T)

go to *[B]* (page 55)

endif

Else if effective rainfall or irrigation or both are less than the cumulative stage I evaporation, then

update stage I evaporation by reducing it by the depth of the water application

go to *[A]* (page 55)

Endif

Set cumulative stage I evaporation to 0.0

[A] Continue

Update cumulative stage I evaporation to include potential evaporation for the current day

If cumulative stage I evaporation exceeds its upper limit, then

reduce excess soil evaporation to fit the stage II evaporation curve

add the adjusted excess evaporation to the stage II curve

update the effective time value found on the drying curve

Else

soil evaporation is equal to the potential soil evaporation

Endif

*** Calculate transpiration. ***

[B] Continue

Set transpiration to equal the product of the potential evaporation rate above the crop canopy and the fraction of light intercepted by the canopy

If transpiration is greater than the difference between potential evaporation and soil evaporation, set transpiration to equal the difference

Set AVGPSI to equal -1 times PSIAVG (**Note:** AVGPSI has a positive value and represents soil suction, in bars.)

If AVGPSI is greater than 0.8 bars, set it to equal 0.8 bars

Calculate net solar radiation from incident solar radiation, and convert units from langleys to Wm^{-2}

Calculate a reduction factor for evaporation from the plant [**Note**: These regression equations are derived from Baker and Hesketh (1969). This value may fall between 0.2 and 1.0, and is a function of net solar radiation, average daily temperature, and average soil water potential.]

Adjust plant evaporation (transpiration) by multiplying it by the reduction factor

Return to SOIL

End ET

```
      SUBROUTINE ET
C   ***************************************************************
C   *                                                             *
C   *            EVAPOTRANSPIRATION SUBROUTINE                    *
C   *                                                             *
C   ***************************************************************
C   SUBROUTINE TAKEN ALMOST ENTIRELY FROM RITCHIE (1972), MODEL
C   FOR PREDICTING EVAPORATION FROM A ROW CROP WITH INCOMPLETE COVER.
C   WATER RESOURCES RESEARCH 8:1204-1213.
C
C
      INCLUDE 'GOSCOM.FOR'
C
      U=6.
      P = RAIN
      RS = RI*.0169491525
C RS = SOLAR RADIATION IN MM H2O/DAY.
      TAVM1 = TAVG-1.
      DEL = VP(TAVG) - VP(TAVM1)
C DEL=SLOPE OF SATURATION VAPOR PRESSURE CURVE AT MEAN AIR TEMP.
      LAMDA = INT*LAMDAC + (1.-INT)*LAMDAS
C LAMDAC & LAMDAS = ALBEDOS OF CROP & SOIL.
C INT=INTERCEPTION (FRACTION OF INCIDENT RS)
C RNO=NET RADIATION ABOVE CANOPY (MM/DAY)
      RNO=(RS-LAMDA*RS)
C TDRY & TWET = DRY AND WET BULB TEMPERATURES.
      TDRY = TAVG
      VPO = VP(TDRY)
      TWET = TMIN
C EO=POTENTIAL EVAPORATION RATE ABOVE CANOPY (MM/DAY)
C MODIFIED PENMAN EQ.
C WIND = WINDSPEED AT 2 METERS (MILES/DAY)
C GAMMA=PSYCHROMETER CONSTANT
      VPA = VP(TWET)
      EO=(RNO*DEL/GAMMA+.262*(1.+0.0061*WIND)*(VPO-VPA))/(DEL/GAMMA+1.)
C THE FOLLOWING CALCULATES ESO(POTENTIAL EVAP. RATE AT SOIL SURFACE1
C RNS=NET RADIATION AT SOIL SURFACE BELOW CANOPY
      RNS=((1.-INT)-(1.-INT)*LAMDAS)*RS
      ESO=DEL*RNS/(DEL+GAMMA)
C STAGE I DRYING
C SESI=CUMULATIVE STAGE ONE EVAPORATION FROM SOIL SURFACE
C U=UPPER LIMIT OF SESI
      IF(SESI.GT.U)GOTO 100
C P=RAINFALL
      IF(P.GE.SESI)GOTO 101
      SESI=SESI-P
 99   SESI=SESI+ESO
      IF(SESI.GE.U)GOTO 102
      ES=ESO
      GOTO 110
 102  ES=ESO-.4*(SESI-U)
      SESII=.6*(SESI-U)
      DUMY01 = SESII / ALPHA
      T = DUMY01 * DUMY01
      GO TO 110
 101  SESI=0.
      GO TO 99
C STAGE II DRYING
 100  IF(P.GE.SESII)GO TO 103
      T=T+1.
      ES = ALPHA * (SQRT(T)-SQRT(T-1.))
      IF(P.GT.0.)GO TO 104
```

```
      IF(ES.GT.ESO)GO TO 105
 106  SESII=SESII+ES-P
      DUMY02 = SESII / ALPHA
      T = DUMY02 * DUMY02
      GO TO 110
 105  ES=ESO
      GO TO 106
 104  ESX=0.8*P
      IF(ESX.LT.ES)GO TO 107
 111  IF(ESX.GT.ESO)GO TO 108
 109  ES=ESX
      GO TO 106
 108  ESX=ESO
      GO TO 109
 107  ESX=ES+P
      GO TO 111
 103  P=P-SESII
      SESI=U-P
      IF(P.GT.U)GO TO 101
      GO TO 99
C TRANSPIRATION IS PROPORTIONAL TO LIGHT INTERCEPTION (INT).
C  THIS REPRESENTS A MODIFICATION TO RITCHIE'S MODEL.
 110  EP = INT * EO
      IF(EP.GT.(EO-ES)) EP=EO-ES
C      E = ES + EP
      AVGPSI = -1. * PSIAVG
      IF(AVGPSI.GT.0.8) AVGPSI = 0.8
      RN = RI*.71536-26.

C  RFEP = REDUCTION FACTOR FOR EVAPORATION FROM PLANT.  BASED ON
C  UNPUBLISHED DATA OF BAKER & HESKETH. 1969.

      RFEPN = 749.5831405 + 0.9659065*RN - 54.6600986*TAVG
     .        - 194.6508431*AVGPSI - 0.0010226*RN*RN +1.0153007*TAVG*
           TAVG + 29.775978*AVGPSI*AVGPSI + 0.0293687*RN*TAVG
     .        - 4.206856*TAVG*AVGPSI
      RFEPD = 749.5831405 + 0.9659065*RN - 54.6600986*TAVG -
     .        19.46508431 - 0.0010226*RN*RN + 1.0153007*TAVG*TAVG +
         0.29775978 + 0.0293687*RN*TAVG - .4206856*TAVG
      RFEP = RFEPN/RFEPD
      IF(RFEP.LT.(0.2*LAI/2.0)) RFEP = 0.2 * LAI/2.0
      IF(RFEP.LT.0.2) RFEP = 0.2
      IF(RFEP.GT.1.0) RFEP = 1.0
      EP = EP * RFEP
      RETURN
      END
```

GLOSSARY

ALPHA	The slope of the curve plotting cumulative soil evaporation against the square root of time. Alpha is given a value of 3.5, which represents the slope of the curve for a Houston clay (mm H_2O days$^{-1/2}$).
AVGPSI	Average soil water potential (bars, positive value); equal to PSIAVG but opposite in sign.
DEL	Difference between saturated water vapor pressures at TAVG and TAVM1; used to estimate the slope of the psychrometric curve at TAVG (millibars).
DUMY01	A temporary variable used to locate effective time on the cumulative evaporation curve. Effective time is determined by the degree of stage II soil drying (days$^{1/2}$).
DUMY02	A temporary variable used to locate effective time on the cumulative evaporation curve. Effective time is determined by the degree of stage II soil drying (days$^{1/2}$).
E	Sum of evaporation and transpiration, EP + ES (mm H_2O).
EO	Potential evaporation rate above the crop canopy (mm H_2O day^{-1}).
EP	Plant evaporation or transpiration rate (mm H_2O day^{-1}).
ES	Evaporation rate from the soil surface (mm H_2O day^{-1}).
ESO	Potential evaporation rate below the plant canopy at the soil surface (mm H_2O day^{-1}).
ESX	Evaporation rate from the soil surface during stage II evaporation (mm H_2O day^{-1}).
GAMMA	Psychrometer constant, rho*c_p*L where rho = density of air (gm cm^{-3}) c_p = specific heat of air (cal/gm-°C) L = latent heat of vaporization of water (cal gm^{-1}). GAMMA is dependent upon temperature and atmospheric pressure, but is treated as a constant (GAMMA = 0.653) in this routine. This value is used in the wet- and dry-bulb psychrometer equation.
INT	Fraction of incident solar radiation intercepted by the canopy (no units).
LAI	Leaf area index; projected leaf area per unit of ground surface area.
LAMDA	Total albedo of the field; takes into account the crop albedo and the soil albedo (dimensionless fraction).
LAMDAC	Crop albedo; the fraction of incident radiation reflected by most green crops with a full cover; 0.23 is commonly used.
LAMDAS	Soil albedo; the fraction of incident radiation reflected by the soil. The average albedo of bare soil is assumed to be 0.1, that of a Houston clay.
P	Amount of rainfall or irrigation or both (mm H_2O day^{-1}).
PSIAVG	Average soil water potential (bars, negative value).

RAIN	Effective rainfall or irrigation or both for this day (mm H_2O).
RFEP	Reduction factor for evaporation from the plant (Lambert and Baker 1984). This factor, calculated by a regression equation, is a function of net solar radiation, average daily temperature, and average soil water potential.
RFEPD	Intermediate value used in calculating RFEP.
RFEPN	Intermediate value used in calculating RFEP.
RI	Incident solar radiation; read from the input weather file (langleys).
RN	Net solar radiation; calculated from RI (Wm^{-2}).
RNO	Net solar radiation above the canopy; units are converted to the amount of water that may be evaporated per day (mm H_2O day^{-1}).
RNS	Net solar radiation at the soil surface below the canopy (mm H_2O day^{-1}).
RS	Solar radiation; units are converted to the amount of water evaporated by the energy of solar radiation (mm H_2O day^{-1}).
SESI	Cumulative stage I drying from the soil surface; cumulative water removed in the linear phase of the drying curve (mm H_2O).
SESII	Cumulative stage II drying from the soil surface; cumulative water removed in the nonlinear phase of the drying curve (mm H_2O).
T	Calculated time when drying will begin in the nonlinear phase; time to switch from stage I to stage II drying (days).
TAVG	Mean daily temperature; calculated from wet-bulb and dry-bulb temperatures (°C).
TAVM1	TAVG minus 1; value is used to estimate the slope of the psychrometric curve at TAVG (°C).
TDRY	Average dry-bulb temperature; equal to TAVG (°C).
TMIN	Wet-bulb temperature (minimum daily temperature) (°C).
TWET	Wet-bulb temperature (°C).
U	Upper limit of cumulative evaporation in the linear stage I phase of drying (mm H_2O).
VPA	Saturated vapor pressure of water in air at the wet-bulb temperature (mbars).
VPO	Saturated vapor pressure of water in air at the (maximum) daily dry-bulb temperature (mbars).
VP(TAVG)	Saturated vapor pressure of water in air at the daily mean air temperature (mbars).
WIND	Wind speed at a height of 2 m (mi day^{-1}).

SUBROUTINE EVPSOIL

Written by S.B. Turner

EVPSOIL adjusts the evaporative losses from the soil surface and updates the volumetric water contents of the surface cells. The soil may lose moisture until it reaches its air-dry volumetric water content. Only surface soil cells not shaded by the crop canopy are used to satisfy evaporative demand. If available water is insufficient to meet the demand, the amount of evaporation is reduced accordingly.

ASSUMPTIONS

1. Evaporative losses occur only from soil cells that are not shaded by the crop canopy.
2. Water may be removed from a cell until the cell is at its air-dry water content.

INPUTS

AIRDR(1)
ES
INT
NEWES
NK
NOITR
TCELL
VCELL
VH2OC(1,K)
WCELL

OUTPUTS

NEWES
VH2OC(1,K)

GENERAL PSEUDOCODE

Include the common variables from GOSCOM.FOR

Initialize the cell counter and soil evaporation accumulator

Divide soil evaporative demand into "per iteration" portions

Determine the number of columns from which water may evaporate

Calculate the evaporative demand per cell for this iteration

Locate the cells from which water may evaporate

For each cell in the surface layer, check whether the available water is sufficient to meet evaporative demand

> If the available water is insufficient, add it to the evaporation accumulator and remove it from the cell
>
> Else, add the evaporative demand from this cell to the evaporation accumulator and remove it from the cell

Increment the cell counter

Calculate the soil evaporation adjustment

Adjust soil evaporation for this iteration

Return to SOIL

End EVPSOIL

PSEUDOCODE

Include the common block of variables from GOSCOM.FOR

*** Units of evaporation are in mm/day and units of soil moisture and water removed from the soil are in cm^3 H_2O cm^{-3} soil. ***

Initialize to 0.0 the cell counter and cumulative soil evaporation for this iteration

Calculate the portion of daily soil evaporation to be considered in this iteration (ESPART) (**Note**: Soil evaporation is divided into NOITR portions to allow the CAPFLO subroutine time to move water to the surface by capillary flow.)

Calculate the number of surface cells from which water may evaporate (NKES) (**Note**: This number is determined by the number of columns of cells that are not shaded by the canopy, but it must be at least 2.)

Calculate the volume of water that must be removed from each cell to satisfy evaporative demand per iteration (H2OREM) [**Note**: This value is determined by converting ESPART (mm day^{-1}) to a volumetric ratio (cm^3 H_2O cm^{-3} soil). The value is then weighted to reflect that the demand over the entire surface is to be met by water in unshaded cells.]

Locate the unshaded cell columns in the profile (**Note**: The unshaded region will extend from column K1 on the left to K2 on the right. Column K1 is located immediately to the right of the canopy on the left edge of the profile. Column K2 is located immediately to the left of the canopy on the right edge of the profile.)

If K1 is to the left of K2, then

    for each unshaded cell, do:

    *** Water content in the cell in excess of the water held at air-dry condition is referred to as "available water"; available water = VH2OC(1,K) -AIRDR(1). ***

        if the evaporative demand per cell is greater than available water in this cell, then

            add all water available for evaporation to the SUMES accumulator

            set volumetric water content of cell (1,K) to its air-dry value

        else

            the evaporative demand from this cell can be met by the available water, therefore add the demand (H2OREM) to the evaporation accumulator (SUMES)

            volumetric water content of the cell is updated to reflect the water removed by evaporation during this iteration

        endif

        cell counter (NCELLS) is incremented (**Note**: This value records the number of cells from which water is evaporated.)

    end do

Endif

Calculate the soil evaporation adjustment value (ADJES) (**Note:** This value includes the cumulative soil evaporation [SUMES], which is converted from cm^3 H_2O cm^{-3} soil to mm H_2O day^{-1}. The weighting factor [included in the calculation of H2OREM] is removed to maintain water balance.)

The value of soil evaporation (NEWES) is updated by adding the adjustment value to it

Return to SOIL

End EVPSOIL

SOURCE CODE

```
      SUBROUTINE EVPSOIL

      INCLUDE 'GOSCOM.FOR'

C  UNITS     ES : MM/DAY     H2OREM : CM**3/CM**3
      NCELLS = 0
      SUMES = 0.0
      ESPART = ES/NOITR
      NKES = MAX0(2, IFIX((1.-INT) * NK))
      H2OREM = ((0.1*ESPART)*WCELL*TCELL/VCELL)*NK/NKES
      K1 = MAX0(1, ((NK-NKES)/2)+1)
      K2 = NK + 1 - K1
      IF (K1 .LE. K2) THEN
         DO K = K1,K2
            IF(H2OREM.GT.VH2OC(1,K)-AIRDR(1)) THEN
               SUMES = SUMES + (VH2OC(1,K)-AIRDR(1))
               VH2OC(1,K) = AIRDR(1)
            ELSE
               SUMES = SUMES + H2OREM
               VH2OC(1,K) = VH2OC(1,K) - H2OREM
            ENDIF
            NCELLS = NCELLS + 1
         ENDDO
      ENDIF
      ADJES = (10.*SUMES*VCELL / (NCELLS*WCELL*TCELL)) * NKES/NK
      NEWES = NEWES + ADJES
      RETURN
      END
```

GLOSSARY

ADJES	Adjustment to correct the soil evaporation value (mm H_2O day^{-1}).
AIRDR(1)	Volumetric water content at air dry for the surface soil horizon; read from the soil hydrology file (cm^3 H_2O cm^{-3} soil).
ES	Evaporation rate from the soil surface (mm H_2O day^{-1}); calculated in ET on the assumption of bare soil.
ESPART	Evaporation from the soil during the portion of the day for which this iteration runs (mm H_2O per fraction of day).
H2OREM	Volume of water removed by evaporation from the soil surface per iteration. This value represents evaporative demand per cell, not actual evaporation (cm^3 H_2O cm^{-3} soil).
INT	Fraction of light intercepted by plant leaves; also the fraction of the soil covered by the crop canopy. INT is used in determining soil cells from which water evaporates (no units).
K1	Column on the left edge of the "evaporation zone"; the first column not shaded by canopy cover from the left side of the profile (no units).
K2	Column on the right edge of the "evaporation zone"; the last column not shaded by canopy cover from the left side of the profile (no units).
NCELLS	Number of surface soil cells from which water may be evaporated (cells).
NEWES	Total daily evaporation from the soil (mm H_2O); the amount actually removed.
NK	Number of vertical columns of soil cells in the profile.
NKES	Number of surface cells from which evaporation will occur; minimum value of 2 (columns).
NOITR	Number of iterations or times the routine will be called per day.
SUMES	Cumulative water removed from the soil by evaporation during this iteration (cm^3 H_2O cm^{-3} soil).
TCELL	Thickness of a soil cell (1 cm).
VCELL	Volume of a soil cell, DCELL*WCELL*TCELL (cm^3).
VH2OC(1,K)	Volumetric water content of the cell in the first cell layer and column K (cm^3 H_2O cm^{-3} soil).
WCELL	Width of a soil cell, 5 cm or row spacing/NK (cm).

SUBROUTINE UPTAKE

Revised by S.M. Bridges

This subroutine is called five times a day to model the withdrawal by roots of water and nitrate from the two-dimensional soil profile. The rate of withdrawal from each soil cell depends on the root density within the cell, the age distribution of these roots, and the water status of the cell as characterized by soil water diffusivity. If sufficient water is available, the total withdrawal of water by the roots is equal to the daily transpiration rate.

Subroutine UPTAKE performs the following functions:

- calculates the proportion of roots capable of uptake in each cell
- apportions transpiration among cells based on the presence of active roots and diffusivity
- reduces the volumetric water content of each cell containing roots by the amount of uptake from that cell
- calculates the active and passive nitrate-N and ammonium-N uptake from rooted cells
- reduces the nitrate-N and ammonium-N content of each cell with roots by the amount of uptake from that cell.

ASSUMPTIONS

1. The total amount of water removed from the soil by uptake is apportioned to individual cells containing roots, based on a weighting factor. This weighting factor is the product of the weight of those roots capable of uptake and the diffusivity of the cell.
2. The amount of water that can be removed from the soil is the estimated transpiration amount calculated in subroutine ET.
3. The youngest roots have 100 percent uptake efficiency, while roots of intermediate and old age have 20 percent uptake efficiency (Graham et al. 1973).
4. Water uptake in each cell has a limit of the maximum of
   - 0 or
   - the minimum of the estimated uptake of the cell and the current available water of that cell.
5. Ten percent of ammonium-N is assumed to be free for passive uptake.
6. Active uptake of nitrate-N is based on curve-fitted data for wheat, using the Michelis-Menten equation (McMaster, personal communication, 1984).
7. All water and remaining nitrate-N in solution are taken up by the roots through mass flow.

INPUTS

ACELLDW
AIRDR(1)
DCELL
DIFF(L,K)
EP
ES
FCININ(J)
INT

KRL(L)
KWIDTH
LR
NK
NL
RTWT(L,K,1)
RTWT(L,K,2)
RTWT(L,K,3)
THETAR(J)
VCELL
VH20C(L,K)
VNH4C(L,K)
VNO3C(L,K)
WCELL

OUTPUTS

SUMEP
SUMES
UPNH4C
UPNO3C
VH20C(L,K)
VNH4C(L,K)
VNO3C(L,K)

GENERAL PSEUDOCODE

Set the fraction of total daily uptake to 0.20 for each of the 5 iterations

Calculate the volume of water to be removed per iteration as a function of plant evaporation

For all cells with roots calculate the total root weight capable of uptake by adding the weight of the young roots (less than 5 days old) and 20 percent of both the intermediate-aged (5–15 days old) and old (more than 15 days old) roots

For each of the rooted cells in the left half of the profile, do:

Calculate the uptake factor as the product of root weight capable of uptake and the soil diffusivity of each cell

Calculate the percentage of total uptake that will be removed from each cell

Set the flags to determine which cells are to be processed in the next iteration

For both the cell and its mirror-image cell, do:

> Calculate the water uptake from the cell as the product of total plant evaporation and the percentage uptake factor
>
> Calculate the passive uptake of nitrate-N from the cell, assuming all nitrate-N is in solution
>
> Calculate the passive uptake of ammonium-N from the cell, assuming 10 percent of the ammonium-N in solution is available
>
> Calculate the active uptake of nitrate-N using the Michelis-Menten equation for the three age categories of roots
>
> Update the volumetric water content of the cell by subtracting the amount of water uptake from the cell

Update the volumetric nitrate-N content of the cell by subtracting the amount of active and passive nitrate-N uptake from the cell (**Note:** The minimum nitrogen flux [FLNMIN] is the lower limit of the volumetric nitrate-N content of the cell.)

Update the volumetric ammonium-N content of the cell by subtracting the amount of passive ammonium-N uptake from the cell (**Note:** The minimum nitrogen flux [FLNMIN] is the lower limit of the volumetric ammonium-N content of the cell.)

End do

End do

If this is the last iteration, then set the new transpiration rate equal to the total plant evaporation rate

Return to SOIL

End UPTAKE

PSEUDOCODE

Set the fraction of total daily uptake done per iteration (DELT)

Calculate the volume of water to be removed on this iteration (DUMY01) from transpiration (EP) (cm^3 H_2O)

Initialize to 0.0 the root weight capable of uptake [RTWTCU(I,J)] and the uptake factor [UPF(L,K)] of all the cells in the NL layer (I) by the NK column (J)

For all of the NL+1 by NK array, do:

initialize the percentage uptake factor [PUPF(L,K)] to 0.0

set the two logical variables TUPF(L,K) and TTUPF(L,K) to TRUE—there is significant water uptake from the cell being considered

End do

For each of the rooted cells, do:

calculate the root weight capable of uptake [RTWTCU(L,K)] by adding the weight of all the roots 5 days old or less and 20 percent of the weight of roots older than 5 days

if the current rooted cell layer plus 2 is less than or equal to the maximum number of layers—NL, then if there are roots capable of uptake in the first rooted cell of this layer L and the logical variable identifying the extent of roots in this layer plus 2 [RTEXNT(L+2)] has a value of FALSE, set RTEXNT(L+2) to TRUE—there are roots 2 layers below the current layer

End do

For each of the rooted cells, add the roots growing from the right row toward the left (into the slab) to obtain the total number of roots taking up water from each soil cell

Set the intermediate variable (NKH) to half the number of columns (NK)

*** Due to symmetry, water uptake needs to be represented in only half of the plane. The other half is handled as an image. ***

Initialize to 0.0 the sum of the uptake factors (SUPF) of all the rooted cells

For each of the rooted cells in the left half of the profile, do:

calculate the uptake factor [UPF(L,K)] as a function of the cell's root weight capable of uptake and soil diffusivity [DIFF(L,K)] (g cm day^{-1})

accumulate the uptake factors (SUPF)

End do

Set the intermediate variable (D01) to the inverse of SUPF

For all rooted cells in the left half of the soil profile, calculate the percentage of total uptake that will be removed from each cell [PUPF(L,K)] as UPF(L,K)/SUPF

For all rooted cells of the left half of the soil profile,

*** Set the flags to determine which cells are to be processed in the next iteration. ***

if the percentage uptake factor of each cell in the current layer and in the layer below is less than or equal to 0.005 percent, then set TUPF(L,K) to FALSE—there is no significant water uptake from the current cell (L,K) or the cell below it (L+1,K)

if the percentage uptake factor of each cell in the current layer and in the layer below is less than or equal to 0.005 percent and TUPF(L,K) is FALSE, then set TTUPF(L,K) to TRUE—there is no significant water uptake from cell (L,K), cell (L+1,K), or cell (L,K+1)

set the index for the mirror-image cell, I, to NK+1-K

set the logical variables TUPF(L,I) to TUPF(L,K) and TTUPF(L,I) to TTUPF(L,K)

End do

Initialize to 0.0 the total uptake of nitrate-N (UPNO3) and ammonium-N (UPNH4)

For each rooted cell in the left half of the soil profile, do:

calculate the uptake of water from each cell (UPTH2O) as the product of total transpiration from plant leaves (EP) for this iteration and the percentage uptake factor for this cell

convert UPTH2O to cm^3 cm^{-3} (H2OUPT)

define the index for the image column (IMGKOL)

set the lower limit on H2OUPT to 0 and the upper limit as the difference in the cell's current volumetric water content and its volumetric water content at 15,000 cm (15 bars)

add the water uptake from the cell and its image to the transpiration accumulator (SUMEP)

convert H2OUPT to cm^3 day^{-1} (UPTH2O)

if there is no water uptake for this cell, ignore the remainder of the calculations for this cell—return to SOIL

*** Calculate the active and passive nitrate-N and ammonium-N uptake. ***

calculate the passive uptake of nitrate-N from the cell (UPNO3C), assuming all nitrate-N is in solution (mg N $cell^{-1}$)

calculate the passive uptake of ammonium-N from the cell (UPNH4C), assuming 10 percent of the ammonium-N is in solution (mg N $cell^{-1}$)

initialize to 0.0 mg N the active uptake of nitrate-N (AUNO3C)

calculate an empirical soil-water status factor (EFOW), which is a ratio of the difference in volumetric water content of the cell at present and the volumetric water content at 15,000 cm *over* the difference in the volumetric water content of the cell at field capacity and the volumetric water content at 15,000 cm

set the lower limit of EFOW to 0.0

if the volumetric nitrate-N content of the cell is negligible, then

> calculate the active uptake of nitrate-N for the three age classes of roots (MMUPN1, MMUPN2, and MMUPN3)
>
> calculate the amount of nitrate-N removed from the cell by active uptake (AUNO3C)
>
> set the lower limit of AUNO3C to 0.

endif

*** Calculate the active and passive nitrogen uptake from the mirror-image cells. ***

calculate the passive uptake of nitrate-N from the mirror-image cell (UPNO3I)

calculate the passive uptake of ammonium-N from the mirror-image cell (UPNH4I)

initialize to 0.0 mg N the active uptake of nitrate-N from the mirror-image cell (AUNO3I)

calculate an empirical soil-water status factor (EFOW), which is a ratio of the difference in volumetric water content of the cell at present and the volumetric water content at 15,000 cm *over* the difference in volumetric water content of the cell at field capacity and the volumetric water content at 15,000 cm

set the lower limit of EFOW to 0.0

if the volumetric nitrate-N content of the cell is negligible, then

> calculate the active uptake of nitrate-N for the three age classes of roots (MMUPN1, MMUPN2, and MMUPN3)
>
> calculate the amount of nitrate-N removed from the cell by active uptake (AUNO3I)
>
> set the lower limit of AUNO3I to 0.0

endif

*** End the active and passive uptake of nitrogen. Decrease the volumetric water content of the current cell and its mirror image by the amount of uptake from the cells. ***

update the volumetric water content of the current cell and its mirror-image cell

update the water uptake of the current cell [ZUPT(L,K)] and its mirror-image cell [ZUPT(L,IMGKOL)]

*** Decrease the volumetric nitrate-N content of the current cell and its mirror image by the amount of active and passive nitrate-N uptake from the cells. ***

if the difference between the volumetric nitrate-N content and the passive nitrate-N uptake of the cell is less than 0.0, then

the passive uptake of nitrate-N from the cell is the difference between the volumetric nitrate-N content of the cell and the minimum nitrogen flux (FLNMIN)

redefine the volumetric nitrate-N content of the cell to equal FLNMIN

set the active nitrate-N uptake of the cell to 0.0

else

the volumetric nitrate-N content of the cell is reduced by passive uptake from the cell

if the difference between the volumetric nitrate-N content of the cell and its active nitrate-N uptake is less than 0.0, then

the active nitrate-N uptake of the cell is the difference between the volumetric nitrate-N content of the cell and the minimum nitrogen flux (FLNMIN)

redefine the volumetric nitrate-N content of the cell to equal FLNMIN

else

the volumetric nitrate-N content of the cell is reduced by active uptake from the cell

end if

end if

*** Decrease the volumetric ammonium-N content of the current cell and its mirror image by the amount of passive ammonium-N uptake from the cells. ***

if the difference between the volumetric ammonium-N content and the passive ammonium-N uptake of the cell is less than 0.0, then

the passive uptake of ammonium-N from the cell is the difference between the volumetric ammonium-N content of the cell and the minimum nitrogen flux (FLNMIN)

redefine the volumetric ammonium-N content of the cell to equal FLNMIN

else

the volumetric ammonium-N content of the cell is reduced by passive uptake from the cell

end if

active and passive uptake of nitrate-N from the cell is added to an accumulator (UPNO3) to derive the amount of nitrate-N taken up from the total profile

the passive uptake of ammonium-N from the cell is added to an accumulator (UPNH4) to derive the amount of ammonium-N taken up from the total profile

End do

If this is the last iteration, then set the new plant evaporation-transpiration rate (NEWEP) to equal total plant evaporation (SUMEP) (cm^3 cm^{-3})

Return to SOIL

End UPTAKE

```
      SUBROUTINE UPTAKE
C*************************************************************
C UPTAKE OF WATER FROM EACH SOIL CELL IS PROPORTIONAL TO    *
C THE PRODUCT OF ROOT WEIGHT CAPABLE OF UPTAKE AND THE      *
C HYDRAULIC CONDUCTIVITY OF THE CELL.  THE SUM OF THE       *
C UPTAKE FROM THE CELLS EQUALS TRANSPIRATION.  ALL NO3 IN   *
C THE WATER TAKEN UP BY THE ROOTS IS ALSO TAKEN UP, AND     *
C ACTIVE UPTAKE OF NO3 IS CALCULATED USING MICHELIS-MENTON  *
C CONSTANTS FOR ROOTS OF 3 DIFFERENT AGE CLASSES.           *
C*************************************************************
C EP - TRANSPIRATION BY PLANTS, MM/DAY.
C SUPNO3 - SUPPLY OF NITRATE FROM SOIL, MG.
C SUPNH4 - SUPPLY OF AMMONIA FROM SOIL, MG.
C UPF - UPTAKE FACTOR, GM CM/DAY.
C ROOT WEIGHT CAPABLE OF UPTAKE, GM/CELL.

      INCLUDE 'GOSCOM.FOR'
C
C  SINCE UPTAKE IS CALLED TWO TIMES PER DAY DELT IS 0.5 OR (1.0/2.0)
C
      DELT = 1./NOITR
      DUMY01 = (.10*NK*WCELL*EP)*DELT

      DO 8 I=1,40
         DO 8 J=1,20
            RTWTCU(I,J) = 0.
            UPF(I,J) = 0.
    8 CONTINUE
      DO 888 I=1,41
         DO 888 J=1,21
            PUPF(I,J)  = 0.
            TUPF(I,J)  = .TRUE.
            TTUPF(I,J) = .TRUE.
  888 CONTINUE

      DO 1 L=1, LR
         KR = KRL(L)
         DO 1 K =1, KR
            RTWTCU(L,K) = (RTWT(L,K,1)+0.20*(RTWT(L,K,2)+ RTWT(L,K,3)))
C
C SUMS THE WEIGHT OF ROOTS 15 DAYS OLD OR LESS IN CELL.
C
            IF(L+2.LE.NL) THEN
               IF((RTWTCU(L,1).GT.0.).AND.(.NOT.RTEXNT(L+2)))
     .                                                 RTEXNT(L+2)=.TRUE.
            ENDIF
 1    CONTINUE
      DO 4 L  = 1, LR
         KR = KRL(L)
         DO 4 K = 1, KR
            RTWTCU(L,K) = RTWTCU(L,K) + RTWTCU(L,NK-K+1)
C
C ADDS THE ROOTS GROWN BY THE PLANTS IN THE NEXT ROW TO GET
C THE TOTAL WEIGHT OF ROOTS CAPABLE OF UPTAKE.
C
 4    CONTINUE
      NKH = NK/2
      SUPF = 0.
      DO 5 L  = 1, LR
         KR = KRL(L)
         IF (KR.GT.NKH) KR=NKH
         DO 5 K = 1, KR
```

```
            UPF(L,K)=RTWTCU(L,K)*DIFF(L,K)
C
C UPTAKE FACTOR FOR EACH CELL, HAS UNITS OF GM CM/DAY.
C
            SUPF = SUPF + UPF(L,K)
C
C SUM OF UPTAKE FACTORS IN THE PROFILE.  USED FOR APPORTIONING
C UPTAKE AMONG CELLS.
C
 5    CONTINUE
      D01 = 1./SUPF
      DO 556 L=1,LR
         KR = KRL(L)
         IF(KR.GT.NKH) KR=NKH
         DO 555 K = 1,KR
            PUPF(L,K)=UPF(L,K)*D01
  555    CONTINUE
  556 CONTINUE
      DO 656 L=1,LR
         KR=KRL(L)
         IF(KR.GT.NKH) KR=NKH
         DO 655 K=1,KR
            IF((PUPF(L,K).LE.0.005).AND.(PUPF(L+1,K).LE.0.005))
     .                                                TUPF(L,K) = .FALSE.
            IF((PUPF(L,K).LE.0.005).AND.(PUPF(L,K+1).LE.0.005).AND.
     .                              (.NOT.TUPF(L,K))) TTUPF(L,K) = .FALSE.
            I = NK + 1 - K
            TTUPF(L,I) = TTUPF(L,K)
            TUPF(L,I) = TUPF(L,K)
  655    CONTINUE
  656 CONTINUE
      UPNH4 = 0.
      UPNO3 = 0.
      DO 6 L  = 1, LR
         J = LYRDPH(L)
         KR = KRL(L)
         IF (KR.GT.NKH) KR=NKH
         DO 6 K = 1, KR
            UPTH2O = (UPF(L,K)/SUPF) * (DUMY01/2.)
            H2OUPT = UPTH2O/ACELLDW
C
C UPTAKE OF WATER FROM EACH CELL, CM**3/DAY. EP HAS UNITS OF MM/DAY.
C
            IMGKOL = NK - K + 1
C
C IMAGE COLUMN, MIRRORED ABOUT CENTERLINE OF PLANE.
C
            H2OUPT = AMAX1(0., AMIN1(VH2OC(L,K)-THETA0(J), H2OUPT))
C
C VOLUMETRIC WATER CONTENT OF CELL IS DECREASED BY AMOUNT
C OF UPTAKE FROM CELL. VOLUMETRIC WATER CONTENT OF IMAGE CELL IS
C ALSO REDUCED.
C
            SUMEP = SUMEP + (2.0*H2OUPT)
            UPTH2O = H2OUPT*ACELLDW
            IF(UPTH2O.LE.0.0)GO TO 6
C
C CALCULATE AMOUNT OF PASSIVE NO3 AND NH4 UPTAKE IN MG N/CM**3
C
            UPNO3C = UPTH2O*(VNO3C(L,K)/VH2OC(L,K))
            UPNH4C = UPTH2O*((0.10*VNH4C(L,K))/VH2OC(L,K))
C
```

```
C CALCULATE ACTIVE UPTAKE OF NO3 FROM CELL, MG N/DAY.
C
            AUNO3C = 0.0
            EFOW = (VH2OC(L,K)-THETA0(J))/(FCININ(J)-THETA0(J))
            IF(EFOW.LT.0.) EFOW=0.
C
            IF(VNO3C(L,K).GT.0.000001) THEN
               MMUPN1 = 9.00/(1.+0.000000248/(VNO3C(L,K)/VH2OC(L,K)))
               MMUPN2 = 9.00/(1.+0.000000248/(VNO3C(L,K)/VH2OC(L,K)))
               MMUPN3 = .900/(1.+0.000000248/(VNO3C(L,K)/VH2OC(L,K)))
               AUNO3C = ((MMUPN1*RTWT(L,K,1)) + (MMUPN2*RTWT(L,K,2)) +
     .                   (MMUPN3*RTWT(L,K,3))) * EFOW * DELT
               IF(AUNO3C.LT.0.0) AUNO3C = 0.0
            ENDIF
C
C PASSIVE NO3 AND NH4 UPTAKE FROM MIRROR IMAGE CELLS, MG N/CM**3 SOIL.
C
            UPNO3I = UPTH2O*(VNO3C(L,IMGKOL)/VH2OC(L,IMGKOL))
            UPNH4I = UPTH2O*((0.10*VNH4C(L,IMGKOL))/VH2OC(L,IMGKOL))
C
C CALCULATE ACTIVE UPTAKE OF NO3 FROM IMAGE CELL, MG N/CM**3.
C
            AUNO3I = 0.0
            EFOW = (VH2OC(L,IMGKOL)-THETA0(J))/(FCININ(J)-THETA0(J))
            IF(EFOW.LT.0.) EFOW=0.
            IF(VNO3C(L,IMGKOL).GT.0.000001) THEN
               MMUPN1 = 9.00/(1.+0.000000248/(VNO3C(L,IMGKOL)/
     .                  VH2OC(L,IMGKOL)))
               MMUPN2 = 9.00/(1.+0.000000248/(VNO3C(L,IMGKOL)/
     .                  VH2OC(L,IMGKOL)))
               MMUPN3 = .900/(1.+0.000000248/(VNO3C(L,IMGKOL)/
     .                  VH2OC(L,IMGKOL)))
               AUNO3I = ((MMUPN1*RTWT(L,K,1)) + (MMUPN2*RTWT(L,K,2)) +
     .                   (MMUPN3*RTWT(L,K,3))) * EFOW * DELT
               IF(AUNO3I.LT.0.0) AUNO3I = 0.0
            ENDIF
C
C NOW UPDATE WATER CONTENT OF CELLS AFTER NITROGEN MOVEMENT AND UPTAKE
C HAVE BEEN ACCOUNTED FOR.
C
            VH2OC(L,K) = VH2OC(L,K) - H2OUPT
            VH2OC(L,IMGKOL) = VH2OC(L,K)
            ZUPT(L,K) = ZUPT(L,K) + H2OUPT
            ZUPT(L,IMGKOL) = ZUPT(L,IMGKOL) + H2OUPT
C
C REDUCE VOLUMETRIC NO3 CONTENT OF CELLS, BY THE AMOUNT OF PASSIVE
C AND ACTIVE NO3 UPTAKE. MG/CM**3
C
            IF((VNO3C(L,K) - UPNO3C/ACELLDW).LT.0.0) THEN
               UPNO3C = (VNO3C(L,K) - FLNMIN/VCELL) * ACELLDW
               VNO3C(L,K) = FLNMIN/VCELL
               AUNO3C = 0.0
            ELSE
               VNO3C(L,K) = VNO3C(L,K) - UPNO3C/ACELLDW
               IF((VNO3C(L,K) - AUNO3C/ACELLDW) .LT.0.0) THEN
                  AUNO3C = (VNO3C(L,K) - FLNMIN/VCELL) * ACELLDW
                  VNO3C(L,K) = FLNMIN/VCELL
               ELSE
                  VNO3C(L,K) = VNO3C(L,K) - AUNO3C/ACELLDW
               ENDIF
            ENDIF
C
```

```
C REDUCE VOLUMETRIC NH4 CONTENT OF CELLS, BY THE AMOUNT OF
C NH4 UPTAKE. MG/CM**3
C
            IF(VNH4C(L,K) - UPNH4C/ACELLDW .LT.0.0) THEN
               UPNH4C = (VNH4C(L,K) - FLNMIN/VCELL) * ACELLDW
               VNH4C(L,K) = FLNMIN/VCELL
            ELSE
               VNH4C(L,K) = VNH4C(L,K) - UPNH4C/ACELLDW
            ENDIF
C
C ACCUMULATE UPTAKE OF NO3 AND NH4 AND THE ACTIVE UPTAKE OF NO3
C FROM EACH CELL, UNITS AREA MG N/CM**3.
C
            UPNO3 = UPNO3 + UPNO3C + AUNO3C
            UPNH4 = UPNH4 + UPNH4C
C
C REDUCE NO3 CONTENT OF IMAGE CELL, BY PASSIVE AND ACTIVE N UPTAKE.
C MG N/CM**3
C
            IF((VNO3C(L,IMGKOL) - UPNO3I/ACELLDW) .LT.0.0) THEN
               UPNO3I = (VNO3C(L,IMGKOL) - FLNMIN/VCELL) * ACELLDW
               VNO3C(L,IMGKOL) = FLNMIN/VCELL
               AUNO3I = 0.0
            ELSE
               VNO3C(L,IMGKOL) = VNO3C(L,IMGKOL) - UPNO3I/ACELLDW
               IF((VNO3C(L,IMGKOL) - AUNO3I/ACELLDW).LT.0.0) THEN
                  AUNO3I = (VNO3C(L,IMGKOL) - FLNMIN/VCELL) * ACELLDW
                  VNO3C(L,IMGKOL) = FLNMIN/VCELL
               ELSE
                  VNO3C(L,IMGKOL) = VNO3C(L,IMGKOL) - AUNO3I/ACELLDW
               ENDIF
            ENDIF
C
C REDUCE NH4 CONTENT OF IMAGE CELL, BY PASSIVE N UPTAKE. MG/CM**3
C
            IF((VNH4C(L,IMGKOL) - UPNH4I/ACELLDW).LT.0.0) THEN
               UPNH4I = (VNH4C(L,IMGKOL) - FLNMIN/VCELL) * ACELLDW
               VNH4C(L,IMGKOL) = FLNMIN/VCELL
            ELSE
               VNH4C(L,IMGKOL) = VNH4C(L,IMGKOL) - UPNH4I/ACELLDW
            ENDIF
C
C ACCUMULATE PASSIVE N UPTAKE OF NO3 AND NH4 AND ACTIVE UPTAKE OF NO3
C FROM IMAGE CELL, UNITS AREA MG N/CM**3.
C
            UPNO3 = UPNO3 + UPNO3I + AUNO3I
            UPNH4 = UPNH4 + UPNH4I
C
 6    CONTINUE
      IF(ITER.EQ.(NINT(1./DELT))) NEWEP = (SUMEP*DCELL*10.)/NK
      RETURN
      END
```

GLOSSARY

AUNO3C	Amount of nitrate-N removed from a soil cell on one iteration (mg N).
AUNO3I	Amount of nitrate-N removed from an image soil cell on one iteration (mg N).
DELT	Fraction of total daily uptake done per iteration; currently set at 0.5.
DO1	Reciprocal of SUPF.
DUMY01	Amount of transpiration water (EP) removed on this iteration (cc H_2O).
EFOW	Empirical factor of water.
EP	Plant evaporation or transpiration rate (mm H_2O day^{-1}).
H2OUPT	Amount of water removed from each cell by uptake (cc H_2O cc^{-1} soil).
IMGKOL	Index of the image column.
KR	Column index of the last rooted cell in layer L; KRL(L).
LR	Index of the last rooted layer; the deepest layer containing roots.
MMUPN1	Michelis-Menten factor 1 for active uptake.
MMUPN2	Michelis-Menten factor 2 for active uptake.
MMUPN3	Michelis-Menten factor 3 for active uptake.
NKH	Column index of the last cell in the left half of the profile.
PUPF(L,K)	Percentage uptake factor; percentage of the total uptake removed from the cell.
RTEXNT(L)	Root extent. This is a logical variable that will be true for layer L if there are roots in the layer L+2.
RTWTCU(L,K)	Roots (dry matter/weight basis) capable of uptake (mg/soil cell).
SUPF	Sum of the uptake factors for all rooted cells; used for apportioning uptake among cells.
TTUPF(L,K)	A logical variable for each cell; indicates TRUE if there is significant water uptake from the current cell, the cell directly below it, or the cell to its right.
TUPF(L,K)	A logical variable for each cell; indicates TRUE if there is significant water uptake from the current cell or the cell directly below it.
UPF(L,K)	Uptake factor for each soil cell, calculated as the product of the diffusivity of the cell and the root weight capable of uptake (mg cm^2 day^{-1}).
UPNH4C	Passive uptake of ammonium-N from a soil cell in one iteration (mg N).
UPNH4I	Passive uptake of ammonium-N from an image soil cell in one iteration (mg N).
UPNO3C	Passive uptake of nitrate-N from a soil cell in one iteration (mg N).

UPNO3I	Passive uptake of nitrate-N from an image soil cell in one iteration (mg N).
UPTH2O	Uptake of water from the soil cell (cm^3 day^{-1}).

SUBROUTINE RIMPED

Review chaired by G. Theseira and F.D. Whisler

The RIMPED subroutine incorporates a "table lookup" procedure and a linear interpolation to estimate physical impedance to root elongation. The values of impedance are selected from a family of curves that relate soil strength to bulk density and gravimetric water content (Campbell et al. 1974; Taylor and Gardner 1963). Root impedance concerns the resistance that a root encounters as it extends through the soil. Impedance values are used by the RUTGRO subroutine to modify the increase in dry weight of roots in each cell.

ASSUMPTIONS

1. Impedance curves were developed from data collected on a Norfolk sandy loam. Presumably, these data sufficiently represent root impedance found in all GOSSYM soils. When sufficient data are available, soil-specific curves will be developed.

2. The bulk density and water content of the soil are the major factors determining the physical impedance of the soil to root growth.

INPUTS

BDI
BD(J)
GH2OC(IK)
INRIM
LYRDPH(L)
NCURVE
NK
TSTBD(IK,JJ)
TSTIMP(IK,JJ)
VH2OC(IK)

OUTPUT

RTIMPD(L,K)

GENERAL PSEUDOCODE

Include the common block of variables from GOSCOM.FOR

Establish the left half of the profile on which operations will be conducted

For each layer of cells, do:

> Determine the soil horizon
>
> Locate the bulk density pointer on the impedance curve table
>
> For each column in the layer, do:
>
> > Calculate the gravimetric water content of the soil cell
> >
> > Locate the gravimetric water content pointer on the impedance curve table
> >
> > If the bulk density and water content pointers correspond to discrete points on the impedance curves, read the impedance value (soil strength) directly from the curve
> >
> > If the bulk density of the layer lies between bulk density points on the curves, interpolate to locate the position of bulk density on the impedance curves

If the soil moisture content of the cell lies between the moisture contents represented by the curves, interpolate to locate the position of soil moisture between the impedance curves

The physical impedance of the soil to root elongation is read as the soil strength corresponding to the bulk density and moisture content of the soil cell

Repeat the above-mentioned operations of locating the soil conditions and corresponding soil strength for the first 3 columns in the top soil layer

Reflect the root impedance values of the left half of the profile onto the right half of the profile

Return to RUTGRO

End RIMPED

PSEUDOCODE

Include the common block of variables from GOSCOM.FOR

Determine the number of columns of cells in one-half of the soil profile

For each layer of soil cells, do:

indicate the soil horizon where the layer is found

initialize the bulk density pointer

[A] Continue

If the bulk density of the horizon is greater than the bulk density on point JJ of curve 1, then

increment the bulk density pointer by moving to the next point on the curve

if the bulk density pointer value is less than the number of points on the curve (if there are more points on this curve), go to *[A]* (above)

Endif

if the value of the bulk density pointer is greater than the number of points on the curve, set the value to the number of points on the curve (**Note**: If the bulk density of the soil exceeds the value in the impedance data file, ignore the bulk density of the soil and use the highest bulk density on the impedance curves.)

for each column in the left half of the soil profile, do:

calculate the gravimetric water content of the soil cell (L,K)

initialize the curve pointer (IK)

if the gravimetric water content of the cell (L,K) is greater than the gravimetric water content for the curve (IK), increment the curve pointer (go to the next curve); repeat this operation until the last curve is reached

if the bulk density pointer indicates that the bulk density of the soil is equal to the lowest value on the curve, then

if the curve pointer indicates that the moisture content of the soil is equal to that of the first curve, or if the gravimetric water content of the cell is the same as that for a curve, then

root impedance in a cell is equal to the soil strength on the first point on the curve; go to [*B*] (below)

endif

else

if the moisture content of the soil is equal to the moisture content of the first curve, or if the water content of the cell is the same as that for a curve, then

root impedance is estimated by linear interpolation on the curve between bulk density points; go to *[C]* (below)

endif

Endif

if the bulk density pointer indicates that bulk density is equal to a low value in the table, then

root impedance is estimated by linear interpolation between two adjacent curves that represent moisture contents above and below the moisture content of the soil cell

else—the bulk density and moisture content of the soil lie between values on the root impedance curves—

linearly interpolate between bulk density points on curve IK and between bulk density points on curve IK-1 (**Note**: These curves represent moisture contents above and below the moisture content of the cell. This step locates soil bulk density on the curves.)

linearly interpolate between the curves discussed in the previous step (**Note**: This step locates the appropriate moisture content on the table.)

root impedance is equal to the soil strength corresponding to the bulk density and moisture content of the soil cell

Endif

[B] Continue/End do

[C] Continue/End do

Set the bulk density pointer to 1

[D] Continue

If the initial bulk density of the surface layer is greater than the bulk density corresponding to the bulk density pointer on the first curve, then

increment the bulk density pointer

if the bulk density pointer is less than the number of points on the curve, go to *[D]* (above)

Endif

If the bulk density pointer is greater than the number of points on the curve, set the pointer to equal the number of points on the curve (the last point on the curve) (**Note**: Bulk density is restricted to the range of values included in the impedance curves.)

For the first 3 soil cell columns in the top layer of the soil profile, do:

calculate the gravimetric water content for the cell

initialize the curve pointer (IK) to 1

[E] Continue

If the water content of the soil cell is greater than the water content of the curve, IK, then

increment IK (go to the next curve)

if IK is less than the number of curves (if there are still more curves), go to *[E]* (above)

Endif

if IK exceeds the number of curves, set the curve pointer (IK) on the last curve (**Note**: The water content of soil is considered only if it is less than or equal to the maximum water content of the impedance curves.)

if the bulk density pointer is less than or equal to 1, then

if the curve pointer is set on the first curve or if the water content of the soil is equal to the water content represented by a curve, then

root impedance is equal to the soil strength on the curve IK at the lowest bulk density (JJ = 1)

go to *[F]* (page 81)

endif

else

if the curve pointer is set on the first curve or if the water content of the soil is equal to the water content of a curve, then

root impedance is estimated by linear interpolation between the bulk density points on the curve

go to *[F]* (page 81)

endif

Endif

If the bulk density pointer is less than or equal to 1, then

root impedance is determined by linear interpolation between the first points on the adjacent moisture content curves immediately above and below the soil moisture content

Else

linearly interpolate between bulk density points on curve IK and between bulk density points on curve IK-1 (**Note**: These curves represent moisture contents above and below the soil moisture content of the cell. This step locates the bulk density on the curves.)

linearly interpolate between the curves discussed in the previous step (**Note**: This step locates the appropriate moisture content on the table. Root impedance is equal to the soil strength that corresponds to the bulk density and moisture content of the soil cell.)

Endif

[F] Continue/End do

Reflect the soil impedance values for the left half of the soil profile onto the right half of the profile

Return to RUTGRO

End RIMPED

```
      SUBROUTINE RIMPED
C  ********************************************************************
C  THIS SUBROUTINE CALCULATES ROOT IMPEDENCE BASED UPON THE BULK      *
C  DENSITY AND WATER CONTENT. THIS IS BASED UPON DATA FROM ARTICLES BY *
C  R.B.CAMPBELL, D.C.REICOSKY, AND C.W.DOTY J.OF SOIL AND WATER CONS.  *
C  29:220-224,1974 AND                                                 *
C  H.M.TAYLOR AND H.R.GARDNER. SOIL SCI.96:153-156,1963.               *
C  A LINEAR TABLE LOOK-UP PROCEDURE IS USED. ASSUME ALL CURVES ARE     *
C  READ AT THE SAME BD.                                                *
C                                                                      *
C  ********************************************************************
C
      INCLUDE 'GOSCOM.FOR'
C
      NKH = NK/2
      DO 99 L = 1,NL
         J = LYRDPH(L)
         JJ = 1
   26    IF(BD(J).GT.TSTBD(1,JJ)) THEN
            JJ = JJ+1
            IF(JJ.LT.INRIM) GO TO 26
         ENDIF
         IF(JJ.GT.INRIM) JJ=INRIM
         DO 98 K = 1,NKH
            TEST1=VH2OC(L,K)/BD(J)
            IK = 1
   32       IF(TEST1.GT.GH2OC(IK)) THEN
               IK = IK+1
               IF(IK.LT.NCURVE) GO TO 32
            ENDIF
C
C  SOIL CELL H2O LESS THAN TEST H2O
C
            IF(JJ.EQ.1) THEN
               IF((IK.EQ.1).OR.(TEST1.EQ.GH2OC(IK))) THEN
                  RTIMPD(L,K)=TSTIMP(IK,JJ)
                  GO TO 98
               ENDIF
            ELSE
               IF((IK.EQ.1).OR.(TEST1.EQ.GH2OC(IK))) THEN
                  RTIMPD(L,K)=TSTIMP(IK,JJ-1)-(TSTIMP(IK,JJ-1)-
     .                        TSTIMP(IK,JJ))*((TSTBD(IK,JJ-1)-BD(J))/
     .                        (TSTBD(IK,JJ-1)-TSTBD(IK,JJ)))
                  GO TO 98
               ENDIF
            ENDIF
C
C  CALCULATE SOIL STRENGTH FOR VALUES OF BD LESS THAN TABLE VALUES
C
            IF(JJ.EQ.1) THEN
               RTIMPD(L,K)=TSTIMP(IK-1,JJ)-(TSTIMP(IK-1,JJ)
     .                     -TSTIMP(IK,JJ))*((TEST1-GH2OC(IK-1))/
     .                     (GH2OC(IK)-GH2OC(IK-1)))
            ELSE
C
C   FOR VALUES OF BD AND H2O BETWEEN TABLE VALUES
C
               TEMP1R=TSTIMP(IK,JJ-1)-(TSTIMP(IK,JJ-1)-TSTIMP(IK,JJ))*
     .                ((TSTBD(IK,JJ-1)-BD(J))/(TSTBD(IK,JJ-1)
     .                -TSTBD(IK,JJ)))
               TEMP2=TSTIMP(IK-1,JJ-1)-(TSTIMP(IK-1,JJ-1)-
     .               TSTIMP(IK-1,JJ))*((TSTBD(IK-1,JJ-1)-BD(J))/
```

```
     .                (TSTBD(IK-1,JJ-1)-TSTBD(IK-1,JJ)))
                RTIMPD(L,K)=TEMP2+(TEMP1R-TEMP2)*((TEST1-GH2OC(IK-1))/
     .                        (GH2OC(IK)-GH2OC(IK-1)))
             ENDIF
  98      CONTINUE
  99   CONTINUE
       JJ = 1
126    IF(BDI.GT.TSTBD(1,JJ)) THEN
          JJ = JJ + 1
          IF(JJ.LT.INRIM) GO TO 126
       ENDIF
       IF(JJ.GT.INRIM) JJ=INRIM
       DO 198 K=1,3
          TEST1 = VH2OC(1,K)/BDI
          IK = 1
132       IF(TEST1.GT.GH2OC(IK)) THEN
             IK = IK + 1
             IF(IK.LT.NCURVE) GO TO 132
          ENDIF
          IF(IK.GT.NCURVE) IK=NCURVE
          IF(JJ.LE.1) THEN
             IF((IK.EQ.1).OR.(TEST1.EQ.GH2OC(IK))) THEN
                RTIMPD(1,K) = TSTIMP(IK,JJ)
                GO TO 198
             ENDIF
          ELSE
             IF((IK.EQ.1).OR.(TEST1.EQ.GH2OC(IK))) THEN
                RTIMPD(1,K) = TSTIMP(IK,JJ-1) - (TSTIMP(IK,JJ-1)-
     .                        TSTIMP(IK,JJ)) * ((TSTBD(IK,JJ-1)-BDI)/
     .                        (TSTBD(IK,JJ-1)-TSTBD(IK,JJ)))
                GO TO 198
             ENDIF
          ENDIF
          IF(JJ.LE.1) THEN
             RTIMPD(1,K) = TSTIMP(IK-1,JJ) - (TSTIMP(IK-1,JJ)-
     .                     TSTIMP(IK,JJ))*((TEST1-GH2OC(IK-1))/
     .                     (GH2OC(IK)-GH2OC(IK-1)))
          ELSE
             TEMP1R = TSTIMP(IK,JJ-1)-(TSTIMP(IK,JJ-1)-TSTIMP(IK,JJ))*
     .               ((TSTBD(IK,JJ-1)-BDI)/(TSTBD(IK,JJ-1)-TSTBD(IK,JJ)))
             TEMP2 = TSTIMP(IK-1,JJ-1)-(TSTIMP(IK-1,JJ-1)
     .               -TSTIMP(IK-1,JJ))*((TSTBD(IK-1,JJ-1)-BDI)/
     .               (TSTBD(IK-1,JJ-1)-TSTBD(IK-1,JJ)))
             RTIMPD(1,K) = TEMP2+(TEMP1R-TEMP2)*((TEST1-GH2OC(IK-1))/
     .                     (GH2OC(IK)-GH2OC(IK-1)))
          ENDIF
198    CONTINUE
       NKH = NKH+1
       DO 109 K=NKH,NK
          NKK=NK+1-K
          DO 109 L = 1,NL
             RTIMPD(L,K)=RTIMPD(L,NKK)
 109   CONTINUE
       RETURN
       END
```

GLOSSARY

BDI	Initial bulk density of the surface soil layer (g soil cm^{-3} soil).
BD(J)	Bulk density of soil in horizon J (g soil cm^{-3} soil).
GH2OC(IK)	Corresponding gravimetric water content for curve IK; read from the soil impedance file (g H_2O g $soil^{-1}$).
INRIM	Number of input (bulk density) data points on the impedance curve.
LYRDPH(L)	Integer counter for the soil horizon number of layer L.
NCURVE	Number of input (soil moisture) curves on the impedance table.
NK	Number of vertical columns of soil cells in the profile.
RTIMPD(L,K)	Root impedance for soil cell (L,K) (kg cm^{-2}).
TSTBD(IK,JJ)	Corresponding bulk density on the impedance table (g cm^{-3}).
TSTIMP(IK,JJ)	Corresponding soil strength on the impedance table (kg cm^{-2}).
VH2OC(L,K)	Volumetric water content of the cell (cm^3 H_2O cm^{-3} soil).

Note: Array variables are designated by the following:

IK represents the curve number (each curve represents a given gravimetric water content),

JJ represents the data point on the curve (each point represents a given bulk density),

K represents the soil cell column number,

L represents the soil cell row number.

SUBROUTINE RRUNOFF

Written by D.O. Porter

RRUNOFF applies the "curve number method" of the Soil Conservation Service (U.S. Department of Agriculture 1972) to estimate daily rainfall or irrigation losses from surface runoff. Prior to the addition of this routine in 1990, it was assumed that all rainfall and irrigation water was taken into the soil profile. Called from the CLYMAT subroutine, RRUNOFF is essentially a "table lookup" procedure designed to account for differences between the amount of water applied to the field and the amount actually available to meet evapotranspiration demands. Because some producers prefer to assume no losses of rainfall or irrigation water or both from runoff, a flag read from the soil hydrology profile allows the routine to be turned on or off.

Equations used in RRUNOFF are as follows:

$$Q = \frac{(I-0.2S)^2}{I + 0.8S}$$

where

Q = depth of direct surface runoff (mm)
I = storm rainfall (mm)
S = maximum potential difference between rainfall and runoff (mm)
0.2S = initial abstraction, consisting of interception losses, surface storage, and infiltration before runoff commences.

$$S = \frac{25400}{N} - 254$$

where

N = the curve number with a value between 0 and 100 (N = 100 represents the greatest runoff potential).

The curve number, N, is selected according to land use or cover, treatment or practice, and hydrologic condition. N is then adjusted for soil group, which represents the soil's infiltration capacity, and for antecedent rainfall condition, which represents changed infiltration capacity due to unusually wet or dry conditions.

ASSUMPTIONS

1. Infiltration capacity is determined primarily by soil texture. Clay soils are assumed to have high runoff potentials and sands, low runoff potentials.

2. The Soil Conservation Service (SCS) curve number is based on straight-row operations on soils of good hydrologic condition. Adjustments are made to accommodate antecedent moisture conditions and hydrologic soil groups.

3. Runoff of irrigation water behaves the same way as does runoff of rainfall.

4. Effective rainfall is determined by subtracting the depth of estimated runoff from the depth of applied rainfall or irrigation.

5. Slope is not a factor in calculating runoff.

INPUTS

AMTIRR(K)
CLIMAT(K,5)
DAYNUM
IDAY

IFGIRR
IFGRAIN
IPCLAY(1)
IPSAND(1)
LDAYIR
RAIN

OUTPUTS

RAIN
RUNOFF(IDAY)

GENERAL PSEUDOCODE

If the runoff flag indicates that the RRUNOFF subroutine will be applied, then

Determine the 5-day antecedent rainfall

Determine the antecedent rainfall condition (condition I, II, or III)

Determine the soil group (B, C, or D) based on soil texture

Set the unadjusted curve number

Adjust the curve number for the soil group

Adjust the curve number for antecedent rainfall conditions

Calculate the depth of expected runoff

Calculate the effective rainfall

Endif

Return to CLYMAT

End RRUNOFF

PSEUDOCODE

*** IDUM allows the runoff subroutine to be turned on or off, depending on whether the user wishes to apply the routine to rainfall or irrigation or both. ***

Calculate the 5-day antecedent moisture application (**Note**: D11 is the sum of all water applied to the field within the last 5 days, converted from inches to millimeters.)

*** Set the variable for the antecedent rainfall condition (I03)(Schwab et al. 1981). ***

If the 5-day antecedent rainfall is less than 36 mm, then set I03 = 1 (condition I, optimal soil condition from about the lower plastic limit to the wilting point)

Else if the 5-day antecedent rainfall is greater than 53 mm, set I03 = 3 (condition III, heavy rainfall)

Else set I03 = 2 (condition II, average value for annual floods)

Endif

*** Determine the texture of the surface horizon. ***

If the soil contains more than 70 percent sand and less than 15 percent clay, it is a sand (I04 = 1)

Else if the clay content is greater than 35 percent, the soil is a clay (I04 = 3)

Else the soil is a loam (I04 = 2)

Endif

*** For purposes of estimating runoff, sands are assumed to have moderately low runoff potentials (group B); clays, high runoff potentials (group D); and loams, moderately high runoff potentials group C) (Schwab et al. 1981; Brady 1984). ***

Set the unadjusted SCS curve number (CRVNUM = 78.0) (**Note**: This value assumes straight-row cropping and good hydrologic conditions. The value will be adjusted for moisture condition and soil group

Set the coefficient D01 to adjust for soil group, if other than soil group B

Set the coefficient D02 to adjust for antecedent rainfall conditions I, II, or III

Adjust CRVNUM for moisture condition and soil group by multiplying it by coefficients D01 and D02

Calculate the maximum potential difference between rainfall and runoff, D03

Calculate the runoff depth for the day [RUNOFF(IDAY)]

If rainfall is less than the assumed initial abstraction (.2*D03), there is no runoff

Calculate the effective water application (RAIN) by subtracting RUNOFF from the total application [AMTIRR() + CLIMAT()]

Return to CLYMAT

End RRUNOFF

```
      SUBROUTINE RRUNOFF
C   ****************************************************************
C   *                                                              *
C   *   RUNOFF SUBROUTINE.  CALCULATES PORTION OF RAINFALL THAT    *
C   *   IS LOST TO RUNOFF, REDUCES RAINFALL TO THAT WHICH IS       *
C   *   INFILTRATED IN THE SOIL.  SUBROUTINE USES THE SOIL         *
C   *   CONSERVATION SERVICE METHOD OF ESTIMATING RUNOFF.          *
C   *   REFERENCE:  BRADY, NYLE C. 1984. THE NATURE AND            *
C   *   PROPERTIES OF SOILS, 9TH ED. MACMILLAN PUBLISHING CO.      *
C   *   REFERENCE:  SCHWAB, FREVERT, EDMINSTER, AND BARNES.        *
C   *   1981. SOIL AND WATER CONSERVATION ENGINEERING, 3RD         *
C   *   ED. JOHN WILEY & SONS, INC.                                *
C   *                                                              *
C   ****************************************************************

      INCLUDE 'GOSCOM.FOR'
      DIMENSION D02(3,3)

      IDUM=IFGRAIN
      IF(DAYNUM.EQ.LDAYIR) IDUM=IFGIRR
      IF(IDUM.EQ.0) THEN

C  LOOP TO ACCUMULATE 5-DAY ANTECEDENT RAINFALL (MM) WHICH
C  WILL AFFECT THE SOIL'S ABILITY TO ACCEPT NEW RAINFALL.
C  ANTECEDENT RAINFALL INCLUDES RAINFALL AND ALL IRRIGATION.
C  D11 = 5-DAY TOTAL

         D11 = 0.0
         I01 = IDAY-6
         IF(I01.LT.1) I01=1
         I02=IDAY-1
         IF(I02.LT.1) I02=1
         DO 10 K = I01,I02
            D11 = D11+(AMTIRR(K)+CLIMAT(K,5))*25.4
   10    CONTINUE

C  I03 IS AN INDICATOR OF THE ANTECEDENT MOISTURE CONDITIONS.
C  1=CONDITION I, LOW MOISTURE, LOW RUNOFF POTENTIAL.
C  3=CONDITION III, WET CONDITIONS, HIGH RUNOFF POTENTIAL.
         IF (D11.LT.36)THEN
            I03=1
         ELSE IF (D11.GT.53)THEN
            I03=3
         ELSE
            I03=2
         ENDIF

C  CALCULATE THE FINAL INFILTRATION RATE FOR Ap HORIZON.  FINAL
C  INFILTRATION IS ESTIMATED FROM THE PERCENT SAND AND PERCENT
C  CLAY IN THE Ap LAYER.  IF CLAY CONTENT IS GREATER THAN 40%,
C  THE SOIL IS ASSUMED TO HAVE A HIGHER RUNOFF POTENTIAL (SOIL
C  GROUP C=3).  IF CLAY CONTENT IS LESS THAN 15% AND SAND IS
C  GREATER THAN 70%, A LOWER RUNOFF POTENTIAL IS ASSUMED (SOIL
C  GROUP A=1).  OTHER SOILS (LOAMS) ASSUMED MODERATELY LOW RUNOFF
C  POTENTIAL (SOIL GROUP B=2).  NO 'IMPERMEABLE' (GROUP D) SOILS
C  ARE ASSUMED.REFERENCES: SCHWAB, BRADY.

         IF(IPSAND(1).GT.70.AND.IPCLAY(1).LT.15)THEN
            I04 = 1
         ELSEIF(IPCLAY(1).GT.35)THEN
            I04 = 3
         ELSE
```

```
            I04=2
         ENDIF

C   ASSUME STRAIGHT ROW, GOOD PRACTICE CROPPING PRACTICE.
C   RUNOFF CURVE NUMBER, UNADJUSTED FOR MOISTURE AND SOIL TYPE.

         CRVNUM=78.0

C   ADJUST CURVE NUMBER FOR SOIL GROUP A,B,C.  COEFFICIENTS ADJUST FOR
C   SOIL GROUPS OTHER THAN 'B', WHICH IS ASSUMED IN THE SCS EQUATION.

         IF(I04.EQ.3)THEN
            D01=1.14
         ELSEIF(I04.EQ.2)THEN
            D01=1.09
         ELSEIF(I04.EQ.1)THEN
            D01=1.0
         ENDIF

C   ADJUST CURVE NUMBER FOR ANTECEDENT RAINFALL CONDITIONS I,II,III.

         D02(1,1)=0.71
         D02(1,2)=0.78
         D02(1,3)=0.83
         D02(2,1)=1.00
         D02(2,2)=1.00
         D02(2,3)=1.00
         D02(3,1)=1.24
         D02(3,2)=1.15
         D02(3,3)=1.10

         CRVNUM = CRVNUM*D01*D02(I03,I04)

C   EFFECTIVE RAINFALL = RAINFALL (OR IRRIGATION) - RUNOFF.
C   D03 = MAX POTENTIAL DIFFERENCE BETWEEN RAINFALL AND RUNOFF.

         D03 = 25400./CRVNUM - 254.0
         RUNOFF(IDAY) = ((RAIN-0.2*D03)**2)/(RAIN+0.8*D03)
         IF (RAIN.LE.0.2*D03) RUNOFF(IDAY)=0.0
         RAIN = RAIN-RUNOFF(IDAY)
      ENDIF
      RETURN
      END
```

GLOSSARY

AMTIRR(K)	Daily irrigation amount (inches H_2O).
CLIMAT(K,5)	Daily rainfall (inches H_2O).
CRVNUM	Soil Conservation Service curve number; used in calculating the maximum potential difference between rainfall and runoff depths; value is determined from a "table lookup" procedure and based on soil texture and moisture status (no units).
D01	A multiplication factor used to adjust the curve number for a soil group (no units).
D02	A multiplication factor used to adjust the curve number for antecedent soil moisture conditions; values depend on the soil group (A, B, C, D) and moisture condition (I, II, III) (no units).
D03	A temporary variable representing the maximum potential difference between rainfall and runoff (mm H_2O).
D11	A temporary variable used to store the cumulative water applied over the preceding 5 days (mm).
DAYNUM	Number of the day of the year (Julian days).
I01	A temporary variable used to begin the 5-day count for determining antecedent moisture (days).
I02	A temporary variable to mark the end of the 5-day count for determining antecedent moisture (days).
I03	A temporary variable used to indicate antecedent moisture conditions (1 = soil condition I, 2 = soil condition II, 3 = soil condition III) (no units).
I04	A temporary varable used to indicate soil group, according to runoff potential (no units).
IDAY	Day of simulation (days).
IDUM	A dummy variable used as a flag to indicate if RRUNOFF will be applied (no units).
IFGIRR	A flag to indicate whether RRUNOFF will apply to irrigation applications in this GOSSYM run (no units).
IFGRAIN	A flag to indicate whether RRUNOFF will apply to rainfall applications in this GOSSYM run (no units).
IPCLAY(1)	Percentage clay content of the top soil layer; used in estimating final infiltration rate potential (%).
IPSAND(1)	Percentage sand content of the top soil layer; used in estimating final infiltration rate potential (%).
LDAYIR	Variable to indicate the day of irrigation (day).
RAIN	Total water application for the day (IDAY), including rainfall and irrigation (mm H_2O).
RUNOFF(IDAY)	Estimated runoff from rainfall or irrigation or both for the day (IDAY) (mm H_2O).

SUBROUTINE TMPSOL

Review chaired by F.D. Whisler

The TMPSOL subroutine, called from the CLYMAT routine, uses daily maximum, minimum, and average air temperatures to create a temperature profile for the soil. This temperature profile is used in the NITRIF and RUTGRO subroutines.

TMPSOL applies multiple regression equations from the work of McWhorter and Brooks (1965) to calculate maximum and minimum soil temperatures at depths of 2, 4, 8, and 16 inches. Daytime and nighttime temperatures are determined from average hourly temperatures and an algorithm by Stapleton et al. (undated). The routine does not account for heterogeneous soil conditions or variations in thermal properties or soil moisture. Data collected in the 1991 season are being used to develop a new soil temperature model.

ASSUMPTIONS

1. The thermal properties of the soil are homogeneous throughout the profile.
2. The effects of soil moisture on soil temperature are neglected.
3. Color, texture, and other properties that may influence the thermal properties of the soil are neglected.

INPUTS

DAYLNG
DTAVG()
NL
TAVG

OUTPUTS

SOILT(LAYER,K)
TSOILD(LAYER)
TSOILN(LAYER)
TSOLAV(LAYER)

GENERAL PSEUDOCODE

Calculate the 7-day average daily air temperature

Calculate the high and low temperatures for soil depths of 2, 4, 8, and 16 inches

Calculate the maximum temperature for each soil cell layer

Calculate the minimum temperature for each soil cell layer

Convert the soil temperatures to Celsius

Determine the hour of sunrise and the hour of sunset

Estimate the hourly temperatures for each soil cell layer

Calculate the daytime and nighttime average temperatures for each cell layer

Calculate the daily average temperature for each soil cell layer

Set the temperature of each soil cell to the average daily temperature of its soil cell layer

Return to CLYMAT

End TMPSOL

PSEUDOCODE

Include the common block of variables from GOSCOM.FOR

Determine the 7-day average air temperature

Use regression equations from McWhorter and Brooks (1965) to calculate maximum and minimum daily soil temperatures at depths of 2, 4, 8, and 16 inches

Approximate the maximum daily soil temperature for each soil cell layer to a depth of 100 cm [**Note**: Use linear approximation between calculated values (2-, 4-, and 8-inch depths) for the first 4 layers.]

Approximate the minimum daily soil temperature for each soil cell layer to a depth of 100 cm [**Note**: Use linear approximation between calculated values (2-, 4-, and 8-inch depths) for the first 4 layers.]

Convert the maximum and minimum soil temperatures of each layer from Fahrenheit to Celsius

Determine the hours of sunrise and sunset (**Note**: These values are determined from daylength.)

For each of the upper 20 soil cell layers in the profile, do:

> calculate the mean temperature of the soil cell layer
>
> calculate the amplitude of fluctuation between the maximum and minimum temperatures of the cell layer
>
> calculate the hourly soil temperatures for the soil cell layer, using cosine functions and the algorithm from Stapleton et al. (undated)
>
> calculate average daytime and nighttime temperatures for the soil in the layer

End do

For the top half of the soil cell layers, restrict the average daytime and nighttime temperatures to values above selected minimum values

Calculate the daily average temperatures, taking into account daytime and nighttime averages and daylength

For each soil layer, do:

> for each column in the layer, do:
>
> if the soil cell layer is within the top 100 cm (the top 20 cell layers), set the temperature of the cell to equal the average temperature of the cell layer; else, set the temperature of the soil cell to 25 °C
>
> end do

End do

Return to CLYMAT

End TMPSOL

```
      SUBROUTINE TMPSOL
C*********************************************************************
C  THIS SUBROUTINE CALCULATES A TEMPERATURE PROFILE IN THE SOIL,    *
C  ASSUMES HORIZONTAL HOMOGENEITY OF TEMPERATURE & DISREGARDS       *
c  MOISTURE CONTENT EFFECTS.  FIRST, MAXIMUM (H) & MINIMUM (L)      *
C  TEMPERATURES ARE CALCULATED AT 2, 4, 8, & 16 INCH DEPTHS BY      *
C  MULTIPLE REGRESSION EQUATIONS OF J.C. MCWHORTER & B.P. BROOKS,   *
C  JR.  1965. CLIMATOLOGICAL AND SOLAR RADIATION RELATIONSHIPS.     *
C  BULL. 715, MISS. AGRI. EXP. STA., STARKVILLE.  NOTE THAT THE     *
C  GRID SIZE (D*W) IS NOT VARIABLE IN THIS SUBROUTINE, BUT THE      *
C  LAYER THICKNESS IS FIXED AT 5 CM.  MAX & MIN SOIL TEMPS FOR       *
C  EACH OF THE LAYERS ARE THEN OBTAINED BY INTERPOLATION &          *
C  EXTRAPOLATION OF THE 2, 4, 8, & 16 INCH TEMPS.                   *
C    FINALLY, DAYTIME AND NIGHTTIME TEMPS (TSMX & TSMN) ARE         *
C  OBTAINED AS AVERAGE HOURLY VALUES FROM 7 A.M. THRU SUNSET, &     *
C  SUNSET THRU 7 A.M., RESPECTIVELY, USING AN ALGORITHM FOR AIR     *
C  TEMP PUBLISHED BY H. N. STAPLETON, D. R. BUXTON, F. L. WATSON,   *
C  D. J. NOLTING, AND D. N. BAKER.  UNDATED.  COTTON: A COMPUTER    *
C  SIMULATION OF COTTON GROWTH.  TECH. BULL. 206, ARIZONA AGRI.     *
C  EXP. STA. TUCSON.                                                 *
C*********************************************************************
C
      INCLUDE 'GOSCOM.FOR'
C
      DO 1 I = 1,6
         J = 8 - I
         JM1 = J - 1
    1 DTAVG(J) = DTAVG(JM1)
      DTAVG(1) = TAVG
      WTAVG = 0.
      DO 2 J = 1,7
    2 WTAVG = WTAVG +  DTAVG(J)
      WTAVG = WTAVG/7.
      WTAVGF = WTAVG*1.8 + 32.
C THE NEXT EIGHT EQUATIONS ARE FROM MCWHORTER AND BROOKS.
      T2H  = 1.1962*WTAVGF + 0.27389
      T2L  = 0.960*WTAVGF  + 1.4404
      T4H  = 1.1493*WTAVGF + 1.1452
      T4L  = 0.9126*WTAVGF + 2.9961
      T8H  = 0.9655*WTAVGF + 8.3121
      T8L  = 0.8700*WTAVGF + 7.9217
      T16H = 0.8409*WTAVGF + 13.988
      T16L = 0.8341*WTAVGF + 13.029
C GET TEMP OF SOIL ( MAX ) BY INTERPOLATION OR EXTRAPOLATION.
      T24 = T2H - T4H
      T48 = T4H - T8H
      TSMX(1) = T2H + (.507874) * T24
      TSMX(2) = T4H + (.523622) * T24
      TSMX(3) = T8H + (.769685) * T48
      TSMX(4) = T8H + (.277559) * T48
      T816 = .0492126 * (T8H - T16H)
      DO 6 I=5,20
         TSMX(I) = T8H - (2.18+(I-5)*5.) * T816
 6    CONTINUE
C GET TEMP OF SOIL (MIN) BY INTERPOLATION OR EXTRAPOLATION.
      T24 = T2L - T4L
      T48 = T4L - T8L
      TSMN(1) = T2L + (.507874) * T24
      TSMN(2) = T4L + (.523622) * T24
      TSMN(3) = T8L + (.769685) * T48
      TSMN(4) = T8L + (.277559) * T48
      T816 = .0492126 * (T8L - T16L)
```

```
      DO 7 I=5,20
         TSMN(I) = T8L - (2.18+(I-5)*5.) * T816
         IF(TSMN(I).LT.TSMX(I)) GO TO 7
         TSMN(I) = (TSMN(I) + TSMX(I))/2.
         TSMX(I) = TSMN(I)
 7    CONTINUE
      DO 8 I=1,20
C CONVERT TEMPS TO CENTIGRADE.
         TSMX(I) = (TSMX(I)-32.)*.555556
         TSMN(I) = (TSMN(I)-32.)*.555556
 8    CONTINUE
      ISR = 12 - IFIX(DAYLNG*.5)
      ISS = ISR + IFIX(DAYLNG+0.5)
C HOUR OF SUNSET.
C SEE PP 37 OF STAPLETON, ET AL. FOR EQUATIONS DETERMINING RECDAT.
      DO 9 LAYER = 1,20
         TMEAN = (TSMX(LAYER)+TSMN(LAYER)) * .5
         SWINGH = (TSMX(LAYER)-TSMN(LAYER)) * .5
         DO 11 IH=7,15
            RECDAT(IH) = TMEAN - SWINGH*COS(0.3927*(IH-7.))
            IH9 = IH + 9
            RECDAT(IH9) = TMEAN + SWINGH*COS(0.19635*(IH9-15.))
 11      CONTINUE
         DO 12 IH=1,6
 12      RECDAT(IH) = TMEAN - SWINGH*COS(0.19635*(7-IH))
         SHRTD = 0.
         SHRTN = 0.
         DO 13 IH=7,ISS
            SHRTD = SHRTD + RECDAT(IH)
C SUM OF HOURLY TEMPS IN DAYTIME.
 13      CONTINUE
         TSOILD(LAYER) = SHRTD/(ISS-6)
C AVERAGE TEMP OF SOIL DURING DAYTIME, DEG C.
         ISS1 = ISS + 1
         DO 14 IH=ISS1,24
            SHRTN = SHRTN + RECDAT(IH)
C SUM OF HOURLY TEMPS IN NIGHTTIME.
 14      CONTINUE
         DO 15 IH=1,6
            SHRTN = SHRTN + RECDAT(IH)
 15      CONTINUE
         TSOILN(LAYER) = SHRTN/(30-ISS)
C AVERAGE TEMP OF SOIL DURING NIGHTTIME.
 9    CONTINUE
      NLH = NL/2
      DO 16 LAYER = 1, NLH
         IF(LAYER.GT.10.AND.TSOILD(LAYER).LT.22.) TSOILD(LAYER)=22.0
         IF(LAYER.GT.10.AND.TSOILN(LAYER).LT.20.) TSOILN(LAYER)=20.0
         IF(LAYER.GT.15.AND.TSOILD(LAYER).LT.25.) TSOILD(LAYER)=25.0
         IF(LAYER.GT.15.AND.TSOILN(LAYER).LT.25.) TSOILN(LAYER)=25.0
         TSOLAV(LAYER)= (TSOILD(LAYER)*DAYLNG+TSOILN(LAYER)*
     .                    (24.-DAYLNG))/24.
 16   CONTINUE
      DO 20 LAYER = 1,NL
         DO 20 K=1,20
            IF(LAYER.LE.20) THEN
               SOILT(LAYER,K)=TSOLAV(LAYER)
            ELSE
               SOILT(LAYER,K)=25.0
            ENDIF
 20   CONTINUE
C AVERAGE SOIL TEMPERATURE, DEG C.
```

```
RETURN
END
```

GLOSSARY

DAYLNG	Length of day; number of daylight hours (hrs).
DTAVG()	Average daily temperature; used in calculating the running 7-day average (°C).
ISR	Time of sunrise (hr).
ISS	Time of sunset (hr).
K	Soil cell column index.
LAYER	Soil cell layer index.
NL	Number of layers (horizontal rows) of soil cells in the profile.
NLH	One-half the number of soil cell layers.
RECDAT(hour)	Hourly temperature for the current soil cell layer (°C).
SHRTD	Sum of hourly temperatures during the daytime (°C).
SHRTN	Sum of hourly temperatures during the nighttime (°C).
SOILT(LAYER,K)	Temperature of soil in cell (LAYER,K) (°C).
SWINGH	One-half the difference between the maximum and minimum daily soil temperatures (°C).
T2H	Maximum soil temperature at a 2-inch depth (°F).
T2L	Minimum soil temperature at a 2-inch depth (°F).
T4H	Maximum soil temperature at a 4-inch depth (°F).
T4L	Minimum soil temperature at a 4-inch depth (°F).
T8H	Maximum soil temperature at an 8-inch depth (°F).
T8L	Minimum soil temperature at an 8-inch depth (°F).
T16H	Maximum soil temperature at a 16-inch depth (°F).
T16L	Minimum soil temperature at a 16-inch depth (°F).
T24	Difference between soil temperatures at 2-inch and 4-inch depths (°F).
T48	Difference between soil temperatures at 4-inch and 8-inch depths (°F).
T816	Intermediate variable equal to the difference between soil temperatures at 8-inch and 16-inch depths multiplied by a constant (°F).
TAVG	Mean daily temperature, calculated from wet-bulb and dry-bulb temperatures (°C).
TMEAN	Mean daily temperature of the soil, equal to the arithmetic mean of maximum and minimum daily temperatures (°C).
TSMN(I)	Minimum daily nighttime soil temperature (°C).
TSMX(I)	Maximum daily daytime soil temperature (°C).

TSOILD(LAYER)	Average daytime temperature of the soil at the depth corresponding to the given layer of soil cells (°C).
TSOILN(LAYER)	Average nighttime temperature of the soil at the depth corresponding to the given layer of soil cells (°C).
TSOLAV(LAYER)	Average daily temperature of the soil at the depth corresponding to the given layer of soil cells (°C).
WTAVG	Running average temperature over a 7-day period (°C).
WTAVGF	Weekly average temperature; 7-day average temperature (°F).

SUBROUTINE RUTGRO

Revised by S.B. Turner
Review chaired by M.Y.L. Boone

This subroutine is called twice a day. On the first call, the day's potential root growth (based on dry matter) is calculated for each soil cell, depending on soil impedance, soil temperature, soil water, soil nitrogen, roots present that are capable of growth, and the carbohydrate supply available for root growth. Growth is determined in response to daytime and nighttime environmental conditions.

On the second call, actual root growth is calculated based on the actual amount of carbohydrate available for root growth. The actual growth for a given cell may occur within the cell and in the cells to the right, left, and below. Growth in the 4 available cells is based on the relative water potentials of the 4, with a heavier weighting given to downward growth. The subroutine also calculates average soil water potential.

ASSUMPTIONS

1. There is no restriction to root growth at leaf water potential above −7 bars (−7,000 cm). However, no growth occurs below this threshold (Boyer 1970). The volumetric nitrogen content in the soil that determines potential root growth is set (arbitrarily) at 5×10^{-4} mg N cm^{-3} of soil.

2. Roots between 5 and 15 days old are capable of being sloughed from the plant. If cotton roots live to be 15 days old, they harden and live until they die from environmental causes or lack of energy for respiration (Huck et al. 1975). The sloughing factor (SLF) is a fraction of young and old roots and has an arbitrary value of 0.02.

3. In order to reduce computational time, only root growth from the left row to the right row is detailed (fig. 3). This configuration is possible because root growth from the right row to the left or into the plane being simulated is assumed to be symmetrical to root growth from the left row to the right. Thus, total root density is the sum of root growth in each of the rows.



Figure 3. The geometry of root growth hypothesizes that roots in soil cell L,K may grow internally in L,K, left into cell L, K−1 right into cell L,K+1, and below into cell (L+1,K).

4. The boundary conditions that place limits on root growth are defined by the symmetry of root growth and the lower limit of the root zone. It is assumed there is no root growth under and across the plane of each row. Roots are further inhibited from spreading beyond a given row to any adjacent row and are not allowed to grow past the bottom of the root zone.

5. Growth originating in any soil cell L,K may add to root weight in any of the 4 cells: cell (L,K), the cell to the right, the cell to the left, or the cell below (fig. 3). How much additional weight applies to each of the 4 cells depends on the relative water potentials of the 4 cells, root impedance, and a threshold limit of root weight (g dry matter cm^{-3} soil). The additional weight given to downward growth accounts for geotropism.

6. Roots are divided into three age classes: Age class 1 includes roots less than 5 days old; age class 2 contains roots 5–15 days old; and age class 3 has roots more than 15 days old. Partitioning coefficients for moving root materials from one age class to another are set to 0.3 for age class 1 and 0.1 for age class 2 (Bar-Yosef and Lambert 1977).

7. Average soil water potential is determined from the portion of the profile occupied by roots. Cells with soil water potential less than −15 bars are not to be included in the calculation.

INPUTS

ACELLDW
ARDRCN(IROW)
DAYNUM
DAYTYM
DWRT (I,J)
EPAVG(J)
FCININ(N)
GEOTR
KALL
KDAY
KRL
KRL(L)
KULDAY(J)
LR
LRT
LYRDPH(IROW)
NK
NL
NYTTYM
PIXCON
POPFAC
PSISFC
PSIS(L,K)
RCH2O
ROOTCN
ROOTN
RTIMPD(L,KM1)
RTIMPD(L,KP1)
RTIMPD(LP1,K)
RTP1
RTP2
RTWT(L,K,1)
RTWT(L,K,2)
RTWT(L,K,3)
SLF
THRLN
THTR(L)
TSOILD(L)
TSOILN(L)

UPF(L,K)
VH2OC(L,K)
VNH4C(L,K)
VNO3C(L,K)
WCELL
ZUPT(L,K)

OUTPUTS

PSIAVG
ROOTN
RTWT(L,K,1)
RTWT(L,K,2)
RTWT(L,K,3)
SPDWRT

GENERAL PSEUDOCODE

If this is the first call to RUTGRO for the day, then

Calculate the potential root growth in each soil cell containing roots based upon the following:

1. root weight capable of growth (comprising roots in age class 1 (less than 5 days old) and class 2 (5–15 days old)
2. root temperatures during the day and night
3. specific root growth rate within the soil layer
4. potential change in root weight as affected by nitrogen status of the soil cell and soil impedance to root expansion

Add together the potential change in root weight in each cell for the total soil profile

Else—RUTGRO has already been called once for the day—calculate, adjust, and distribute carbohydrates for root growth

Calculate the root growth correction factor (that is, the ratio of root carbohydrate supply to total potential root growth); the root weight distribution in each age class; and the actual root growth in each soil cell

Find the direction or directions of root growth as determined by soil impedance to root growth, soil water potential, and geotropism within the soil cell itself, in the cell to the left, in the cell to the right, and in the cell below

Calculate the sloughing factor

determine the total weight of sloughed roots and live roots in the soil profile

determine the nitrogen lost through root death

find the amount of PIX (mepiquat chloride, a plant growth regulator) remaining in the plant root

Endif

Calculate the average soil water potential

Sort the root-containing soil cells by a factor calculated as the product of volumet-

ric soil moisture content, soil diffusivity, and plant available water

Determine the average transpiration for the past 5 days

Determine the average soil moisture content of the rooted cells in each horizon

Determine the number of rooted cells required to satisfy the transpiration demand

Determine the average soil moisture content of each horizon

Determine the soil water potential and average soil water potential (PSIAVG) with use of the Marani equation

Return to GROWTH

End RUTGRO

PSEUDOCODE

If this is the first call from GROWTH for the day, then

for all the cells in the profile, do:

initialize to 0.0 the actual increment of root weight for a given cell [DWRT(I,J)]

initialize to 0.0 the potential change in root weight [PDWRT(I,J)]

initialize to 0.0 the weight of roots capable of growth [RTWTCG(I,J)]

end do

initialize to 0.0 the sum of the potential change in root weight over all cells (SPDWRT)

initialize to 0.0 the reduction in the potential change in the root weight due to nitrogen stress (PDWCUT)

for all cells containing roots, add the increment of root weights [RTWTCG(L,K)] for a given cell from age class 1 (less than 5 days old) [RTWT(L,K,1)] and age class 2 (5–15 days old) [RTWT(L,K,2)]

call the subroutine RIMPED to calculate root impedance

for each soil layer with roots, do:

define the intermediate variables for day and night soil temperatures, TSDL and TSNL, respectively

set the upper temperature limit to 30 °C and the lower limit to 13.5 °C (Bar-Yosef and Lambert 1977, fig. 1)

calculate the daytime and nighttime root temperatures, RUTDAY and RUTNYT, respectively

calculate the specific root growth rate within the cell (ROOTXP) as the sum of RUTDAY and RUTNYT

for each soil column with roots, do:

potential root weight increment—if there is abundant photosynthate—[PDWRT(L,K)] is the product of root weight capable of growth and specific root growth rate

*** Reduce potential root growth in cells with low nitrogen content (Landivar, personal communication, 1988). ***

if the volumetric nitrate content of the cell is less than the limit of root growth [calbrt(34) = 0.005] or the soil water potential is less than 1.0 bar, then

set the intermediate variable (FULGRO) to PDWRT(L,K)

adjust the potential root growth toward the nitrogen gradient in the soil

accumulate the reduction in potential growth in the weight of roots due to nitrogen stress (PDWCUT)

else

adjust potential root growth toward the nitrogen gradient in the soil by adding 10 percent of PDWCUT

decrease PDWCUT by 10 percent

endif

*** Reduce potential root growth due to root impedance (Whisler, personal communication, 1983). ***

define the column indices KP1 and KM1

set the intermediate variable (TEST) to maximum root impedance of cell L,K [RTIMPD(L,K)], the cell to its left, the cell to its right, and the cell below

calculate an adjustment factor (RTPCT) to root growth due to root impedance (TEST)

set the lower and upper limits of RTPCT to 0.5 and 1.0, respectively

adjust the potential in root growth as a result of the impedance factor

accumulate potential root growth over the entire profile

end do

end do

Else

*** The second time RUTGRO is called from GROWTH it returns here and distributes the actual (adjusted) carbohydrates available for root growth. ***

if the sum of the potential root growth is greater than 0.0—there are roots with the potential to grow—then

the root-growth correction factor (RGCF) is the ratio of root carbohydrate supply (RCH2O) *over* the total potential root growth for the profile

else

RGCF is set to 0.0

endif

*** For all cells with roots, determine if there is shifting of roots from one age class to another. If there is shifting, update the current values of each age class by the partitioning coefficients for moving root material from one age class to another. ***

for all the cells with roots, do:

if the number of days since emergence is more than 5, then

update the root weight of age class 2 by shifting or adding a fraction of the roots in age class 1

update the root weight of age class 1 by removing the roots added to age class 2

if the number of days since emergence is more than 15, then

update the root weight of age class 3 by shifting or adding a fraction of the roots in age class 2

update the root weight of age class 2 by removing the roots added to age class 3

endif

endif

DWRT(L,K) is the actual root growth after reduction due to carbohydrate stress

end do

*** After determining the growth originating from each cell already occupied by roots, define the direction of that root growth. Growth may occur within the cell itself, to its right (KR1), left (KL1), or downward (LD1). ***

define indices LRT and NLR as the deepest layer with roots (LR) before this day's root growth

for all the layers with roots, do:

define the layer-down coefficient (LDC) as equal to the weighting factor for geotropism (GEOTR)

define the index of the next layer down (LD1)

define the number of columns occupied by roots in layer L (KR)

*** For all the cells with roots, determine if the root weight capable of growth is greater than the threshold weight. If it is, then the roots have traversed the soil cell and can extend into the adjacent cells (Taylor and Gardner 1963). ***

for all cells with roots, do:

if the root dry weight in this cell (ROOTSV) is greater than or equal to the threshold root weight (THRLN), then

*** The effect of KR1 and IRC is to keep roots from spreading more than 1 row spacing, while the effect of KL1 and ILC is to keep roots originating on the right side of the left row from

growing into the left side. The plane under the row is assumed to be impenetrable to roots due to symmetry. ***

define indices KR1 (column to the right) and IRC (right coefficient)

define indices KL1 (column to the left) and ILC (left coefficient)

*** Calculate the root impedance and soil-water-potential weighting factors to determine relative growth from the soil cell in each of the four directions: internally in the cell itself (1), left (L), right (R), and down (D) ***

calculate the terms for root impedance (STR1, STRL, STRR, and STRD) in each of the 4 directions (Whisler, personal communication, 1983)

set the lower and upper limits of the root impedance terms to 0.0 and 1.0, respectively

calculate the sum of soil-water-potential weighting factors (SWFAC), where each factor is the inverse of the cube of the soil water potential of the cell and its neighbors

calculate the soil-water-potential weighting factor (WEFAC1, WEFACL, WEFACR, and WEFACD) in each of the 4 directions

calculate the sum of the root impedance weighting factors (SRIMPD)

if SRIMPD is greater than 0.0, then

calculate the root impedance weighting factor (RIMP1, RIMPL, RIMPR, and RIMPD) in each of the 4 directions

else

set the values of the 4 root impedance weighting factors to 1.0

endif

the terms reflecting the interaction of soil water potential and root impedance (EFACl, EFACL, EFACR, and EFACD) are the products of water potential and root impedance weighting factors

calculate the sum of the interaction terms for root impedance and soil water potential (SRWP)

added to the current young roots in the appropriate cell is the following fraction of root growth:

within the cell [RTWT(L,K,1]

in the right-hand cell [RTWT(L,KL1,1)] (**Note**: If K = l and ILC = 0, the boundary condition of no growth across the plane under the row is satisfied.)

in the left-hand cell [RTWT(L,KR1,1)] (**Note**: If K = NK and IRC = 0, the boundary condition of no growth across the

plane under the next row is satisfied.)

into the cell below [RTWT(LD1,K,1)] (**Note**: If L = NL and LDC = 0, the boundary condition of no growth beyond the bottom of the root zone is satisfied.)

*** Traverse the matrix from left to right by layer. ***

if the current cell is the rightmost cell containing roots in the layer and if the number of columns occupied by roots is less than the total number of columns, increment the number of columns occupied by roots in the layer [KRL(L)] (**Note**: This occurs only when growth in the rightmost cell of the layer containing roots is being considered and current root weight capable of growth exceeds the threshold value.)

if the current cell is not in the bottom layer occupied by roots or all layers in the slab are already occupied by roots, go to *[A]* below. [**Note**: There is no need to increase the number of layers occupied by roots (LR) since that number is already defined.]

*** Growth from the lowest layer occupied by roots downward increases the number of layers occupied by roots. It must be possible to increment LR only once across the layer. ***

if the current column is the first one, increment the temporary number of layers occupied by roots (LRT) [**Note**: LR is not incremented until the complete matrix has been traversed so that the comparison (L.NE.LR) can continue accurately.]

increment the number of columns occupied by roots in what will be the lowest occupied layer during the next traverse of the matrix

go to *[A]*

endif

all growth occurs within the cell itself because the threshold limit has not been exceeded

[A] end do

end do

for each of the 5 members of the cultivation time array, do:

if the date of cultivation is not equal to the number of days since the start of the simulation, go to the end of do loop *[B]* (below)

for the soil cells in layer 1 and columns 4–10, do:

for all three age classes of roots, do:

*** The roots up to a depth of 15 cm on both sides of the plants are damaged due to cultivation (needs refinement in the width to be considered). ***

set the root weight to 0.0

end do

end do

[B] end do

set the number of layers occupied by roots (LR) to LR or LRT, depending on whether a new layer has roots growing in it

initialize to 0.0 g the dry weight of all living roots in the profile

initialize to 0.0 g the total weight of sloughed roots

for all cells with roots, do:

> determine the root weight to be sloughed today (WTBSLF)
>
> reduce the root weight in age class 2 by the amount of sloughed roots
>
> accumulate the weight of sloughed roots (WTSLFD)
>
> the total weight of live roots in each soil cell [ROOTSV(L,K)] due to the left row is the sum of all root weight in each of the three age classes
>
> the total weight of live roots in the profile (ROOTS) due to the left row is the sum over all the cells

end do

calculate root weight per plant (ROOTWT) (**Note**: The "2" accounts for both halves of the root system.)

calculate the total weight of sloughed roots per plant (RUTOFF)

update total root nitrogen (ROOTN) by subtracting the amount of nitrogen loss in sloughed roots

update the total nitrogen loss from dead roots (NLOSS)

calculate PIX loss (PIXLOS) in sloughed roots

update the total PIX loss in the plant from dead roots (PIXPLT)

*** Calculate the average water potential. ***

initialize to 0.0 the total soil water potential (PSITOT), the number of cells from which average soil water potential will be calculated (PSINUM), the sum of water in the cells considered to satisfy water demand (SUMH2O), and the number of cells with roots (NUMCWR)

set index J to 0

initialize available nitrogen (AVAILN) to 0.0

for all the rooted cells, do:

> calculate available nitrogen as the sum of the volumetric nitrate-N and ammonium-N over all cells
>
> calculate plant available water to roots capable of uptake (PAWRCU) as the greater of (1) 0.0 and (2) the product of the uptake factor and the difference in volumetric water content of the cell at present and its residual value
>
> increment the number of cells with roots (NUMCWR)
>
> if J is less than or equal to 600 (assuming a maximum of 600 rooted cells), then
>
> > increment J

set the temporary variable [SORT(J,1)] to the difference in the current and residual volumetric water content of the cell

set the temporary variable [SORT(J,2)] as the greater of 0.0 and the water uptake of the cell

set the temporary variable [SORT(J,3)] as the plant available water to roots capable of uptake

define index [INDX(J)] = [(L−1)*NK] + K (**Note**: This is an encoding of the row and column indices of the cell.)

endif

end do

index N is the minimum of 600 and the number of rooted cells

call subroutine SORTIT to sort the array in descending order by PAWRCU values

*** Calculate the running average of EP for the last 5 days. (**Note**: This is done to smooth the sharp fluctuations in daily plant evaporation or transpiration. ***

for the last 4 days, do:

set the value of the evaporation rate from plant leaves [EPAVG(J)] to the previous day's EP value [move each entry to the next location in EPAVG(J)]

end do

set the EPAVG(1) to today's EP [store today's EP in the first location of EPAVG(1)]

initialize to 0.0 the running average of plant transpiration (WEP)

accumulate the 5 EPAVG(J) values

calculate the running average of the 5 EPAVG(J) values

*** Calculate the average volumetric water content of the rooted cells in each soil horizon. ***

for each soil horizon (maximum of 9 horizons), do:

initialize to 0.0 the number of soil cells to be considered [NCELLS(I)] and the sum of the volumetric water content of the cells [TVH2OC(I)]

end do

*** Determine how many cells with roots are needed to satisfy the transpiration demand (WEP). ***

for each of the cells with roots (NUMCWR), do:

calculate the row and column indices of the cell (IROW and ICOL, respectively)

*** Find the horizon index of this cell. ***

if the soil water potential of this cell [PSIS(IROW,ICOL)] is greater than −15 bars (less negative), then

define the current horizon index (ICUR) as that of the soil horizon of the current row

define the intermediate variable ARDX(ICUR) as the inverse exponent of

the Marani equation [ARDRCN(IROW), brought in from subroutine INITIALIZE]

the plant available water of a cell (TWAT) is the water uptake of the cell (m)

accumulate the water uptake to satisfy the plant water demand over all cells (SUMH20)

increment the number of cells from which average soil water potential is calculated (PSINUM) and NCELLS(ICUR)

accumulate the sum of the volumetric water content over all cells [TVH2OC(ICUR)]

if the total water uptake is greater than the 5-day running average of plant evaporation, exit the do loop

endif

end do

for each of the soil horizons, do:

if the number of cells [NCELLS(I)] is more than 0, calculate the average water content of each horizon with roots [AVGH20(I)]

end do

*** Calculate PSI for each layer from the average water content, then calculate the weighted average of PSI values using the Marani equation. ***

initialize average soil water potential (PSIAVG) to 0.0

for each soil horizon, do:

if the number of cells [NCELLS(I)] is more than 0, then

define the intermediate variable (TEMP1G) of the Marani equation

calculate the soil water potential (PSI) using the Marani equation

if PSI is less than −15 bars, set PSI to −15 bars

average soil water potential (PSIAVG) is the sum of the weighted average soil water potentials over all horizons

endif

end do

Endif

Return to SOIL

End RUTGRO

```
      SUBROUTINE RUTGRO
C*********************************************************************
C THIS SUBROUTINE CALCULATES THE GROWTH (IN TERMS OF DRY             *
C MATTER) OF ROOTS IN EACH CELL FOR THE DAY.  FIRST, THE POTENTIAL*
C GROWTH PDWRT FOR THE EXISTING SOIL WATER POTENTIAL (PSIS)          *
C AND TEMPERATURE (TSOILD & TSOILN) IS CALCULATED FOR EACH           *
C SOIL CELL, BASED ON THE WEIGHT OF ROOTS CAPABLE OF GROWTH          *
C IN EACH CELL RTWTCG.  THEN THE ACTUAL GROWTH IS                    *
C DETERMINED, BASED ON THE CARBOHYDRATE SUPPLY FOR ROOT GROWTH       *
C AND THE POTENTIAL GROWTH FOR THE CELL.  THE ACTUAL GROWTH          *
C OCCURRING FOR A GIVEN CELL MAY OCCUR WITHIN THE CELL OR IN         *
C THE CELLS TO THE RIGHT OR LEFT & BELOW.                            *
C    GROWTH IN THE 4 AVAILABLE CELLS IS BASED ON RELATIVE            *
C WATER POTENTIALS OF THE FOUR, WITH A HEAVIER WEIGHTING             *
C GIVEN TO DOWNWARD GROWTH.                                          *
C    THIS SUBROUTINE DRAWS HEAVILY ON THE IDEAS AND THEORIES OF      *
C  M.G. HUCK, USDA-ARS, AUBURN, AL.  THIS IS ESPECIALLY              *
C  AS REGARDS SLOUGHING.  CF. "A MODEL FOR SIMULATING ROOT           *
C  GROWTH AND WATER UPTAKE ," M.G. HUCK, F.W.T. PENNING DE           *
C  VRIES, AND M.G. KEIZER 1975.                               *
C*********************************************************************

      DIMENSION DWRT(40,20),PDWRT(40,20),RTWTCG(40,20)
      DIMENSION NCELLS(9), TVH2OC(9), ARDX(9), AVGH2O(9)
      DIMENSION PAWRCU(40,20), SORT(600,3), INDX(600)

      INCLUDE 'GOSCOM.FOR'
      IF(KALL.NE.1) THEN
         DO 100 I=1,NL
            DO 100 J=1,NK
               DWRT(I,J)=0.
               PDWRT(I,J)=0.
               RTWTCG(I,J)=0.
  100    CONTINUE
         SPDWRT = 0.
         PDWCUT = 0.0
         DO 140 L = 1, LR
            KR = KRL(L)
            DO 140 K = 1, KR
               RTWTCG(L,K) = RTWT(L,K,1) + RTWT(L,K,2)
  140    CONTINUE
         CALL RIMPED
         DO 160 L = 1, LR
            LP1 = L + 1 - (L/NL)
            TSDL = TSOILD(L)
            TSNL = TSOILN(L)
            IF(TSDL.GT.30.)TSDL = 30.
            IF(TSNL.GT.30.)TSNL = 30.
            IF(TSDL.LT.13.5)TSDL = 13.5
            IF(TSNL.LT.13.5)TSNL = 13.5
            RUTDAY =(-0.2120865+0.016079*TSDL)*DAYTYM
            RUTNYT =(-0.2120865+0.016079*TSNL)*NYTTYM
            ROOTXP = RUTDAY + RUTNYT
C
C IF THERE IS ABUNDANT PHOTOSYNTHATE, INCREASE THE ROOT EXPANSION POTENT
C CSTORE IS THE PROPORTION OF THE LEAVES CARBOHYDRATE STORAGE CAPACITY.
C
            KR = KRL(L)
            DO 160 K = 1, KR
               PDWRT(L,K)=RTWTCG(L,K)*ROOTXP
C
C POTENTIAL ROOT GROWTH IS REDUCED IN CELLS WITH LOW NITROGEN CONTENT
```

```
C
               IF(VNO3C(L,K).LT.CALBRT(34).OR.PSIS(L,K).LT.-1.0) THEN
                  FULGRO = PDWRT(L,K)
                  PDWRT(L,K) = PDWRT(L,K) * CALBRT(4)
                  PDWCUT = PDWCUT + (FULGRO - PDWRT(L,K))
               ELSE
                  PDWRT(L,K) = PDWRT(L,K) + PDWCUT*0.10
                  PDWCUT = PDWCUT * 0.90
               ENDIF
C
               KP1 = K + 1 - (K/NK)
               KM1 = K - 1 + (1/K)
               TEST = RTIMPD(L,K)
               IF(TEST.GE.RTIMPD(L,KM1)) TEST = RTIMPD(L,KM1)
               IF(TEST.GE.RTIMPD(L,KP1)) TEST = RTIMPD(L,KP1)
               IF(TEST.GE.RTIMPD(LP1,K)) TEST = RTIMPD(LP1,K)
               RTPCT= (104.6 - 3.53*TEST/1.0216)*.01
               IF(RTPCT.GT.1.0) RTPCT = 1.0
               IF(RTPCT.LT.0.5) RTPCT = 0.5
               PDWRT(L,K) = PDWRT(L,K)*RTPCT
               SPDWRT = SPDWRT + PDWRT(L,K)
  160     CONTINUE
      ELSE
C
C THE SECOND TIME RUTGRO IS CALLED FROM GROWTH IT RETURNS HERE AND
C DISTRIBUTES THE ACTUAL (ADJUSTED) CARBOHYDRATE AVAILABLE FOR
C ROOT GROWTH.
C
          IF(SPDWRT.GT.0.0) THEN
             RGCF = RCH2O / SPDWRT
          ELSE
             RGCF = 0.0
          ENDIF
          DO 200 L = 1, LR
             KR = KRL(L)
             DO 200 K = 1, KR
                IF(KDAY.GT.5) THEN
                   RTWT(L,K,2) = RTWT(L,K,2) + RTP1 * RTWT(L,K,1)
                   RTWT(L,K,1) = RTWT(L,K,1) * (1.-RTP1)
                   IF(KDAY.GT.15) THEN
                      RTWT(L,K,3) = RTWT(L,K,3) + RTP2 * RTWT(L,K,2)
                      RTWT(L,K,2) = RTWT(L,K,2) * (1.-RTP2)
                   ENDIF
                ENDIF
                DWRT(L,K) = RGCF * PDWRT(L,K)
  200     CONTINUE
          LRT = LR
          NLR = LR
          DO 240 L=1,NLR
             LDC = GEOTR
             LD1 = L + 1 - L/NL
             KR  = KRL(L)
             DO 240 K=1,KR
                IF(ROOTSV(L,K).GE.THRLN) THEN
                   KR1 = K + 1 - K/NK
                   KL1 = K - 1 + 1/K
                   IRC = 1 - K/NK
                   ILC = 1 - 1/K
                   STR1 = (104.6 - 3.53*RTIMPD(L,K)/1.0216)*.01
                   IF(STR1.GT.1.)  STR1 = 1.
                   IF(STR1.LT.0.)  STR1 = 0.
                   STRL = (104.6 - 3.53*RTIMPD(L,KL1)/1.0216)*.01
```

```
              IF(STRL.GT.1.)  STRL = 1.
              IF(STRL.LT.0.)  STRL = 0.
              STRR = (104.6 - 3.53*RTIMPD(L,KR1)/1.0216)*.01
              IF(STRR.GT.1.)  STRR = 1.
              IF(STRR.LT.0.)  STRR = 0.
              STRD = (104.6 - 3.53*RTIMPD(LD1,K)/1.0216)*.01
              IF(STRD.GT.1.)  STRD = 1.
              IF(STRD.LT.0.)  STRD = 0.
C
              SWFAC = (1./PSIS(L,K)**3) + (ILC/PSIS(L,KL1)**3) +
     .                (IRC/PSIS(L,KR1)**3) + (LDC/PSIS(LD1,K)**3)
              WEFAC1 = (1.0/PSIS(L,K)   **3)/SWFAC
              WEFACL = (ILC/PSIS(L,KL1)**3)/SWFAC
              WEFACR = (IRC/PSIS(L,KR1)**3)/SWFAC
              WEFACD = (LDC/PSIS(LD1,K)**3)/SWFAC
              SRIMPD = STR1 + STRL + STRR + STRD
              IF(SRIMPD.GT.0.0) THEN
                 RIMP1 = STR1/SRIMPD
                 RIMPL = STRL/SRIMPD
                 RIMPR = STRR/SRIMPD
                 RIMPD = STRD/SRIMPD
              ELSE
                 RIMP1 = 1.
                 RIMPL = 1.
                 RIMPR = 1.
                 RIMPD = 1.
              ENDIF
C
              EFAC1 = WEFAC1 * RIMP1
              EFACL = WEFACL * RIMPL
              EFACR = WEFACR * RIMPR
              EFACD = WEFACD * RIMPD
              SRWP = EFAC1 + EFACL + EFACR + EFACD
C
              RTWT(L,K,1)  =RTWT(L,K,1)  +DWRT(L,K)*(EFAC1/SRWP)
              RTWT(L,KL1,1)=RTWT(L,KL1,1)+DWRT(L,K)*(EFACL/SRWP)
              RTWT(L,KR1,1)=RTWT(L,KR1,1)+DWRT(L,K)*(EFACR/SRWP)
              RTWT(LD1,K,1)=RTWT(LD1,K,1)+DWRT(L,K)*(EFACD/SRWP)
C
              IF(K.EQ.KR.AND.KR.LT.NK) KRL(L)=KRL(L)+1
              IF(L.NE.LR.OR.LR.GE.NL) GO TO 240
              IF(K.EQ.1) LRT = LR + 1
              KRL(L+1) = KRL(L+1) + 1
              GO TO 240
           ENDIF
           RTWT(L,K,1) = RTWT(L,K,1) + DWRT(L,K)
  240   CONTINUE
        DO 280 J=1,5
           IF(KULDAY(J).NE.DAYNUM) GO TO 280
           DO 260 K=4,10
              DO 260 I=1,3
                 RTWT(1,K,I) = 0.0
  260      CONTINUE
  280   CONTINUE
        LR = LRT
        ROOTS = 0.
        WTSLFD = 0.
        DO 300 L = 1, LR
           KR = KRL(L)
           DO 300 K = 1, KR
              WTBSLF = RTWT(L,K,2)
              RTWT(L,K,2) = WTBSLF*(1. - SLF)
```

```
              WTSLFD = WTSLFD + (WTBSLF-RTWT(L,K,2))
              ROOTSV(L,K) = RTWT(L,K,1)+RTWT(L,K,2)+RTWT(L,K,3)
              ROOTS = ROOTS + ROOTSV(L,K)
 300    CONTINUE
        ROOTWT = ROOTS * POPFAC * 2.
        RUTOFF = RUTOFF + (WTSLFD * POPFAC * 2.)
C ADJUST ROOTN AND CALCULATE NITROGEN LOSS FROM DEAD ROOTS
        ROOTN  = ROOTN  - (WTSLFD * POPFAC * 2.) * ROOTCN
        NLOSS  = NLOSS  + (WTSLFD * POPFAC * 2.) * ROOTCN
        PIXLOS = (WTSLFD * POPFAC * 2.) * PIXCON
        PIXPLT = PIXPLT - PIXLOS
C
        PSITOT = 0.
        PSINUM = 0.
        SUMH2O = 0.
        NUMCWR = 0
C
C PAWRCU IS PLANT AVAILABLE WATER TIMES ROOT WEIGHT CAPABLE OF UPTAKE
C NUMCWR IS NUMBER OF CELLS WITH ROOTS
C
        J = 0
        AVAILN = 0.0
        DO 360 L = 1, LR
           KR = KRL(L)
           DO 360 K = 1, KR
              AVAILN = AVAILN + VNO3C(L,K) + VNH4C(L,K)
              PAWRCU(L,K) = AMAX1(0.,VH2OC(L,K)-THTR(L))*UPF(L,K)
              NUMCWR = NUMCWR + 1
              IF (J .LE. 600) THEN
                 J = J + 1
                 SORT(J,1) = VH2OC(L,K)-THTR(L)
                 SORT(J,2) = AMAX1(0.0,ZUPT(L,K))
                 SORT(J,3) = PAWRCU(L,K)
                 INDX(J) = ((L-1)*NK) + K
              ENDIF
  360   CONTINUE

        N = MIN(J,600)

C  ORDER SORT IN DESCENDING ORDER BY PAWRCU VALUES
        CALL SORTIT(N,SORT,INDX)

C
C CALCULATE RUNNING AVERAGE OF EP FOR THE LAST FIVE DAYS. THIS IS_DONE
C TO SMOOTH THE SHARP FLUCTUATIONS IN DAILY EP.
C
C  MOVE ALL VALUES DOWN ONE SLOT, PUT NEW VALUE IN FIRST SLOT, AVERAGE
        DO 1 J =5,2,-1
           EPAVG(J) = EPAVG(J-1)
    1   CONTINUE
        EPAVG(1) = EP
        WEP = 0.
        DO 2 J =1,5
           WEP = WEP + EPAVG(J)
    2   CONTINUE
        WEP = WEP/5.

C  CALCULATE THE AVERAGE VOLUMETRIC WATER CONTENT OF THE ROOTED CELLS IN
C  EACH SOIL HORIZON. ICUR IS THE INDEX OF THE CURRENT HORIZON.
C  DETERMINE HOW MANY CELLS WITH ROOTS WILL TAKE TO SATISFY THE
C  TRANSPIRATION DEMAND (WEP). TWAT IS PAW OF A CELL IN m. SUMH2O IS
C  THE ACCUMULATED TWAT UP TO THAT ITERATION.
```

```
        DO 365, I = 1, 9
           NCELLS(I) = 0
           TVH2OC(I) = 0.0
 365    CONTINUE
C
        DO 350 J = 1, NUMCWR
C           CALCULATE INDICES OF THIS CELL
           IROW = (INDX(J) / NK) + 1
           ICOL = MOD(INDX(J), NK)
C           FIND HORIZON INDEX OF THIS CELL
           IF (PSIS(IROW,ICOL) .GT. -15) THEN
              ICUR = LYRDPH(IROW)
              ARDX(ICUR) = ARDRCN(IROW)
              TWAT = (SORT(J,2)*ACELLDW)/(WCELL*.1*NK)
              SUMH2O = SUMH2O + TWAT
              PSINUM = PSINUM + 1.0
              NCELLS(ICUR) = NCELLS(ICUR) + 1
              TVH2OC(ICUR) = VH2OC(IROW,ICOL) + TVH2OC(ICUR)
              IF(SUMH2O.GT.WEP)GO TO 370
           ENDIF
  350   CONTINUE
  370   CONTINUE

C       CALCULATE AVERAGE WATER CONTENT OF EACH HORIZON WITH ROOTS
        DO 385, I = 1,9
           IF (NCELLS(I) .GT. 0) AVGH2O(I) = TVH2OC(I)/NCELLS(I)
 385    CONTINUE

C       CALCULATE PSI FOR AVERAGE WATER CONTENT, CALCULATE WEIGHTED
C          AVERAGE OF PSI VALUES
        PSIAVG = 0.0
        DO 395, N = 1,9
           IF (NCELLS(N) .GT. 0) THEN
              TEMP1G = (AVGH2O(N)-AIRDR(N))/(FCININ(N)-AIRDR(N))
              PSI= PSISFC * TEMP1G**ARDX(N)
              IF (PSI . LT. -15.0) PSI = -15.0
              PSIAVG = PSIAVG + (PSI*(NCELLS(N)/PSINUM))
           ENDIF
 395    CONTINUE
        PSIPCNT = PSINUM/NUMCWR
     ENDIF

     RETURN
     END
```

GLOSSARY

ACELLDW	Cross-sectional area of a cell in depth and width, DCELL*WCELL (cm^2).
AIRDR(N)	Volumetric water content at "air dry" of soil horizon N (cm^3 H_2O cm^{-3} soil).
ARDRCN(IROW)	Inverse exponent in the Marani equation.
ARDX(ICUR)	ARDRCN(IROW).
AVAILN	Available nitrogen (mg N cm^{-3} soil).
AVGH2O(I)	Average water content of each soil horizon with roots (cm^3 H_2O cm^{-3} soil).
DAYNUM	Number of days of the year (Julian days).
DAYTYM	Daytime fraction of a 24-hour day.
DWRT(I,J)	The actual increment of root weight for a given cell (g cm^{-1} day^{-1}).
EFAC1	A weighting factor reflecting the interaction of soil water potential and root impedance in the cell itself.
EFACD	A weighting factor reflecting the interaction of soil water potential and root impedance in the cell below.
EFACL	A weighting factor reflecting the interaction of soil water potential and root impedance in the cell to the left.
EFACR	A weighting factor reflecting the interaction of soil water potential and root impedance in the cell to the right.
EP	Plant evaporation or transpiration rate (mm day^{-1}).
EPAVG(J)	Evaporation rate from plant leaves for each of the 5 days (mm day^{-1}).
FCININ(N)	Soil water potential at field capacity of the layer N (cm^3 H_2O cm^{-3} soil).
FULGRO	PDWRT(L,K).
GEOTR	A weighting factor for geotropism.
ICOL	Column (vertical row) index.
ICUR	Index of the current horizon.
ILC	Coefficient for weighting root growth to the left in response to water potential.
INDX(J)	Index.
IRC	Coefficient for weighting root growth to the right in response to water potential.
IROW	Layer (horizontal row) index.
J	Index for the number of rooted cells.
KALL	A flag that reads 0 if the first call from GROWTH and 1 if the second call.
KDAY	Number of days since emergence (day).
KL1	Column index to the left of the source cell of root growth.

KM1	Column index implying column number minus 1.
KP1	Column index implying column number plus 1.
KR	Column counter for the layer.
KR1	Column index to the right of the source cell of root growth.
KRL(L)	Column index of the last rooted cell in layer L.
KULDAY(J)	Cultivation day.
LD1	Row index below ("layer down") the source cell of root growth.
LDC	Coefficient for weighting root growth downward ("layer down") in response to geotropism.
LP1	Layer index implying column number minus 1.
LR	Deepest layer with roots.
LRT	The deepest layer containing roots before the current day's root growth.
LYRDPH(IROW)	Integer counter for the number of the soil horizon.
N	Index passed to the SORT subroutine.
NCELLS(ICUR)	Number of soil cells considered in this horizon.
NK	Number of vertical columns of soil cells in the soil profile.
NL	Number of layers or horizontal rows of soil cells in the soil profile.
NLOSS	Total nitrogen concentration in sloughed roots (g N).
NLR	Number of layers containing roots.
NUMCWR	Number of cells with roots.
NYTTYM	Nighttime fraction of a 24-hour day.
PAWRCU	Plant available water to roots capable of uptake (mg-cm^{-2} day^{-1}).
PDWCUT	Reduction in potential growth in the root weight from nitrogen stress (g).
PDWRT(I,J)	Potential change in weight of root (g).
PIXCON	PIX concentration (mg acre^{-1} per g of plant dry weight).
PIXLOS	Total PIX concentration of sloughed roots per plant (mg acre^{-1}).
PIXPLT	Total PIX concentration in roots per plant (mg acre^{-1})
POPFAC	Population factor (dm^2 plant^{-1}).
PSIAVG	Average soil water potential (bars, negative value).
PSINUM	Number of cells from which average soil water potential (PSIAVG) is calculated (bars).
PSISFC	Soil water potential at field capacity (bars).

PSIS(L,K)	Soil water potential of layer L and column K (bars).
PSITOT	Total soil water potential (bars).
RCH2O	Root carbohydrate supply per plant (g $plant^{-1}$).
RGCF	Root growth correction factor.
RIMP1	A weighting factor related to the root impedance of the cell itself.
RIMPD	A weighting factor related to the root impedance of the cell below.
RIMPL	A weighting factor related to the root impedance of the cell to the left.
RIMPR	A weighting factor related to the root impedance of the cell to the right.
ROOTCN	Average nitrogen concentration in roots (g N per g roots).
ROOTN	Total root nitrogen (g N).
ROOTS	Dry weight of all living roots in the profile (g).
ROOTSV(L,K)	The array of total dry root weight in each soil cell (g).
ROOTWT	Dry weight of all living roots per plant (g).
ROOTXP	Root growth exponent.
RTIMPD(L,K)	Root impedance of cell (L,K) (kg cm^2).
RTIMPD(L,KM1)	Root impedance of cell (L,KM1) (kg cm^2).
RTIMPD(L,KP1)	Root impedance of cell (L,KP1) (kg cm^2).
RTIMPD(LP1,K)	Root impedance of cell (LP1,K) (kg cm^2).
RTP1	Partitioning coefficient for moving root material from age class 1 to age class 2.
RTP2	Partitioning coefficient for moving root material from age class 2 to age class 3.
RTPCT	Adjustment to root growth from root impedance (%).
RTWTCG(I,J)	Weight of roots capable of growth (g $cell^{-1}$).
RTWT(L,K,1)	Array of root weight by cell for roots less than 5 days old.
RTWT(L,K,2)	Array of root weight by cell for roots 5–15 days old.
RTWT(L,K,3)	Array of root weight by cell for roots older than 15 days old.
RUTDAY	Root temperature during the daytime (°C).
RUTNYT	Root temperature during the nighttime (°C).
RUTOFF	Total weight of sloughed roots per plant (g).
SLF	Sloughing factor; the fraction of young and old roots that are sloughed each day.
SORT(J,1)	Temporary sorting variables.

SORT(J,2)	Temporary sorting variables.
SORT(J,3)	Temporary sorting variables.
SPDWRT	Sum of potential change in the weight of roots over all cells (g).
SRIMPD	Sum of weighting factors attributed to root impedance.
SRWP	Sum of the interaction terms for root impedance and soil water potential.
STR1	A root impedance factor; used to modify root growth in the cell itself.
STRD	A root impedance factor; used to modify root growth below the cell.
STRL	A root impedance factor; used to modify root growth to the left of the cell.
STRR	A root impedance factor; used to modify root growth to the right of the cell.
SUMH2O	Sum of water in the cells considered to satisfy plant water demand (mm day^{-1}).
SWFAC	Sum of the water-potential-related weighting factors.
TEMP1G	(AVGH2O(N)−AIRDR(N))/(FCININ(N)−AIRDR(N)).
TEST	RTIMPD(L,K).
THRLN	Threshold weight limit for those roots reaching the opposite boundaries of the cell where growth originated (g dry matter cm^{-3} soil).
THTR(L)	Residual volumetric water content (cm^3 H_2O cm^{-3} soil).
TSDL	Temperature of the soil layer during the daytime (°C).
TSNL	Temperature of the soil layer during the nighttime (°C).
TSOILD	Average temperature of the layer during the daytime (°C).
TSOILN	Average temperature of the layer during the nighttime (°C).
TVH2OC(I)	Sum of the volumetric water content of the cells (cm^3 H_2O cm^{-3} soil).
TWAT	Plant available water of a cell (m).
UPF(L,K)	Uptake factor of the cell at layer L and column K (mg cm^2 day^{-1}).
VH2OC(L,K)	Volumetric water content of the soil (cm^3 H_2O cm^{-3} soil).
VNH4C(L,K)	Volumetric ammonium-N content of the cell(mg N cm^{-3} soil).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N $soil^{-1}$).
WCELL	Width of a soil cell (cm).
WEFAC1	A weighting factor related to the reciprocal of soil water potential; used to modify root growth in the cell itself.
WEFACD	A weighting factor related to the reciprocal of soil water potential; used to modify root growth to the cell below.

WEFACL	A weighting factor related to the reciprocal of soil water potential; used to modify root growth to the left of the cell.
WEFACR	A weighting factor related to the reciprocal of soil water potential; used to modify root growth to the right of the cell.
WEP	Running average plant transpiration for the last 5 days (mm day^{-1}).
WTBSLF	Weight of roots to be sloughed (g).
WTSLFD	Total weight of sloughed roots (g).
ZUPT(L,K)	Water uptake of the cell (cm^3 H_2O cm^{-3} soil).

NITROGEN MOVEMENT AND TRANSFORMATION

At the start of the simulation, the subroutine INITIALIZE performs the following functions:

1. Soils are grouped into one of two sets (SOILNO) based on their water potential at field capacity (PSISFC). Coarse loamy-loam soil has PSISFC greater than −0.2 and a potential mineralizable nitrogen factor of 0.10. Fine loamy-clay soil has PSISFC less than −0.2 and a potential mineralizable factor of 0.06.

2. For the first cell of each layer, the volumetric nitrogen contents—representing ammonium-N (VNH4C), nitrate-N (VNO3C), and organic matter-N (VNC)—are all initially set to 0.0. They are then reset to the residual values read from the initial soil fertility data file. If 0's are read in, they are reset to 4 lb N per 6 inches of layer, or 2.0 and 0.2 lb N acre^{-1} for residual nitrate-N and residual ammonium-N, respectively.

   For the first cell of each layer, the potential mineralizable nitrogen in organic matter (POTN) is calculated as a function of depth. It is assumed there is about 5 percent nitrogen in organic matter, 20 percent of which is mineralizable.

   In the first cell of each layer, the volumetric organic nitrogen content (VNC) is determined as a function of the percentage of organic matter content (OMA), the bulk density of each soil layer (BDL), POTN, and SOILNO.

3. The VNH4C, VNO3C, and VNC of all the remaining cells are initialized to the corresponding value of the first cell for each layer.

4. The total ammonium-N ($TNNH_4$) and nitrate-N content ($TNNO_3$) of all the cells is added and converted from mg N cm^{-3} soil into lb N acre^{-1}.

In subroutine FRTLIZ, fertilizers are added and incorporated into the soil as broadcast, sidedress, or foliar applications. Mineralization of organic matter and urea and nitrification of ammonium are accomplished in NITRIF. Nitrogen is moved in solution through the soil profile by GRAFLO or CAPFLO and is taken up by the plants by mass flow in UPTAKE.

SUBROUTINE FRTLIZ

Revised by S.M. Bridges
Review chaired by G. Stevens

FRTLIZ is called each day and activates on dates that inorganic nitrogen fertilizers are applied. FRTLIZ initializes the ammonium-N and nitrate-N content and the organic matter, by chemical component, of the upper 4 layers of the soil profile. The subroutine is divided into three sections based on the method of application—broadcast, foliar, and sidedress. If a nitrogen application is scheduled for a particular day, the program prints the date and total amount of all forms of nitrogen applied for the day.

Initial conditions in the subroutine include any residual nitrate-N and ammonium-N in the profile and any organic matter plowed down before planting. Inputs on days of fertilizer application include the rate and method of nitrogen applications and, if sidedress application is used, the positions of the band width (horizontal and vertical fertilizer placements) with respect to the row.

MAJOR PROCESSES

Broadcast Applications

- Rates of ammonium-N, nitrate-N, and urea application are converted from lb N acre^{-1} to mg N cm^{-3}. The units are then converted back to lb N acre^{-1}. All broadcast applications are assumed to be incorporated in the top 20 cm of soil.

- The appropriate amount of nitrogen in mg N cm^{-3} is added to each cell in the plow layer. Urea is assumed to be immediately converted to ammonium.

Foliar Applications

- The amount of foliar urea absorbed by leaves is calculated by multiplying the amount applied by the fraction of solar radiation intercepted by the crop canopy. Only urea intercepted by leaves is assumed to be absorbed.

- Intercepted foliar urea is converted to g N plant^{-1} and added to the leaf nitrogen reserves.

- All nitrogen not intercepted by the plants is assumed to go into the first layer of soil cells. Nitrogen added to the soil is converted to lb N acre^{-1} and accumulated in a variable for available soil nitrogen. Nitrogen is added to the appropriate array in mg N cm^{-3}.

Sidedress Applications

- The center of fertilizer application is determined from the vertical and horizontal placements entered into the data file. If possible, the nitrogen fertilizer is spread over 100 cm^{3} of soil volume or a maximum of 4 cells and a minimum of 1 cell.

ASSUMPTIONS

1. Broadcast applications are always incorporated into the top 8 inches or 20 cm of soil. Simulated cotton growth may show immediate relief from nitrogen stress following topdress of nitrogen fertilizer on dry soil. In reality, rain or irrigation is usually required to move this nitrogen into the rooting zone.

2. All of the foliar-applied urea intercepted by the plants is absorbed. Oosterhuis et al. (1990) found that a maximum of 70 percent of the N^{15} painted onto cotton leaves was absorbed in a day. Under drought conditions in which leaves wilted by noon, the leaves were too inactive to absorb foliar nitrogen.

3. Foliar-applied nitrate cannot be absorbed by cotton plants. Potassium nitrate has been used in foliar applications on cotton without causing noticeable leaf burn.

4. In the section of the code concerning sidedress, the row width is assumed to be 1 m, and nitrogen is converted to mg cm^{-3} over the designated 1–4 cells. An error could occur if narrower rows are used.

5. Only inorganic nitrogen fertilizers are accepted. If residues from previous crops are not incorporated several weeks before initial soil samples are collected, their influence on soil nitrate-N and ammonium-N are not considered in the model.

6. Organic matter is well mixed in the upper 4 soil layers.

UNITS

In the subroutine, 1 lb N $acre^{-1}$ = 0.0112186 mg N cm^{-2}.

For broadcast applications, the fertilizer is incorporated to a depth of 20 cm; the factor 0.0112186 mg N cm^{-2} is divided by 20 cm resulting in the conversion factor 0.0005609 mg N cm^{-3}. This factor is valid for all cells in the plow layer only because 20 cm is evenly divisible by cell depth.

For sidedress, the fertilizer is applied to a single point in the soil profile. The product of (1) the application rate in lb N $acre^{-1}$, (2) 0.0112186 mg N cm^{-2}, (3) number of columns (NK), (4) width of the cell (WCELL), and (5) cell thickness (TCELL = 1 cm) converts lb N $acre^{-1}$ to mg N. This amount is distributed throughout the cells receiving sidedress. The constant 1.12 comes from the assumption that rows are 1 m wide.

The factor 0.891 is the inverse of 1.12; it derives from the assumption that rows are 1 m wide.

INPUTS

ACELLDW
DCELL
IDAY
INT
LPLOW
NAPS
NFERT(I,1)
NFERT(I,2)
NFERT(I,3)
NFERT(I,4)
NFERT(I,5)
NFERT(I,6)
NFERT(I,7)
NK
POPPLT
SLEAFN
VNH4C(L,K)
VNO3C(L,K)
WCELL

OUTPUTS

ADDEDN
SLEAFN
VNH4C(L,K)
VNO3C(L,K)

PSEUDOCODE

Initialize total inorganic nitrogen fertilizer (FERN) to 0.0

for each day of nitrogen application, do:

  if today is the date of nitrogen application, then

    add all the inorganic nitrogen applied to the soil (FERN)

    if broadcast application is used, then

    *** Initialize the nitrogen content of the upper 20 cm of the soil profile. ***

      convert the nitrogen fertilizer applied to mg N cm^{-3}, ADDAMM for ammonium-N, ADDNIT for nitrate-N, and ADDURA for urea

      find the total amount of all forms of nitrogen fertilizers applied (ADDEDN)

      for all cells within the 20-cm plow layer, do:

        update the volumetric nitrate-N content [VNO3C(L,K)]

        update the volumetric ammonium-N content [VNH4C(L,K)] for ammonium-N and urea applications

      end do

    endif

  if foliar application is used, then

  *** Only urea and ammonium are intercepted by the plants. The intercepted fertilizer is converted to g $plant^{-1}$ and 70 percent of this amount is added directly to the leaf nitrogen pool. The remaining 30 percent is lost. ***

    calculate the amount of foliar-applied nitrogen intercepted by the leaves (FERINT)

    calculate the amount of foliar-applied nitrogen that falls on the soil (FERSOI)

    calculate the amount of foliar-applied nitrogen intercepted by the plants (FOLIARN)

    update the total amount of nitrogen in the leaves (SLEAFN)

    convert the nitrogen fertilizer applied to mg N cm^{-3}

    find the total amount of all forms of nitrogen fertilizers applied (ADDEDN)

    for all the cells in the first layer of the profile, do:

      update the volumetric nitrate-N content [VNO3C(L,K)] and the volumetric ammonium-N content [VNH4C(L,K)]

    end do

  endif

  if the fertilizers are applied as sidedress, then

    define K and L, which are the horizontal and vertical placements of the fertilizer

*** The nitrogen goes into a volume of 100 cm^3—more than 1 cell can be used. Limits are set so that a maximum of 4 cells and a minimum of 1 cell are used. The column and layer limits are calculated and checks are made to ensure the cells fall into the 40-row by 20-column array. ***

limit the value of K to no less than 1.0

set the intermediate variable K1 to K+1

if K is greater than or equal to the number of columns, then set K to NK and K1 to NK-1

limit the value of L to no less than 1.0

set the intermediate variable L1 to L+1

if L is greater than or equal to the number of layers, then set L to NL and L1 to NL-1

define the number of cells where nitrogen will stay after sidedress is applied (N00)

limit the number of cells from 1 to 4 cells only

calculate the nitrate-N added as sidedress (FERADD)

if FERADD is greater than 0.0—sidedress application has been done—then

    calculate the total amount of nitrogen applied

    update the volumetric nitrate-N content of the cell (L,K)

    update the volumetric nitrate-N content of the cells, depending on the N00 index

endif

calculate the ammonium-N added as sidedress (FERADD)

*** The amount of nitrogen each cell receives is expressed in mg N cm^{-3}. ***

if FERADD is greater than 0.0—sidedress application has been done—then

    calculate the total amount of nitrogen applied

    update the volumetric ammonium-N content of the cell (L,K)

    update the volumetric ammonium-N content of the cells, depending on the N00 index

endif

calculate the urea added as sidedress (FERADD)

if FERADD is greater than 0.0—sidedress application has been done—then

    calculate the total amount of nitrogen applied

    update the volumetric ammonium-N content of the cell (L,K)

    update the volumetric ammonium-N content of the cells, depending on the N00 index

```
            endif
        endif
    endif
End do
Return to SOIL
End FRTLIZ
```

```
      SUBROUTINE FRTLIZ
C*******************************************************************
C SUBROUTINE ADDS FERTILIZER TO PROFILE. MUST BE CALLED AT         *
C PLANTING DATE TO INITIALIZE NITROGEN & ORGANIC MATTER            *
C PROFILE.  MAY BE CALLED FOR SIDE DRESSING.  INPUTS ARE:          *
C FERN:  FERTILIZER INORGANIC NITROGEN, LBS N/ACRE.                *
C FNH4:  FRACTION OF INORGANIC N  IN AMMONIA FORM. 0 TO 1          *
C FNO3:  FRACTION OF INORGANIC N  IN NITRATE FORM. 0 TO 1          *
C DR:  DISTANCE TO RIGHT OF ROW OF BAND OF FERTILIZER, INCHES.     *
C      EQUALS 0 IF BROADCAST.                                      *
C DD:  DISTANCE BELOW SOIL SURFACE OF BAND OF FERTILIZER,          *
C      INCHES.  IGNORED IF DR = 0.                                 *
C OMA:  ORGANIC MATTER PLOWED AT BEGINNING OF SEASON, %,           *
C       MUST BE .GT. 0 TO INITIALIZE N & ORGANIC MATTER ARRAYS.    *
C RNNO3: RESIDUAL N  AS NITRATE IN UPPER 20 CM, LBS/ACRE.          *
C RNNH4: RESIDUAL N AS AMMONIUM IN UPPER 20 CM, LBS/ACRE.          *
C 1 LB/ACRE = .0112186 MG/CM**2   OR  .0005609 MG/20CM**3          *
C*******************************************************************

      INCLUDE 'GOSCOM.FOR'

C OMA .GT. 0. IMPLIES INITIAL FERTILIZATION AT PLANTING DATE &
C PLOWDOWN OF ORGANIC MATTER.

      FERN=0.
      DO 200 I=1,NAPS
         IF(IDAY.EQ.NFERT(I,1)) THEN
            FERN=NFERT(I,2)+NFERT(I,3)+NFERT(I,4)+FERN

C BROADCAST NITROGEN APPLICATION

            IF(NFERT(I,5).EQ.0) THEN
               ADDAMM = NFERT(I,2)*0.0005609
               ADDNIT = NFERT(I,3)*0.0005609
               ADDURA = NFERT(I,4)*0.0005609
               ADDEDN = ADDEDN+((ADDAMM+ADDNIT+ADDURA)*LPLOW*NK)
     .                         *0.891*ACELLDW
               DO 100 L=1,LPLOW
                  DO 100 K=1,NK
                     VNO3C(L,K) = VNO3C(L,K)+ADDNIT
                     VNH4C(L,K) = VNH4C(L,K)+ADDAMM
                     VNH4C(L,K) = VNH4C(L,K)+ADDURA
  100          CONTINUE
            ENDIF

C  FOLIAR APPLICATIONS. FOLIARN IS GRAMS OF FOLIAR N RECEIVED PER PLANT.
C  454 CONVERTS LBS TO GRAMS. 70% OF UREA & AMMONIA INTERCEPTED BY
C  CANOPY IS ADDED TO LEAF N RESERVES. NITRO SUBROUTINE USES FOR PLANT
C  GROWTH.

            IF(NFERT(I,5).EQ.2) THEN
               FERINT = (NFERT(I,2)+NFERT(I,4))*INT
               FERSOI = NFERT(I,2)+NFERT(I,4)-FERINT
               FOLIARN = (FERINT*454.)/POPPLT*.70
               SLEAFN = SLEAFN+FOLIARN
               ADDAMM = FERSOI*.0112186/DCELL
               ADDNIT = NFERT(I,3)*.0112186/DCELL
               ADDEDN = ADDEDN+((ADDAMM+ADDNIT)*NK)*0.891*ACELLDW
               DO 120 K=1,NK
                  VNO3C(1,K) = VNO3C(1,K)+ADDNIT
                  VNH4C(1,K) = VNH4C(1,K)+ADDAMM
  120          CONTINUE
```

```
          ENDIF
C   SIDEDRESS NITROGEN APPLICATION
          IF(NFERT(I,5).EQ.1) THEN
             K = NFERT(I,6) * 2.54 / WCELL + 1.5
             L = NFERT(I,7) * 2.54 / DCELL + 1.5
             IF(K.LE.0) K=1
             K1=K+1
             IF(K.GE.NK) THEN
                K=NK
                K1=NK-1
             ENDIF
             IF(L.LE.0) L=1
             L1=L+1
             IF(L.GE.NL) THEN
                L=NL
                L1=NL-1
             ENDIF
             N00=100./ACELLDW+.5
             IF(N00.LT.1) N00=1
             IF(N00.GT.4) N00=4
             FERADD = NFERT(I,3)*1.12/ACELLDW/N00
             IF(FERADD.GT.0.) THEN
                ADDEDN = ADDEDN + (FERADD*N00) * 0.891 * ACELLDW
                VNO3C(L,K)=VNO3C(L,K) + FERADD
                IF(N00.GE.2)VNO3C(L,K1)=VNO3C(L,K1) + FERADD
                IF(N00.GE.3)VNO3C(L1,K)=VNO3C(L1,K) + FERADD
                IF(N00.EQ.4)VNO3C(L1,K1)=VNO3C(L1,K1) + FERADD
             ENDIF
             FERADD = NFERT(I,2)*1.12/ACELLDW/N00
             IF(FERADD.GT.0.) THEN
                ADDEDN = ADDEDN + (FERADD*N00) * 0.891 * ACELLDW
                VNH4C(L,K)=VNH4C(L,K) + FERADD
                IF(N00.GE.2)VNH4C(L,K1)=VNH4C(L,K1) + FERADD
                IF(N00.GE.3)VNH4C(L1,K)=VNH4C(L1,K) + FERADD
                IF(N00.EQ.4)VNH4C(L1,K1)=VNH4C(L1,K1) + FERADD
             ENDIF
             FERADD = NFERT(I,4)*1.12/ACELLDW/N00
             IF(FERADD.GT.0.) THEN
                ADDEDN = ADDEDN + (FERADD*N00) * 0.891 * ACELLDW
                VNH4C(L,K)=VNH4C(L,K) + FERADD
                IF(N00.GE.2)VNH4C(L,K1)=VNH4C(L,K1) + FERADD
                IF(N00.GE.3)VNH4C(L1,K)=VNH4C(L1,K) + FERADD
                IF(N00.EQ.4)VNH4C(L1,K1)=VNH4C(L1,K1) + FERADD
             ENDIF
          ENDIF
       ENDIF
  200 CONTINUE
      RETURN
      END
```

GLOSSARY

ACELLDW	Area of cell in terms of depth and width (cm^2).
ADDAMM	Amount of ammonium-N fertilizer applied (mg N cm^{-3}).
ADDEDN	Total amount of all forms of nitrogen fertilizers (mg N cm^{-3}).
ADDNIT	Amount of nitrate-N fertilizer applied (mg N cm^{-3}).
ADDURA	Amount of urea nitrogen fertilizer applied (mg N cm^{-3}).
DCELL	Depth of the soil cell (cm).
FERADD	Amount of nitrogen added to a cell as sidedress (mg cm^{-3}).
FERINT	Amount of foliar-applied urea intercepted by the leaves (lb N $acre^{-1}$).
FERN	Amount of inorganic nitrogen fertilizer (lb N $acre^{-1}$).
FERSOI	Amount of foliar-applied urea not intercepted by the cotton plants (lb N $acre^{-1}$).
FOLIARN	Amount of foliar applied urea intercepted by the cotton plants (g N $plant^{-1}$).
LPLOW	Number of cell layers in the 20-cm plow layer.
N00	Number of cells on which to apply sidedress nitrogen.
NAPS	Number of nitrogen fertilizer applications.
NFERT(I,1)	Date of nitrogen fertilizer application.
NFERT(I,2)	Amount of ammonium-N fertilizer applied (lb N $acre^{-1}$).
NFERT(I,3)	Amount of nitrate-N fertilizer applied (lb N $acre^{-1}$).
NFERT(I,4)	Amount of urea nitrogen fertilizer applied (lb N $acre^{-1}$).
NFERT(I,5)	Application method for nitrogen fertilizer (0 = broadcast, 1 = sidedress, 2 = foliar).
NFERT(I,6)	Vertical placement of fertilizer sidedressing (inches).
NFERT(I,7)	Horizontal placement of fertilizer sidedressing (inches).
SLEAFN	Total supply of leaf nitrogen (g N $plant^{-1}$).
VNH4C(L,K)	Volumetric ammonium-N content of the cell (mg N cm^{-3} soil).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N cm^{-3} soil).
WCELL	Width of a soil cell (cm).

SUBROUTINE NITRIF

Revised by S.B. Turner
Review chaired by M.Y.L. Boone and J. Varco

Nitrogen is assumed to occur within the profile in one of three forms: organic, ammonium, or nitrate. This subroutine, called once a day, calculates the amount of ammonium released by the breakdown of organic matter and the amount of ammonium converted to nitrate in each cell of the soil profile. Both processes depend on temperature and soil water content.

One question this subroutine does not answer is how much nitrogen is available from the crop residues of a fallow field.

ASSUMPTIONS

1. Mineralization varies with depth (Stanford and Smith 1972).
2. Of the 5 percent nitrogen available in organic matter, 20 percent of that is mineralizable.
3. There is no organic matter below 1 m in the soil profile.

INPUTS

ACELLDW
BETAK(L)
CONSK(L)
NK
NL
ORGN
PSIS(L)
TSOLAV(L)
VNC(L,K)
VNH4C(L,K)
VNO3C(L,K)

OUTPUTS

VNH4C(L,K)
VNO3C(L,K)

GENERAL PSEUDOCODE

For the upper half layer of the soil profile (NLH), calculate:

> The temperature factor affecting mineralization (TFMIN) [**Note**: TFMIN is based on Q_{10} of 2 (rate doubling at every 10 °C rise in temperature) as proposed by Briones (1988).]
>
> Temperature factor affecting nitrification (TFNIT)
>
> For all the columns (NK) of each layer, calculate:
>
>> Mineralized organic matter and urea (WFMIN) as a function of TFMIN and a logarithmic decay function relating soil water potential (PSIS) to mineralization
>>
>> Nitrified ammonium (WFNIT) as a function of TFNIT and a logarithmic decay function relating soil water potential (PSIS) to nitrification
>>
>> Updated values of volumetric organic nitrogen content (VNC), volumetric nitrogen content as ammonium-N (VNH4C), volumetric nitrogen content as

nitrate-N (VNO3C), and amount of mineralized organic nitrogen (ORGN)

End do

End do

Return to SOIL

End NITRIF

PSEUDOCODE

Set index (NLH) to half the number of the layer

For each layer in the left half of the soil profile, do:

set the intermediate variable (TT) to soil temperature [TSOLAV(L)]

calculate the temperature-dependent factors affecting mineralization (TFMIN) and nitrification (TFNIT) rates

for all the columns (NK) of each layer, do:

*** Begin mineralization of organic matter and urea. ***

depending on the value of the soil water potential of the cell L,K [PSIS(L,K)], calculate the soil-water-dependent factor affecting the mineralization rate (WFMIN) as a logarithmic decay function relating soil water potential to the mineralization rate

set the lower and upper limits of WFMIN to 0.0 and 1.0, respectively

the product of TFMIN and WFMIN defines the combined water- and temperature-dependent mineralization rate factor for organic matter and urea (DMINF)

set the lower and upper limits of DMINF to 0.0 and 1.0, respectively

calculate the amount of mineralized nitrogen (DNMIN)

update the values of the volumetric organic nitrogen content [VNC(L,K)] and the volumetric ammonium-N content of the cell [VNH4C(L,K)]

update the value of mineralized organic nitrogen (ORGN)

*** Mineralization process ends. ***

*** Begin nitrification of ammonium. ***

depending on the value of the soil water potential of the cell L,K [PSIS(L,K)], calculate the soil-water-dependent factor affecting the nitrification rate (WFNIT) as a logarithmic decay function relating soil water potential to the nitrification rate

set the lower and upper limits of WFNIT to 0.0 and 1.0, respectively

the combined water- and temperature-dependent nitrification rate factor for ammonium is the product of TFNIT, WFNIT, and a calibration factor (based on 1990 field observation of F.D. Whisler and J. Varco)

set the lower and upper limits of DNITF to 0.0 and 1.0, respectively

calculate the amount of nitrified ammonium-N (DNIT)

update the volumetric ammonium-N content of the cell [VNH4C(L,K)]

update the volumetric nitrate-N content of the cell [VNO3C(L,K)]

*** Nitrification process ends. ***

end do

End do

Return to SOIL

End NITRIF

```
      SUBROUTINE NITRIF
C   ***************************************************************
C   *                                                             *
C   *   SIMPLIFIED VERSION BASED ON KAFKAFI, BAR-YOSEF, AND HADAS*
C   *   1978. SOIL SCI 125:261-268.                               *
C   *                                                             *
C   ***************************************************************
C
      INCLUDE 'GOSCOM.FOR'
C
      NLH = NL/2
      DO 140 L=1,NLH
         M=LYRDPH(L)
         TT = TSOLAV(L)
         TFMIN = CONSK(L) + BETAK(L) * EXP(TT/10.)
         TFNIT = (10.0**12.02) * (10.0**(-3573.0/(TT+273)))
         DO 120 K=1,NK
C
C MINERALIZATION OF ORGANIC MATTER AND UREA.
C
C TFMIN AND WFMIN ARE TEMPERATURE AND WATER DEPENDENT FACTORS AFFECTING
C THE NITROGEN MINERALIZATION RATE. TFMIN WAS DERIVED BY BRIONES (1988).
C WFMIN WAS DERIVED FROM THE DATA OF MILLER AND JOHNSON (1964).
C
            IF(PSIS(L,K).GE.-0.07) THEN
               WFMIN = 0.58 - 0.30*LOG10(-100*PSIS(L,K))
            ELSE
               IF(PSIS(L,K).GT.-1.0) THEN
                  WFMIN = -0.7 + 1.25*LOG10(-100*PSIS(L,K))
               ELSE
                  WFMIN = 1.5 - 0.266*LOG10(-100*PSIS(L,K))
               ENDIF
            ENDIF
            IF(WFMIN.GT.1.0)WFMIN = 1.0
            IF(WFMIN.LT.0.0)WFMIN = 0.0
C
            DMINF = TFMIN * WFMIN
            IF(DMINF.GT.1.0) DMINF = 1.0
            IF(DMINF.LT.0.0) DMINF = 0.0
C
            DNMIN = VNC(L,K) * DMINF
            VNC(L,K) = VNC(L,K) - DNMIN
            VNH4C(L,K) = VNH4C(L,K) + DNMIN
            ORGN = ORGN + (DNMIN*0.891*ACELLDW)
C
C NITRIFICATION OF NH4.
C
C TFNIT AND WFNIT ARE TEMPERATURE AND WATER DEPENDENT FACTORS AFFECTING
C THE NITRIFICATION RATE. TFNIT COMES FROM KAFKAFI, BAR-YOSEF, AND HADAS
(1978).  WFNIT C  WAS DERIVED FROM THE DATA OF SABEY (1969).
C
            IF(PSIS(L,K).GT.-0.1) THEN
               WFNIT = -0.80 + 1.33*LOG10(-100*PSIS(L,K))
            ELSE
               WFNIT = 1.25 - 0.33*LOG10(-100*PSIS(L,K))
            ENDIF

            IF(WFNIT.GT.1.0)WFNIT = 1.0
            IF(WFNIT.LT.0.0)WFNIT = 0.0

            DNITF = TFNIT * WFNIT * 0.05 * CALBRT(35)
C 15. IN THE ABOVE EQUATION COMES FROM J.VARCO & FDW FIELD OBS.
```

```
         IF(DNITF.GT.1.0) DNITF = 1.0
         IF(DNITF.LT.0.0) DNITF = 0.0
         DNIT = VNH4C(L,K) * DNITF
         VNH4C(L,K) = VNH4C(L,K) - DNIT
         IF(VNH4C(L,K).LE.0.000001)VNH4C(L,K)=0.0
         VNO3C(L,K) = VNO3C(L,K) + DNIT
120    CONTINUE
140 CONTINUE
    RETURN
    END
```

GLOSSARY

ACELLDW	Area of a soil cell in depth and width (cm^2).
BETAK(L)	Coefficients of the temperature-dependent equation affecting the mineralization rate; values are defined in BLOCK data.
CONSK(L)	Intercepts of the temperature-dependent equation affecting the mineralization rate; values are defined in BLOCK data.
DMINF	Mineralization rate factor, defined as TFMIN*WFMIN.
DNIT	Nitrification of ammonium (mg N cm^{-3}).
DNITF	Nitrification rate factor, defined as TFNIT*WFNIT.
DNMIN	Amount of mineralized organic nitrogen (mg N cm^{-3}).
LYRDPH(L)	Integer counter for the soil horizon number (L) of the soil profile.
NK	Number of vertical columns of soil cells in the soil profile.
NL	Number of layers or horizontal rows of soil cells in the soil profile.
NLH	NL*1/2.
ORGN	Amount of mineralized organic nitrogen (lb N $acre^{-1}$).
PSIS(L,K)	Soil water potential of layer (L) and column (K) (bars).
TFMIN	Temperature-dependent factor affecting the mineralization rate.
TFNIT	Temperature-dependent factor affecting the nitrification rate.
TSOLAV(L)	Average soil temperature (°C).
VNC(L,K)	Volumetric organic nitrogen content of the cell (mg N cm^{-3}).
VNH4C(L,K)	Volumetric ammonium-N content of the cell (mg N cm^{-3}).
VNO3C(L,K)	Volumetric nitrate-N content of the cell (mg N cm^{-3}).
WFMIN	Soil-water-dependent factor affecting the mineralization rate.
WFNIT	Soil-water-dependent factor affecting the nitrification rate.

APPENDIX A

OTHER RELATED SOURCE CODES

SUBROUTINE GBLOCK

```
SUBROUTINE GBLOCK

CC **SIMULATED BLOCK DATA FOR GOSSYM **** DATE NOVEMBER 9, 1989 **
CC **CONVERTED BY WENDELL LADNER AND SUSAN BRIDGES FOR C **

      INCLUDE 'GOSCOM.FOR'
C
C      DATA
C     &     ABEND/.FALSE./,ABZ/0./,ADDEDN/0./,
C     &     AGE/450*0./,AGEABZ/450*0.0/,AGEBOL/450*0.0/,AGETOP/0./,
C     &     AGEPFN/9*0./,AIRDR/9*0./,
C     &     AIRDRI/0./,AIRDRW/0./,ALPHA/3.5/,APDAY/365*0/,
C     &     AREA/0./,AT/.2/,AVGT/450*0./,
C     &     AVTEMP/20./,AVTPFN/9*0./

      ABEND=.FALSE.
      ABZ=0.
      ADDEDN=0.
      AGETOP=0.
      AIRDRI=0.
      AIRDRW=0.
      ALPHA=3.5
      APRES=0.0
      AREA=0.
      AT=.2
      AVAILN=0.
      AVGTSD=0.
      AVGTSP=0.
      AVTEMP=20.

C      DATA BD/9*0./,BDL/40*1./,BDI/0./,BDRATO/0./,BDSLOP/0./,BDW/0./,
C     &     BETA/9*0./,BETAI/0./,BETAW/0./,BOLL1/0./,
C     &     BETAK/3*0.000800,3*0.000602,3*0.000457,11*0.000308/,
C     &     BOLOSS/366*0./,BOLTMP/450*0./,BOLWGT/450*0./,BURCN/0./,
C     &     BURMIN/0./,BURR1/0./,BURRN/0./,BSIZE/450*0.0/,BLUM/366*0./

      BDI=0.
      BDRATO=0.
      BDSLOP=0.
      BDW=0.
      BETAI=0.
      BETAW=0.
      BOLL1=0.
      BURCN=0.
      BURMIN=0.
      BURR1=0.
      BURRN=0.

C      DATA CD/0.0/,CDBOLL/0./,CDLEAF/0./,CDROOT/0./,
C     &     CDSQAR/0./,CDSTEM/0./,
C     &     CLIMAT/2555*0./,CO2PARM/1.0235,1.0264,1.0285,
C     &     1.0321,1.0335,1.0353,1.0385,1.0403,1.0431,1.0485,1.0538,
C     &     1.0595,1.0627,1.0663,1.0716,1.0752,1.0784,1.0823,1.0880,
C     &     1.0923,1.0968,1.1019,1.1087,1.1172,1.1208,1.1243,1.1311,
C     &     1.1379/,COTXX/0./,
C     &     CONSK/3*0.005417,3*0.004038,3*0.002913,11*0.003214/,
C     &     CPOOL/0./,CSTRES/1./,CSTORE/1.0/,
C     &     CUMEP/0./,CUMES/0./,CUMRAN/0./,CUMSOK/0./

      CD=0.0
      CDBOLL=0.
      CDLEAF=0.
```

```
      CDROOT=0.
      CDSQAR=0.
      CDSTEM=0.
      CHARI = 'I'
      CMXIRR = 0.
      COTXX=0.
      CO2 = 0
      CPOOL=0.
      CSTRES=1.
      CSTORE=1.0
      CUMEP=0.
      CUMES=0.
      CUMRAN=0.
      CUMSOK=0.

C      DATA D/5./,
C     &     DAY1PN/0./,DAY1SN/0./,DAYINC/1./,DEFDAY/0/,DEFPPA/0./,
C     &     DAYLNG/13./,DAYNUM/1/,DAYWTF/0./,DAZE/0./,
C     &     DEHISS/450*45./,DELAY/90*0./,DIFF/800*258.3/,DIFF0/9*0./,
C     &     DIFFOI/0./,DIFFOW/0./,DTAVG/7*20./,DZ/0.0/,DZDAY/10*0/,
C     &     DZOBS/10*0./

      DCELL=5.
      DAY1PN=0.
      DAY1SN=0.
      DEFBGN=0
      DEFDAY=0
      DEFMTH=0
      DEFPPA=0.
      DAYLNG=13.
      DAYNUM=1
      DAYWTF=0.
      DAZE=0.
      DIFFOI=0.
      DIFFOW=0.
      DZ=0.0

C      DATA EMERGE/0./,EP/0./,EPAVG/5*.15/,ES/0./

      EMERGE=0.
      EP=0.
      ES=0.

C      DATA F2/.5/,FBLOOM/0./,FC/40*.267/,FCININ/9*.0/,FCINIC/.4/,
C     &    FCFCTI/0./,FCINIW/.2/,FCODE/450*0./,FFRUT/450*0./,FLOSS/0.0/,
C     &    FLNMIN/1.0E-6/,FLXMAX/9*0./,FLXMIN/9*0./,FNH4/0./,
C     &    FNL/840*0./,FNO3/1./,FNU/820*0./,FRATIO/0./,FRUTP/450*0.0/,
C     &    FSQ/0./,FSTRES/0./,FSTAVG/450*0.5/,FWU/410*0.0/

      F2=.5
      FBLOOM=0.
      FCINIC=.4
      FCFCTI=0.
      FCINIW=.2
      FERN=0.
      FILFRM='formatted'
      IF((OPSYS.EQ.'dos').OR.(OPSYS.EQ.'DOS')) FILFRM='binary'
      FL = 0.
      FLOSS=0.0
      FLNMIN=1.0E-6
      FNH4=0.
      FNO3=1.
```

```
      FRATIO=0.
      FS = 0.
      FSQ=0.
      FSTRES=0.

C      DATA GEOTR/10./,GAMMA/0.653/,GBLOS/0./,GBOLWT/0./,GBZ2/0./,
C     &     GH2OC/9*0./,GINP/0.0/

      GEOTR=10.
      GAMMA=0.653
      GBLOS=0.
      GBOLWT=0.
      GBZ2=0.
      GIN=0.
      GINP=0.0
      GSUBR=.375

C      DATA H2OINT/14*100./

C      DATA IDAY/1/,INRIM/0/,INT/0/,ISQ/0/,IPIX/1/,IDZ/1/

      IDAY=0
      IFGIRR=0
      IFGRAIN=0
      INRIM=0
      INT=0
      IPIX=1
      ISCRN=6
      ISQ=0


C      DATA KRAIN/0/,KRL/2,1,1,1,1,35*0/,KULDAY/5*0/,KWIDTH/0/

      KRAIN=0
      KWIDTH=0

C      DATA JDAY/1/

      JDAY=1

C      DATA LAGE/450*0./,LAI/0.001/,LAMDAC/0.23/,LAMDAS/0.10/,
C     &     LAREA/450*0.04/,LATUDE/35/,
C     &     LDEPTH/9*0/,LEAFCN/.037/,LEAFR1/0/,
C     &     LEAFRS/0/,LEAFW/450*0/,LEAFWT/.2/,LEFABS/0/,
C     &     LOSSQR/366*0/,LR/5/,LTYPE/0/,LYTRES/0./


      LAI=0.001
      LAMDAC=0.23
      LAMDAS=0.10
      LATUDE=35
      LDAYAW=0
      LDAYIR=0
      LDAYFW=0
      LDAYPW=0
      LEAFCN=.037
      LEAFR1=0
      LEAFRS=0
      LEAFWT=.2
      LEFABS=0
      LEFCNT=0
```

```
      LINE=51
      LMAX=0
      LR=5
      LTYPE=0
      LVSLOS=0
      LYTRES=0.

C      DATA MATURE/450*0/,MCODE/450*0/,MH2O/0/,MLAREA/90*.04/,
C     &     MLEAFW/90*0./,MMUPN1/0./,MMUPN2/0./,MMUPN3/0./,MO/1/


      MH2O=0
      MMUPN1=0.
      MMUPN2=0.
      MMUPN3=0.
      MO=1

C      DATA NAPS/0/,NDLAY/0./,
C     &     NEWEP/0.0/,NEWES/0/,NF/0/,NFBR/3*0/,NFERT/1825*0/,
C     &     NFRQ/1/,NFRQX/0/,NK/20/,NL/40/,NLOSS/0./,NNOD/90*0/,
C     &     NOITR/5/,NOPEN/0./,NPN/0/,NPT/0/,NPOOL/0/,
C     &     NPP/0/,NPR/0/,NPW/0/, NUMPFN/1/,NR/0./,NV/0./,NVBRCH/1/,
C     &     NYTTYM/0./,NYTWTF/0/

      NAPS=0
      NDLAY=0.
      NEWEP=0.0
      NEWES=0
      NF=0
      NFBR(1)=0
      NFBR(2)=0
      NFBR(3)=0
      NFRQ=1
      NK=20
      NL=40
      NLOSS=0.
      NOITR=5
      NOPEN=0.
      NPOOL=0
      NUMPFN=1
      NR=0.
      NV=0.
      NVBRCH=1
      NYTTYM=0.
      NYTWTF=0

C      DATA OMA/1.,13*0./,ORGN/0./

      ORGN=0.

C      DATA PDADAY/450*0./,PDAMLD/90*0./,PDAMLN/90*0./,PDANYT/450*0./,
C     &     PDBOLL/0./,PDLEAF/0.0001/,PDROOT/.0001/,PDSQ/0./,
C     &     PDSTEM/.0001/,PDWBOD/450*0./,PDWBON/450*0./,PDWFLD/450*0./,
C     &     PDWFLN/450*0./,PDWMLD/90*0./,PDWMLN/90*0./,PDWSQ/450*0./,
C     &     PLANTN/0./,PFAL/9*0.04/,PFAREA/0./,
C     &     PFDAL/0./,PFDALD/9*0./,PFDALN/9*0./,PFDWL/0./,PFDWLD/9*0./,
C     &     PFDWLN/9*0./,PFWL/9*0./,PIN/0.0/,PLANTW/0./,
C     &     PIXCON/0./,PIXDAY/10*999/,PIXLOS/0./,PIXPLT/0./,
C     &     PIXPPA/10*0./,PIXDA/1./,PIXDN/1./,PIXDZ/1./,
C     &     PLTN/0./,PN/0./,POLYNA/0/,POPFAC/0./,
C     &     POPPLT/41000./,PQFLR/0./,PSIAVG/-.175/,PSILD/-0.8/,
C     &     PSILN/-0.8/,PSIMAX/0.0/,PSINUM/0./,PSIS/800*-.175/,
```

```
C     &     PTSRED/1./,PUPF/861*0./

      PDBOLL = 0.
      PDLEAF = 0.0001
      PDROOT = .0001
      PDSQ = 0.
      PDSTEM = .0001
      PI = 3.14159
      PLANTN = 0.
      PFAREA = 0.
      PFDAL = 0.
      PFDWL = 0.
      PIN = 0.0
      PLANTW = 0.
      PLEFABS = 0.
      PIXCON = 0.
      PIXLOS = 0.
      PIXPLT = 0.
      PIXDA = 1.
      PIXDN = 1.
      PIXDZ = 1.
      PLTN = 0.
      PN = 0.
      POLYNA = 0
      POPFAC = 0.
      POPPLT = 41000.
      PQFLR = 0.
      PRPDAY = 0
      PRPKGH = 0.
      PRPPPA = 0
      PSIAVG = -.175
      PSICMX = -0.5
      PSILD = -0.8
      PSILN = -0.8
      PSIMAX = 0.0
      PSINUM =  0.
      PTSRED = 1.

C      DATA RAIN/0./,RCH2O/.0002/,REQ1/0./,RESC/.06/,
C     &     RESN/0./,RI/0./,RN/0./,RNNH4/50.,13*0./,RNNO3/10.,13*0./,
C     &     ROOTCN/.037/,ROOTN/.00450/,ROOTR1/0./,ROOTRS/0./,ROOTS/0./,
C     &     ROOTSV/800*0./,ROOTWT/.200/,ROWSP/101.6/,RTEXNT/40*.FALSE./,
C     &     RTIMPD/800*0./,RTP1/.3/,RTP2/.1/,RTWT/2400*0./,
C     &     RTWTCU/800*0./,RUTOFF/0./

      RAIN = 0.
      RCH2O = .0002
      REQ1 = 0.
      RESC = .06
      RESN = 0.
      RI = 0.
      RN = 0.
      ROOTCN = .037
      ROOTN = .00450
      ROOTR1 = 0.
      ROOTRS = 0.
      ROOTS = 0.
      ROOTWT = .200
      ROWSP = 101.6
      RSUBO = .0032
      RTP1 = .3
      RTP2 = .1
```

```
      RUTOFF = 0.
C       DATA SDWBOL/0./, SDWLEF/0./,
C      &     SDWSQR/0./,SDWSTM/0./,SEASON/0./,SEEDCN/0./,
C      &     SEEDN/0./,SEEDR1/0./,SEND/.FALSE./,SESI/0./,SESII/0./,
C      &     SITES/0./,SITEZ/0./,SKPFLG/.FALSE./,SLEAFN/.0074/,SLF/.02/,
C      &     SOAKN/820*0./,SOAKW/41*0./,
C      &     SPDWBO/0./,SPDWLD/0./,SPDWLN/0./,
C      &     SPDWRT/0./,SPDWSQ/0./,SPN/0./,SQLOSS/366*0./,
C      &     SQRWT/450*0./,SQRZ/0./,SQWT/0./,STEMCN/.037/,STEMN/.0074/,
C      &     STEMRS/0./,STEMWT/.2/,STMWT/366*0./,
C      &     SUMEP/0.0/,SUMES/0.0/,SUMSTRS/0.0/,SUPNO3/0./,SUPNH4/0./,
C      &     SUMSUB/0./,SUBIRR/0./,SUPF/0./

        SBOLL = 0.
        SDWBOL = 0.
        SDWLEF = 0.
        SDWSQR = 0.
        SDWSTM = 0.
        SEEDCN = 0.
        SEEDN = 0.
        SEEDR1 = 0.
        SEND = .FALSE.
        SESI = 0.
        SESII = 0.
        SITES = 0.
        SITEZ = 0.
        SKPFLG =  .FALSE.
        SLEAF = 0.
        SLEAFN = .0074
        SLF = .02
        SPDWBO = 0.
        SPDWLD = 0.
        SPDWLN = 0.
        SPDWRT = 0.
        SPDWSQ = 0.
        SPN = 0.
        SQUAR = 0.
        SQRZ = 0.
        SQWT = 0.
        SROOT = 0.
        SSTEM = 0.
        STEMCN = .037
        STEMN  = .0074
        STEMRS = 0.
        STEMWT = .2
        SUMEP = 0.0
        SUMES = 0.0
        SUMSTRS = 0.0
        SUPNO3 = 0.
        SUPNH4 = 0.
        SUMSUB = 0.
        SUBIRR = 0.
        SUPF = 0.
C       DATA T/0./,TAIR/15*0./,TAVG/0./,TD/0./,
C      &     TEMP1C/0./,TEMP1G/0./,TEMP1R/0./,TH2O/0./,
C      &     TDAY/0./,THETA0/9*0./,
C      &     THETAI/0./,THETAR/9*0./,THETAS/9*0./,THRLN/0.3E-3/,
C      &     THTA0I/0./,THTA0W/0./,THTARC/0./,THTARI/0./,THTARW/0./,
C      &     THTASC/0./,THTASI/0./,THTASW/0./,THTS/40*0./,THTR/40*0./,
C      &     THAD/40*0./,TIH2OC/0./,
C      &     TIRRIG/365*0.0/, TSMN/40*25./,TSMX/40*25./,
C      &     TMAX/0./,TMIN/0./,TNNH4/0./,TNNO3/0./,TNYT/0./,
```

```
C     &     TSOILD/40*25./,TSOILN/40*25./,TSOLAV/20*0./,
C     &     TSTBD/360*0./,TSTIMP/360*0./,TUPF/861*.TRUE./,
C     &     TTUPF/861*.TRUE./

      T = 0.
      TAVG = 0.
      TCELL = 1.
      TD = 0.
      TEMP1C = 0.
      TEMP1G = 0.
      TEMP1R = 0.
      TH2O = 0.
      TDAY = 0.
      THETAI = 0.
      THRLN = 0.3E-3
      THTA0I = 0.
      THTA0W = 0.
      THTARC = 0.
      THTARI = 0.
      THTARW = 0.
      THTASC = 0.
      THTASI = 0.
      THTASW = 0.
      TIH2OC = 0.
      TMAX = 0.
      TMIN = 0.
      TNNH4 = 0.
      TNNO3 = 0.
      TNH4UP = 0.
      TNO3UP = 0.
      TNYT = 0.

C      DATA UPNH4/0./,UPNO3/0./,UPF/800*0./

      UPNH4 = 0.
      UPNO3 = 0.

C      DATA VDELAY/3*0./,VH2OC/800*.267/,VNC/800*0./,
C     &     VNH4C/800*0./,VNO3C/800*0./,VSTRES/0./

      V = 0.
      VERSION = 'May 1991'
      VSTRES = 0.
C      DATA W/5./,WATTBL/200./,WATTSM/0./,WIND/88./,
C     &     WSTRSD/1./,WSTRSN/1./,WTSLFD/0./,WSTRS/1./
      WCELL = 5.
      WATTBL = 200.
      WATTSM = 0.
      WIND = 88.
      WSTRSD = 1.
      WSTRSN = 1.
      WTSLFD = 0.
      WSTRS = 1.

C      DATA XTRAC/0./

      XTRAC = 0.
      XTRAN = 0.

C      DATA YIELD/0./,IYEAR/74./
      YIELD = 0.
      IYEAR = 74.
```

```
C      DATA Z/.1/
      Z = .1
      ZPIXD = 0.

CC  LEAF AREA & LEAF WEIGHT INITIALIZED ACCORDING TO COTYLEDON DATA
CC  FOR 'M-8' COTTON OF CHRISTIANSEN, M. N. (1962) A METHOD OF
CC  MEASURING AND EXPRESSING EPIGNEOUS SEEDLING GROWTH RATE.
CC  CROP SCI. 2:487-488.
C      DATA LMAX/0./, PI/3.1416/, C1/0.3964D-0,0.3631D+1,0.3838D-1,
C     .      0.7659D-1,0.0000D+0,-0.2297D+2,-0.3885D+0,-0.1587D-0,
C     .      -0.01021D-1/

C      DATA CO2/0./, RSUBO/.0032/, GSUBR/.375/
C      DATA TNO3UP/0./,TNH4UP/0./,XTRAN/0./

C      DATA SLEAF/0./,SBOLL/0./,SQUAR/0./,SSTEM/0./,SROOT/0./
C      DATA V/0./, GIN/0./, FL/0./, FS/0./

C      DATA KA/' ','0','1','2','3','4','5','6','7','8','9','*'/
C      DATA KHAR/800*' '/
C      DATA CHAR1/'-X','-*','-$','-A','-A','-A','-B'/
C      DATA CHAR2/'X-','*-','$-','A-','A-','A-','B-'/
C      DATA CHARI/'I'/

C      DATA CHAR3/'-1','-2','-3','-4','-5','-6','-7','-8','-9','-0',
C     .           '-A','-$','-X'/
C      DATA CHAR4/'1-','2-','3-','4-','5-','6-','7-','8-','9-','0-',
C     .           'A-','$-','X-'/
C      DATA PRT/450*'  '/,PRI/90*' '/
C      DATA DATAID/20*'    '/,HDCPY/20*'    '/,IDC/20*'    '/,
C     .      IDP/20*'    '/,INTSOL/20*'    '/

C*********************************************************************
C   INITIALIZATION OF ARRAYS IS LISTED IN ASCENDING ORDER BY THE
C   FIRST SUBSCRIPT ON THE ARRAY.  WITHIN AN INITIALIZATION OF
C   ARRAYS OF THE SAME SIZE, THEY ARE LISTED ALPHABETICALLY.
C*********************************************************************

C**********************(3),(3,30),(3,30,5)*************************
      DO 100 I=1,3
         VDELAY(I) = 0.
         DO 100 J=1,30
            DELAY(I,J)=0.
            MLAREA(I,J)=.04
            MLEAFW(I,J)=0.
            NNOD(I,J)=0
            PDAMLD(I,J) = 0.
            PDAMLN(I,J) = 0.
            PDWMLD(I,J) = 0.
            PDWMLN(I,J) = 0
            PRI(I,J) = '  '
            DO 100 K=1,5
               AGE(I,J,K)=0.
               AGEABZ(I,J,K)=0.0
               AGEBOL(I,J,K)=0.0
               AVGT(I,J,K)=0.
               BOLTMP(I,J,K)=0.
               BOLWGT(I,J,K)=0.
               BSIZE(I,J,K)=0.0
               DEHISS(I,J,K)=45.
               FCODE(I,J,K)=0.
```

```
            FFRUT(I,J,K)=0.
            FRUTP(I,J,K)=0.0
            FSTAVG(I,J,K)=0.5
            LAGE(I,J,K)=0.
            LAREA(I,J,K)=0.04
            LEAFW(I,J,K)=0
            MATURE(I,J,K)=0
            MCODE(I,J,K)=0
            PDADAY(I,J,K)=0.
            PDANYT(I,J,K) = 0.
            PDWBOD(I,J,K) = 0.
            PDWBON(I,J,K) = 0.
            PDWFLD(I,J,K) = 0.
            PDWFLN(I,J,K) = 0.
            PDWSQ(I,J,K) = 0.
            SQRWT(I,J,K) = 0.
            PRT(I,J,K) = '  '
 100  CONTINUE

C********************************(5)**********************************
      DO 110 I=1,5
         EPAVG(I)=.15
         KULDAY(I)=0
         FMTHOD(I)='      '
         HMTHOD(I)='      '
         IMTHOD(I)='      '
  110 CONTINUE
      FMTHOD(1)='BDCAST'
      FMTHOD(2)='SDRESS'
      FMTHOD(3)='FOLIAR'
      HMTHOD(1)='BANDED'
      HMTHOD(2)='SPKLER'
      HMTHOD(3)='BDCAST'
      IMTHOD(1)='SPKLER'
      IMTHOD(2)='FURROW'
      IMTHOD(3)=' DRIP '

C********************************(7)**********************************
      DO 120 I=1,7
         DTAVG(I)=20.
         PGRUNT(I)='      '
  120 CONTINUE
      PGRUNT(1)=' pts/a'
      PGRUNT(2)=' gal/a'
      PGRUNT(3)=' ozs/a'
      PGRUNT(4)=' lbs/a'
      PGRUNT(5)=' a/lb '
      PGRUNT(6)=' a/gal'

      CHAR1(1) = '-X'
      CHAR1(2) = '-*'
      CHAR1(3) = '-$'
      CHAR1(4) = '-A'
      CHAR1(5) = '-A'
      CHAR1(6) = '-A'
      CHAR1(7) = '-B'

      CHAR2(1) = 'X-'
      CHAR2(2) = '*-'
      CHAR2(3) = '$-'
```

```
      CHAR2(4) = 'A-'
      CHAR2(5) = 'A-'
      CHAR2(6) = 'A-'
      CHAR2(7) = 'B-'

C**************************(9)(9,40)******************************
       C1(1) = 0.3964D-0
       C1(2) = 0.3631D+1
       C1(3) = 0.3838D-1
       C1(4) = 0.7659D-1
       C1(5) = 0.0000D+0
       C1(6) = -0.2297D+2
       C1(7) = -0.3885D+0
       C1(8) = -0.1587D-0
       C1(9) = -0.01021D-1

      DO 130 I=1,9
         AGEPFN(I)=0.
         AIRDR(I)=0.
         AVTPFN(I)=0.
         BD(I)=0.
         BETA(I)=0.
         DIFFO(I)=0.
         FCININ(I)=.0
         FLXMAX(I)=0.
         FLXMIN(I)=0.
         GH2OC(I)=0.
         IPCLAY(I) = 30
         IPSAND(I) = 35
         PFAL(I) = 0.04
         PFDALD(I) = 0.
         PFDALN(I) = 0.
         PFDWLD(I) = 0.
         PFDWLN(I) = 0.
         PFWL(I) = 0.
         LDEPTH(I) = 0
         THETAO(I) = 0.
         THETAR(I) = 0.
         THETAS(I) = 0.
         DO 130 J = 1,40
            TSTBD(I,J) = 0.
            TSTIMP(I,J) = 0.
  130 CONTINUE


C*******************************(10)*********************************
      DO 140 I=1,10
         PIXDAY(I) = 0
         PIXMTH(I) = 0
         PIXPPA(I) = 0.
  140 CONTINUE

C*******************************(12)*********************************

      KA(1) = ' '
      KA(2) = '0'
      KA(3) = '1'
      KA(4) = '2'
      KA(5) = '3'
      KA(6) = '4'
      KA(7) = '5'
      KA(8) = '6'
```

```
      KA(9) = '7'
      KA(10) = '8'
      KA(11) = '9'
      KA(12) = '*'

C********************************(13)*********************************

      CHAR3(1) = '-1'
      CHAR3(2) = '-2'
      CHAR3(3) = '-3'
      CHAR3(4) = '-4'
      CHAR3(5) = '-5'
      CHAR3(6) = '-6'
      CHAR3(7) = '-7'
      CHAR3(8) = '-8'
      CHAR3(9) = '-9'
      CHAR3(10) = '-0'
      CHAR3(11) = '-A'
      CHAR3(12) = '-$'
      CHAR3(13) = '-X'
      CHAR4(1) = '1-'
      CHAR4(2) = '2-'
      CHAR4(3) = '3-'
      CHAR4(4) = '4-'
      CHAR4(5) = '5-'
      CHAR4(6) = '6-'
      CHAR4(7) = '7-'
      CHAR4(8) = '8-'
      CHAR4(9) = '9-'
      CHAR4(10) = '0-'
      CHAR4(11) = 'A-'
      CHAR4(12) = '$-'
      CHAR4(13) = 'X-'

C********************************(14)*********************************
      DO 150 I=1,14
         H2OINT(I)=100.
  150 CONTINUE

      OMA(1)=1.
      RNNH4(1) = 50.
      RNNO3(1) = 10.
      DO 160 I=2,14
         OMA(I)=0.
         RNNH4(I) = 0.
         RNNO3(I) = 0.
  160 CONTINUE


C********************************(15)*********************************
      DO 170 I = 1,15
         MSADTE(I) = 0
         DO 165 J=1,6
           NODPMAP(I,J)=0
  165    CONTINUE
         TAIR(I) = 0.
 170  CONTINUE

C********************************(20)*********************************
      DO 180 I=1,20
         TSOLAV(I) = 0.
 180  CONTINUE
```

```
      BETAK(1)=0.000800
      BETAK(2)=0.000800
      BETAK(3)=0.000800
      BETAK(4)=0.000602
      BETAK(5)=0.000602
      BETAK(6)=0.000602
      BETAK(7)=0.000457
      BETAK(8)=0.000457
      BETAK(9)=0.000457

      CONSK(1)=0.005417
      CONSK(2)=0.005417
      CONSK(3)=0.005417
      CONSK(4)=0.004038
      CONSK(5)=0.004038
      CONSK(6)=0.004038
      CONSK(7)=0.002913
      CONSK(8)=0.002913
      CONSK(9)=0.002913


      DO 190 I=10,20
         BETAK(I)=0.000308
         CONSK(I)=0.003214
  190 CONTINUE

C*******************************(28)************************************

      CO2PARM(1)=1.0235
      CO2PARM(2)=1.0264
      CO2PARM(3)=1.0285
      CO2PARM(4)=1.0321
      CO2PARM(5)=1.0335
      CO2PARM(6)=1.0353
      CO2PARM(7)=1.0385
      CO2PARM(8)=1.0403
      CO2PARM(9)=1.0431
      CO2PARM(10)=1.0485
      CO2PARM(11)=1.0538
      CO2PARM(12)=1.0595
      CO2PARM(13)=1.0627
      CO2PARM(14)=1.0663
      CO2PARM(15)=1.0716
      CO2PARM(16)=1.0752
      CO2PARM(17)=1.0784
      CO2PARM(18)=1.0823
      CO2PARM(19)=1.0880
      CO2PARM(20)=1.0923
      CO2PARM(21)=1.0968
      CO2PARM(22)=1.1019
      CO2PARM(23)=1.1087
      CO2PARM(24)=1.1172
      CO2PARM(25)=1.1208
      CO2PARM(26)=1.1243
      CO2PARM(27)=1.1311
      CO2PARM(28)=1.1379

C*******************(40)(40,20)(40,20,3)(40,21)***********************
      DO 260 I=1,40
         BDL(I)=1.
         FC(I)=.267
```

```
         RTEXNT(I) = .FALSE.
         TSMN(I) = 25.
         TSMX(I) = 25.
         TSOILD(I) = 25.
         TSOILN(I) = 25.
         THTS(I) = 0.
         THTR(I) = 0.
         THAD(I) = 0.
         DO 270 J=1,20
            DIFF(I,J)=258.3
            KHAR(I,J) = ' '
            PSIS(I,J) =-.175
            ROOTSV(I,J) = 0.
            RTIMPD(I,J) = 0.
            RTWTCU(I,J) = 0.
            UPF(I,J) =0.
            VH2OC(I,J) = .267
            VNC(I,J) = 0.
            VNH4C(I,J) = 0.
            VNO3C(I,J) = 0.
 270     CONTINUE
         DO 290 J=1,21
            FNL(I,J)=0.
            TUPF(I,J) = .TRUE.
            TTUPF(I,J) = .TRUE.
 290     CONTINUE
 260  CONTINUE


      KRL(1)=2
      KRL(2)=1
      KRL(3)=1
      KRL(4)=1
      KRL(5)=1

      DO 300 I=6,40
         KRL(I)=0
  300 CONTINUE

C***************************(41)(41,10)(41,20)(41,21)********************
      DO 440 I=1,41
         DO 450 J=1,10
            FWU(I,J)=0.0
  450    CONTINUE
         DO 460 J = 1,21
            PUPF(I,J) = 0.
  460    CONTINUE
         DO 465 J = 1,20
             FNU(I,J)=0.
  465    CONTINUE
  440 CONTINUE

C***************************(365)(365,7)(365,5)***********************
      DO 530 I=1,365
         AMTIRR(I) = 0.
         MTHIRR(I) = 0
         RUNOFF(I) = 0.
         DO 540 J=1,7
            NFERT(I,J)=0
            CLIMAT(I,J)=0.
 540     CONTINUE
 530  CONTINUE
```

```
C********************************(366)*******************************
      DO 610 I = 1, 366
         BOLOSS(I)=0.
         BLUM(I)=0.
         LOSSQR(I)=0
         SQLOSS(I) = 0.
         STMWT(I) =  0.
 610  CONTINUE
      RETURN
      END
```

SUBROUTINE INITIALIZE

```
SUBROUTINE INITIALIZE

C                         I N I T I A L I Z E
C   ***********************************************************
C   *   INITIAL SETUP CALCULATIONS FOR THE MODEL              *
C   ***********************************************************

      INCLUDE 'GOSCOM.FOR'

      WCELL = ROWSP/20.0
      ACELLDW = DCELL * WCELL
      VCELL = DCELL * WCELL * TCELL
      LPLOW = 20 / DCELL
      THRLN = THRLN*(ACELLDW/25.)
      POPFAC = 404685.6/POPPLT
C
C SAVE INITIAL SOIL VARIABLE
C
      DIFFOI = DIFFO(1)
      THTAOI = THETAO(1)
      BETAI = BETA(1)
      THTASI = THETAS(1)
      THTARI = THETAR(1)
      AIRDRI = AIRDR(1)
      BDI = BD(1)
      FCFCTI = FCININ(1)
C
      K = 9
      DO 100 J=1,8
        DIFFO(K) = DIFFO(K-1)
        THETAO(K) = THETAO(K-1)
        BETA(K) = BETA(K-1)
        LDEPTH(K) = LDEPTH(K-1)
        THETAS(K) = THETAS(K-1)
        FCININ(K) = FCININ(K-1)
        THETAR(K) = THETAR(K-1)
        AIRDR(K) = AIRDR(K-1)
        BD(K) = BD(K-1)
        IPSAND(K) = IPSAND(K-1)
        IPCLAY(K) = IPCLAY(K-1)
        K = K - 1
  100 CONTINUE
      LYRSOL=LYRSOL+1
      LDEPTH(1) = 5.05
      KULKNT = 3
C  DM**2 GROUND AREA / PLANT
      DELT = 1./NOITR
      KUPPER = DCELL*(NK+1)-KWIDTH
      J=1
      DO 200 L = 1,NL
  120   CONTINUE
        IF((L*DCELL.GT.LDEPTH(J)).AND.(J.LT.9)) THEN
           J = J+1
           GO TO 120
        ENDIF
        IF(J.GT.LYRSOL) J=LYRSOL
        LYRDPH(L) = J
        ARDRCN(L) = ALOG(-15./PSISFC) / ALOG((THETAR(J)-AIRDR(J)) /
                                        (FCININ(J)-AIRDR(J)))
C
C FLXMAX AND FLXMIN ARE MAX AND MIN WATER FLUXES. CM**2
C FC(L) AND THTS(L) ARE FIELD CAPACITY AND SATURATED VOLUMETRIC WATER
```

```
C WATER CONTENTS, RESPECTIVELY, BY LAYER.
C
        FLXMAX(J)=ABS(DIFFO(J)*((FCININ(J)-THETAR(J))/DCELL)*
     .       (WCELL*DELT)*EXP(BETA(J)*(FCININ(J)-THETAO(J))))
        FC(L) = FCININ(J)
        BDL(L) = BD(J)
        THTS(L) = THETAS(J)
        THTR(L) = THETAR(J)
        THAD(L) = AIRDR(J)
        FLXMIN(J) = ABS(DIFFO(J) * ((THETAO(J)-THETAR(J))/DCELL)
     .                                              *WCELL*DELT)
        IF(FLXMIN(J).LE.0.0001) FLXMIN(J) = 0.0001
  200 CONTINUE

      DO 210 I=1,40
        DO 210 J=1,20
          DO 210 K=1,3
            RTWT(I,J,K)=0
  210 CONTINUE
      RTWT(1,1,1)  = .035 / POPFAC
      RTWT(2,1,1)  = .025 / POPFAC
      RTWT(3,1,1)  = .015 / POPFAC
      RTWT(4,1,1)  = .010 / POPFAC
      RTWT(5,1,1)  = .005 / POPFAC
      RTWT(1,2,1)  = .010 / POPFAC
      RTWT(1,NK,1) = .035 / POPFAC
      RTWT(2,NK,1) = .025 / POPFAC
      RTWT(3,NK,1) = .015 / POPFAC
      RTWT(4,NK,1) = .010 / POPFAC
      RTWT(5,NK,1) = .005 / POPFAC
      RTWT(1,NK-1,1) = .010 / POPFAC

C
C SOILNO OF 0.10 IS POT. MIN. NIT. FOR GROUP 1 SOILS (COARSE LOAMY/LOAM)
C SOILNO OF 0.06 IS POT. MIN. NIT. FOR GROUP 2 SOILS (FINE LOAMY/CLAY)
C
      IF(PSISFC.LT.-0.2) THEN
         SOILNO = 0.06
      ELSE
         SOILNO = 0.10
      ENDIF
C
      DO 220 L=1,NL
        VNH4C(L,1)=0.
        VNO3C(L,1)=0.
        VNC(L,1)=0.
  220 CONTINUE
      DO 240 I=1,197
        J=(I-1)/15+1
        IF(J.GT.14) J=14
        L=(I-1)/DCELL+1
        IF(L.GT.NL) L=NL
C
C INITIALIZE NO3 AND NH4 ARRAY WITH INITIAL RESIDUAL NO3 OR WITH
C 4.0 LBS/6 INCHES LAYER
C
        IF(RNNO3(J).LE.0.) RNNO3(J) = 2.0
        IF(RNNH4(J).LE.0.) RNNH4(J) = 0.2
        VNO3C(L,1) = VNO3C(L,1)+RNNO3(J)/15.*.011219
        VNH4C(L,1) = VNH4C(L,1)+RNNH4(J)/15.*.011219
C
C CONVERT VNC FROM % OM TO MG/CM**3.  1000. CONVERTS GRAMS TO MG.
```

```
C POTN IS POTENTIAL MINERALIZABLE N IN ORGANIC MATTER. I IS DEPTH IN CM
C
       POTN = (AMAX1(0.0, 0.055 - 0.0241*LOG10(FLOAT(I))))*1.1
       POTN = AMIN1(0.05, POTN)

       VNC(L,1)    = VNC(L,1)+(OMA(J)/100.)*BDL(L)*1000.*POTN*SOILNO

  240 CONTINUE
      DO 260 L=1,NL
        VNO3C(L,1) = VNO3C(L,1)/DCELL
        VNH4C(L,1) = VNH4C(L,1)/DCELL
        VNC(L,1)   = VNC(L,1)/DCELL
        DO 260 K=2,NK
          VNO3C(L,K) = VNO3C(L,1)
          VNH4C(L,K) = VNH4C(L,1)
          VNC(L,K)   = VNC(L,1)
  260 CONTINUE
      J=1
      DO 300 L=1,NL
        IF((L-1)*DCELL.GT.J*15) THEN
          J=J+1
          IF(J.GT.14) J=14
        ENDIF
        N=LYRDPH(L)
        DO 280 K=1,NK
          VH2OC(L,K)=FC(L)*H2OINT(J)/100.
          IF(VH2OC(L,K).LT.THETA0(N)) VH2OC(L,K) = THETA0(N)
          TIH2OC = TIH2OC + VH2OC(L,K)
          DIFF(L,K)=DIFF0(N)*EXP(BETA(N)*(VH2OC(L,K)-THETA0(N)))
          TEMP1G = (VH2OC(L,K)-AIRDR(N))/(FCININ(N)-AIRDR(N))
          PSIS(L,K)= PSISFC * TEMP1G**ARDRCN(L)
  280   CONTINUE
  300 CONTINUE
      TBL = NL * DCELL
      IF(WATTBL.GT.TBL) THEN
        THETAI = FCININ(LYRSOL)*H2OINT(14)/100.
        IF(THETAI.LE.0.0)THETAI = FCININ(LYRSOL)
      ENDIF
      TIH2OC = (TIH2OC * DCELL * 10.)/NK
      TNNO3 = 0.
      TNNH4 = 0.
      DO 320 L=1,NL
         DO 320 K=1,NK
            TNNO3 = TNNO3 + VNO3C(L,K)
            TNNH4 = TNNH4 + VNH4C(L,K)
 320  CONTINUE
      TNNO3 = TNNO3 * 0.891*DCELL*WCELL
      TNNH4 = TNNH4 * 0.891*DCELL*WCELL
      DAY1SN = TNNO3 + TNNH4
      DAY1PN = ROOTN + STEMN + SLEAFN + SEEDN + BURRN
      WTDAY1 = ROOTWT+STEMWT+GBOLWT+LEAFWT+SQWT+XTRAC+COTXX+RESC
C
C GET THE CO2 CORRECTION FACTOR FOR PHOTOSYNTHESIS.
C
      IF(IYEAR.GT.100) THEN
         INDXCO2=IYEAR-1959
      ELSE
         INDXCO2=IYEAR-59
      ENDIF
      IF(INDXCO2.LT.1) INDXCO2=1
      IF(INDXCO2.GT.28) INDXCO2=28
      PNETCOR=CO2PARM(INDXCO2)
```

```
RETURN
END
```

SUBROUTINE SOILHYDR

```
SUBROUTINE SOILHYDR

      INTEGER*4 POINTR
      CHARACTER FILNAM*13,DSCRIP*51,FILEPATH*55,XDUM1*3,XDUM2*3
      INCLUDE 'GOSCOM.FOR'

 1220 FORMAT(//
     .  7X,'NUMBER OF SOIL LAYERS ... ',I1,
     . 10X,'WATER TABLE DEPTH (IN) .. ',F4.0, //
     . 16X,'LAYER NO.    LAYER DEPTH    % SAND    % CLAY' /
     . 16X,'                (IN)')
 1240 FORMAT(18X,I3,10X,F6.0,7X,I4,6X,I4)
 1260 FORMAT(/ 8X, 'Rainfall Runoff: ',A3,
     .         18X,'Irrigation Runoff: ',A3)

      CALL BLDPATH(ROOTDIR,'hydr    ',OPSYS,SOLHYD,FILEPATH,IO)
      OPEN(11,FILE=FILEPATH,FORM=FILFRM,STATUS='OLD',ERR=200)
      IF((OPSYS.EQ.'dos').OR.(OPSYS.EQ.'DOS')) THEN
        READ(11,ERR=200,END=300) FILNAM,DSCRIP,LYRSOL,DIFFOC,THTAOC,
     .          BETAC,THTASC,FCINIC,THTARC,AIRDRC,BDC,DIFFOW,THTAOW,
     .          BETAW,THTASW,FCININW,THTARW,AIRDRW,BDW,TD,THETAI,
     .          BDSLOP,BDRATO,PSISFC,WATTBL,POINTR
        DO 100 I=1,LYRSOL
          READ(11,ERR=200,END=300) LDEPTH(I),DIFFO(I),THETAO(I),
     .          BETA(I),THETAS(I),FCININ(I),THETAR(I),AIRDR(I),BD(I),
     .          IPSAND(I),IPCLAY(I),POINTR
  100   CONTINUE
        READ(11,ERR=150,END=150) IFGRAIN,IFGIRR
      ELSE
        READ(11,*,ERR=200,END=300) FILNAM,DSCRIP,LYRSOL,DIFFOC,THTAOC,
     .          BETAC,THTASC,FCINIC,THTARC,AIRDRC,BDC,DIFFOW,THTAOW,
     .          BETAW,THTASW,FCININW,THTARW,AIRDRW,BDW,TD,THETAI,
     .          BDSLOP,BDRATO,PSISFC,WATTBL,POINTR
        DO 120 I=1,LYRSOL
          READ(11,*,ERR=200,END=300) LDEPTH(I),DIFFO(I),THETAO(I),
     .          BETA(I),THETAS(I),FCININ(I),THETAR(I),AIRDR(I),BD(I),
     .          IPSAND(I),IPCLAY(I),POINTR
  120   CONTINUE
        READ(11,*,ERR=150,END=150) IFGRAIN,IFGIRR
      ENDIF

  150 CONTINUE
      CLOSE(11)
      IF(WATTBL.LT.1) WATTBL=200
      IF(PSISFC.GT.0.) PSISFC=-PSISFC
      WRITE(20,1220) LYRSOL,WATTBL/2.54
      DO 170 I=1,LYRSOL
          WRITE(20,1240) I,LDEPTH(I)/2.54,IPSAND(I),IPCLAY(I)
  170 CONTINUE
      XDUM1='ON '
      XDUM2='ON '
      IF(IFGRAIN.GT.0) XDUM1='OFF'
      IF(IFGIRR.GT.0) XDUM2='OFF'
      WRITE(20,1260) XDUM1,XDUM2
      RETURN
  200 CONTINUE
        PRINTBUF=' error on read in soilhydr'
        CALL WRITERR
        CLOSE(11)
        ABEND = .TRUE.
        RETURN
  300 CONTINUE
```

```
  PRINTBUF=' end-of-file while reading soilhydr'
  CALL WRITERR
  CLOSE(11)
  ABEND = .TRUE.
  RETURN
END
```

SUBROUTINE SOILIMPD

```
SUBROUTINE SOILIMPD
CHARACTER FILEPATH*55

      INCLUDE 'GOSCOM.FOR'

 1000 FORMAT(A12)

      CALL BLDPATH(ROOTDIR,'hydr    ',OPSYS,'soilimpd.dat',
     .                                        FILEPATH,IO)
      OPEN(11,FILE=FILEPATH,FORM='FORMATTED',
     .                                STATUS='OLD',ERR=200)

      READ(11,1000,ERR=200,END=300) SNAME
      READ(11,*,ERR=200,END=300) NCURVE
      DO 180 I=1,NCURVE
        READ(11,*,ERR=200,END=300) INRIM,GH2OC(I)
        DO 160 J=1,INRIM
          READ(11,*,ERR=200,END=300) TSTBD(I,J),TSTIMP(I,J)
  160   CONTINUE
  180 CONTINUE
      CLOSE(11)
      RETURN
  200 CONTINUE
        PRINTBUF=' error on read in soilimpd'
        CALL WRITERR
        ABEND = .TRUE.
        CLOSE(11)
        RETURN
  300 CONTINUE
        PRINTBUF=' end-of-file while reading soilimpd'
        CALL WRITERR
        ABEND = .TRUE.
        CLOSE(11)
        RETURN
      END
```

```
SUBROUTINE INITSOIL
      INTEGER*2 ISOILW(12)
      CHARACTER FILNAM*13,DSCRIP*51,FILEPATH*55

      INCLUDE 'GOSCOM.FOR'

 1200 FORMAT(//
     .'  DEPTH     RESIDUAL N (LB/ACRE)     ORGANIC     H2O CONTENT' /
     .'  (IN.)       NITRATE    AMMONIA     MATTER (%)    % FIELD CAP.'/
     .'    1- 6',2(4X,F7.1),8X,F6.2,8X,F7.0 /
     .'    7-12',2(4X,F7.1),8X,F6.2,8X,F7.0 /
     .'   13-18',2(4X,F7.1),8X,F6.2,8X,F7.0 /
     .'   19-24',2(4X,F7.1),8X,F6.2,8X,F7.0 /
     .'   25-30',2(4X,F7.1),8X,F6.2,8X,F7.0 /
     .'   31-36',2(4X,F7.1),8X,F6.2,8X,F7.0)

      CALL BLDPATH(ROOTDIR,'init    ',OPSYS,INTSOL,FILEPATH,IO)
      OPEN(11,FILE=FILEPATH,FORM=FILFRM,STATUS='OLD',ERR=200)
      DO 60 I=1,12
        ISOILW(I)=100
   60 CONTINUE
      IF((OPSYS.EQ.'dos').OR.(OPSYS.EQ.'DOS')) THEN
        READ(11,ERR=200,END=80) FILNAM,DSCRIP,(RNNH4(I),RNNO3(I),
     .          OMA(I),ISOILW(I),I=1,12)
      ELSE
        READ(11,*,ERR=200,END=80) FILNAM,DSCRIP,(RNNH4(I),RNNO3(I),
     .          OMA(I),ISOILW(I),I=1,12)
      ENDIF

   80 CONTINUE
      DO 100 I=1,12
        H2OINT(I) = ISOILW(I)
  100 CONTINUE
      WRITE(20,1200) (RNNO3(I),RNNH4(I),OMA(I),H2OINT(I),I=1,6)
      CLOSE(11)
      RETURN
  200 CONTINUE
        PRINTBUF=' error on read in initsoil'
        CALL WRITERR
        ABEND = .TRUE.
        CLOSE(11)
        RETURN
      END
```

```
SUBROUTINE AGINPUTS
      REAL CSTIRR,CSTCUL,AMTAMM,AMTNIT,AMTURA,CSTFRT
      INTEGER*2 ICDPTH,ITRWTH,MTHFRT,ISDHRZ,ISDDPH
      CHARACTER FILNAM*13,DSCRIP*51,CDATE*10,FILEPATH*55
      INTEGER*4 POINTR

      INCLUDE 'GOSCOM.FOR'

 1000 FORMAT( '1' /
     .'           ------IRRIGATION------  -----------------FERTILIZER',
     .'--------------------' /
     .'   DATE    AMOUNT  METHOD   COST    NH4    NO3    UREA   COST ',
     .'  METHOD   LOCATION' /
     .'             (in)           ($/a)  -----(lb N / a)-----  ($/a)',
     .'           dpth side' /)
 1020 FORMAT(1X,A8,1X,F5.2,3X,A6,1X,F7.2,3F7.1,2X,F6.2,2X,A6,2I5)

      IF(IRRFRT.NE.'             ') THEN
        CALL BLDPATH(ROOTDIR,'irrfert ',OPSYS,IRRFRT,FILEPATH,IO)
        OPEN(11,FILE=FILEPATH,FORM=FILFRM,STATUS='OLD',ERR=300)
        NAPS=0
        IDUM=0
  100   CONTINUE
          IF((OPSYS.EQ.'dos').OR.(OPSYS.EQ.'DOS')) THEN
            READ(11,END=300,ERR=300) FILNAM,DSCRIP,CDATE
          ELSE
            READ(11,*,END=300,ERR=300) FILNAM,DSCRIP,CDATE
          ENDIF
          IF(IDUM.EQ.0) THEN
            WRITE(20,1000)
            IDUM=1
          ENDIF
          CALL CALTOJULAN(CDATE,IDUM1,IDUM2,IDUM3,J)
          I = J-JDSTRS+1
          IF(I.LT.1) I=1
          IF((OPSYS.EQ.'dos').OR.(OPSYS.EQ.'DOS')) THEN
            READ(11,END=300,ERR=300) H2OAMT,CSTIRR,MTHIRR(I),
     .            ICDPTH,CSTCUL,ITRWTH,AMTAMM,AMTNIT,AMTURA,CSTFRT,
     .            MTHFRT,ISDHRZ,ISDDPH,POINTR
          ELSE
            READ(11,*,END=300,ERR=300) H2OAMT,CSTIRR,MTHIRR(I),
     .            ICDPTH,CSTCUL,ITRWTH,AMTAMM,AMTNIT,AMTURA,CSTFRT,
     .            MTHFRT,ISDHRZ,ISDDPH,POINTR
          ENDIF
          IDUM1=4
          IF(H2OAMT.GT.0.) THEN
            IDUM1=MTHIRR(I)+1
            AMTIRR(I)=AMTIRR(I)+H2OAMT
          ENDIF
          IDUM2=4
          IF((AMTAMM.GT..01).OR.(AMTNIT.GT..01).OR.(AMTURA.GT..01)) THEN
            IDUM2=MTHFRT+1
            NAPS=NAPS+1
            NFERT(NAPS,1) = I
            NFERT(NAPS,2) = AMTAMM+.5
            NFERT(NAPS,3) = AMTNIT+.5
            NFERT(NAPS,4) = AMTURA+.5
            NFERT(NAPS,5) = MTHFRT
            NFERT(NAPS,6) = ISDHRZ
            NFERT(NAPS,7) = ISDDPH
          ENDIF
          WRITE(20,1020) CDATE,AMTIRR(I),IMTHOD(IDUM1),
```

```
     .           CSTIRR,AMTAMM,AMTNIT,AMTURA,CSTFRT,FMTHOD(IDUM2),
     .           ISDDPH,ISDHRZ
        GO TO 100
      ENDIF
  300 CONTINUE
      CLOSE(11)
  650 CONTINUE
      RETURN
      END
```

```
SUBROUTINE WEATHER
C                       W E A T H E R
C   *************************************************************
C   *   SUBROUTINE TO READ IN WEATHER, FERTILIZATION, AND       *
C   *       IRRIGATION                                          *
C   *************************************************************
C
      DIMENSION C13(7)
      CHARACTER A00*5,A01*1,FILEPATH*55

      INCLUDE 'GOSCOM.FOR'

 1000 FORMAT(/
     .  4X,'Last Actual Weather ',I2.2,'/',I2.2,'/',I2.2,
     . 13X,'Last Future Weather ',I2.2,'/',I2.2)
 1020 FORMAT(A1)
 1040 FORMAT(' Error, missing data in actual weather on day ',A5)
 1060 FORMAT(' Error, missing data in final weather on day ',I5)
 1080 FORMAT(' Error, final weather is missing day ',I5)
 1100 FORMAT(A80)

      OPEN(23,FILE=ERRFLE,STATUS='UNKNOWN')
  100 CONTINUE
        READ(23,*,END=120,ERR=100) A01
      GO TO 100
  120 CONTINUE
      BACKSPACE(23)

C   READ FUTURE WEATHER FILE

      CALL BLDPATH(ROOTDIR,'wthfls  ',OPSYS,FURWTH,FILEPATH,IO)
      OPEN(11,FILE=FILEPATH,FORM='FORMATTED',
     .                                 STATUS='OLD',ERR=220)

      C13(4)=0
      C13(7)=1
  200 CONTINUE
        READ(11,1020,END=240,ERR=230) A01
        IF(A01.EQ.'#') THEN
          GO TO 200
        ELSE
          BACKSPACE(11)
          READ(11,*,END=240,ERR=230) C13(7),A00,(C13(I),I=1,3),
     .                         C13(5),C13(6)
          J=C13(7)-JDSTRS+1
          IF(C13(6).LE.0.) C13(6)=88.
          IF(J.GT.0) THEN
             DO 210 IC=1,7
                 CLIMAT(J,IC) = C13(IC)
  210        CONTINUE
          ENDIF
        ENDIF
      GO TO 200
  220 CONTINUE
        PRINTBUF=' OPEN ERROR IN FUTURE WEATHER FILE @#*$!&'
        WRITE(23,1100) PRINTBUF
        ABEND=.TRUE.
        CLOSE(23)
        RETURN
  230 CONTINUE
        PRINTBUF=' READ ERROR IN FUTURE WEATHER FILE @#*$!&'
        WRITE(23,1100) PRINTBUF
```

```
  240 CONTINUE
      LDAYFW = C13(7)
      C13(7) = 1
      JDAYLW = LDAYFW
      CLOSE(11)

C   READ PREDICTED WEATHER FILE

      IF(PRDWTH.NE.'              ') THEN
        CALL BLDPATH(ROOTDIR,'wthfls  ',OPSYS,PRDWTH,FILEPATH,IO)
        OPEN(11,FILE=FILEPATH,FORM='FORMATTED',
     .                              STATUS='OLD',ERR=290)

  250   CONTINUE
          READ(11,1020,END=280,ERR=280) A01
          IF(A01.EQ.'#') THEN
            GO TO 250
          ELSE
            BACKSPACE(11)
            READ(11,*,END=280,ERR=280) C13(7),A00,(C13(I),I=1,3),
     .                         C13(5),C13(6)
            J=C13(7)-JDSTRS+1
            IF(C13(6).LE.0.) C13(6)=88.
            IF(J.GT.0) THEN
              DO 260 IC=1,7
                 CLIMAT(J,IC) = C13(IC)
  260         CONTINUE
            ENDIF
          ENDIF
        GO TO 250
  280   CONTINUE
        LDAYPW = C13(7)
        C13(7) = 1
        IF(LDAYPW.GT.JDAYLW) JDAYLW=LDAYPW
        CLOSE(11)
      ENDIF
  290 CONTINUE

C   READ IN CURRENT WEATHER FILE ON TOP OF FUTURE WEATHER

      CALL BLDPATH(ROOTDIR,'wthfls  ',OPSYS,ACTWTH,FILEPATH,IO)
      OPEN(11,FILE=FILEPATH,FORM='FORMATTED',
     .                              STATUS='OLD',ERR=330)

  300 CONTINUE
        READ(11,1020,END=360,ERR=340) A01
        IF(A01.EQ.'#') THEN
          GO TO 300
        ELSE
          BACKSPACE(11)
          READ(11,*,END=360,ERR=340) C13(7),A00,(C13(I),I=1,3),
     .                                  C13(5),C13(6)
          IF((C13(7).GE.JDSTRS).AND.(C13(7).LE.JDSTPS)) THEN
            DO 310 I=1,6
              IF(ABS(ABS(C13(I))-6999.0).LT..01) THEN
                WRITE(PRINTBUF,1040) A00
                WRITE(23,1100) PRINTBUF
                GO TO 300
              ELSEIF(ABS(ABS(C13(I))-9111.0).LT..01) THEN
                WRITE(PRINTBUF,1040) A00
                WRITE(23,1100) PRINTBUF
                GO TO 300
```

```
          ENDIF
310     CONTINUE
        ENDIF
        J=C13(7)-JDSTRS+1
        IF(C13(6).LE.0.) C13(6)=88.
        IF(J.GT.0.AND.J.LE.365) THEN
          DO 320 IC=1,7
            CLIMAT(J,IC) = C13(IC)
320       CONTINUE
        ENDIF
      ENDIF
    GO TO 300
330 CONTINUE
      PRINTBUF=' OPEN ERROR IN ACTUAL WEATHER FILE @#*$!&'
      WRITE(23,1100) PRINTBUF
      ABEND=.TRUE.
      CLOSE(23)
      RETURN
340 CONTINUE
      PRINTBUF=' READ ERROR IN ACTUAL WEATHER FILE @#*$!&'
      WRITE(23,1100) PRINTBUF
360 CONTINUE
    CLOSE(11)
    LDAYAW = C13(7)
    IF(LDAYAW.GT.JDAYLW) JDAYLW=LDAYAW
    CALL JULANTOCAL(LDAYAW,IYEAR,I0,I1)
    CALL JULANTOCAL(LDAYFW,IYEAR,I2,I3)
    WRITE(20,1000) I0,I1,IYEAR,I2,I3
    K=0
    DO 400 I=JDSTRS,JDSTPS
      K=K+1
      DO 380 J=1,6
        IF((ABS(CLIMAT(K,J))-6999..LT..01).AND.
     .      (ABS(CLIMAT(K,J))-6999..GT.-.01)) THEN
          WRITE(PRINTBUF,1060) CLIMAT(K,7)
          WRITE(23,1100) PRINTBUF
          ABEND=.TRUE.
        ELSEIF((ABS(CLIMAT(K,J))-9111..LT..01).AND.
     .          (ABS(CLIMAT(K,J))-9111..GT.-.01)) THEN
          WRITE(PRINTBUF,1060) CLIMAT(K,7)
          WRITE(23,1100) PRINTBUF
          ABEND=.TRUE.
        ENDIF
380   CONTINUE
      IF(CLIMAT(K,7).LT..01) THEN
        L=JDSTRS+K-1
        WRITE(PRINTBUF,1080) L
        WRITE(23,1100) PRINTBUF
        ABEND=.TRUE.
      ENDIF
400 CONTINUE
    CLOSE(23)
    RETURN
    END
```

SUBROUTINE CLYMAT

```
SUBROUTINE CLYMAT
C   ***************************************************************
C   *                                                             *
C   *                    CLIMATE SUBROUTINE                       *
C   *                                                             *
C   ***************************************************************

      INCLUDE 'GOSCOM.FOR'

      DAYNUM = JDSTRS+IDAY-1
      RI   = CLIMAT(IDAY,1)
      TMAX = (CLIMAT(IDAY,2)-32.) * .5555556
      TMIN = (CLIMAT(IDAY,3)-32.) * .5555556
      RAIN = (CLIMAT(IDAY,5) + AMTIRR(IDAY)) * 25.4
      MH2O = 0
      IF(AMTIRR(IDAY).GT.0.) THEN
        LDAYIR = DAYNUM
        J = MTHIRR(IDAY)+1
        IF(IMTHOD(J).EQ.'SPKLER') MH20 = 1
      ENDIF

      CALL RRUNOFF

      WIND = CLIMAT(IDAY,6)
      IF(RAIN.LE.1.5) RAIN = 0.0
      IF(TMAX.LE.0) TMAX = 2.0
      IF(TMIN.LE.0) TMIN = 1.0

C     *** CALCULATE SOLAR DECLINATION USING D. W. STEWART'S PROGRAM
C         WHICH INCORPORATES THE ALGORITHM OF ROBINSON AND RUSSELO. ***
      DEGRAD=3.1416/180.0
      XLAT=LATUDE*DEGRAD
      DEC=C1(1)
      DO 2 I=2,5
        N=I-1
        J=I+4
        PHI=N*0.01721*DAYNUM
    2 DEC=DEC+C1(I)*DSIN(PHI)+C1(J)*DCOS(PHI)
      DEC=DEC*DEGRAD
C     ----------------------------------------------------------------
C     *** CALCULATE DAYLENGTH. ***
      DAYLNG=DACOS((-0.014544-DSIN(XLAT)*DSIN(DEC))/(DCOS(XLAT) *
     &       DCOS(DEC)))*7.6394

      WATTSM = RI * 697.45 / (DAYLNG*60.)
      RN = WATTSM * .8 - 26.
C RN = NET RADIATION IN WATTS/M**2
      JOULES = RI*41860.
C JOULES = NET RADIATION IN JOULES/M**2
      TMINT  = (CLIMAT(IDAY+1,3)-32.) * .5555556
      IF(TMINT.LE.0.0) TMINT = 1.0
      IF(RAIN.LE.2.54) KRAIN = 0
      IF(RAIN.GT.2.54.AND.RAIN.LE.25.4) KRAIN = 1
      IF(RAIN.GT.25.4) KRAIN = 2
C     ----------------------------------------------------------------
C     ----------------------------------------------------------------
C      THE FOLLOWING STATEMENTS CALCULATE TDAY AND TNYT FOLLOWING LOGIC
C      OF B.ACOCK.
       WATACT = PI * JOULES/3600./2./DAYLNG
       X5 = 0.0945-(WATACT*8.06*10.0**(-5))+(TMAX*6.77*10.0**(-4))
       TMX5WA = TMAX/X5/WATACT
       IF(TMX5WA.GE.1.)TMX5WA = 1.0
```

```
        IF(TMX5WA.LE.-1.)TMX5WA = -1.0
        TMAXHR = DAYLNG/PI*(PI-(ASIN(TMX5WA)))
C      -----------------------------------------------------------------
       X6 = ((TMAX-TMIN)/2.)
       X7 = (PI/TMAXHR)
       X8 = 1.5 * PI
       TDAY=X6*(1-(TMAXHR/PI/DAYLNG*COS((X7*DAYLNG)+X8)))+TMIN
C      -----------------------------------------------------------------
C      *** CALCULATE AIR TEMPERATURE AT DUSK. ***
       TAIRSS=(X6*(1.0+SIN((X7*DAYLNG+X8))))+TMIN
C      -----------------------------------------------------------------
C      *** CALCULATE AVERAGE NIGHT-TIME AIR TEMPERATURE. ***
       IF(TAIRSS-TMINT.LE.0.5)  TAIRSS = TMINT + 0.5
       TNYT=(TAIRSS-TMINT)/(-ALOG(TMINT/TAIRSS))
C      -----------------------------------------------------------------
C      *** POLLINATION SWITCH DEPENDS ON RAIN. ***
       POLYNA = 1
       IF(RAIN.GE.12.7.AND.MH2O.EQ.0.) POLYNA = 0

       CALL JULANTOCAL(DAYNUM,IYEAR,MO,DAZE)

       TAVG=(TDAY*DAYLNG+TNYT*(24.-DAYLNG))/24.

C  LTYPE=1 OKRA LEAF.   LTYPE=0 NORMAL LEAF
C TEST OF CHANGE IN CANOPY LIGHT INTERCEPTION DUE TO PIX   06/29/90

       ZDUM = Z + ZPIXD * .6
       IF(LTYPE.EQ.1) THEN
          INT=(-2.05595+1.64301*ZDUM+(-.00648851*(ZDUM**2)))/100.
       ELSE
          INT = 1.0756*ZDUM/ROWSP
       ENDIF

       IF(INT.LT.0.) INT = 0.
       IF(INT.GE.0.95) INT = 0.95
C
C  INT = FRACTION OF INCIDENT LIGHT INTERCEPTED BY PLANT CANOPY.
C  BAKER ET. AL. CANOPY ARCHITECTURE IN RELATION TO YIELD.
C  CHAPTER 3 IN 'CROP PHYSIOLOGY' ED. V. S. GUPTO.
C
       IF(LAI.GT.LMAX) LMAX = LAI
       IF(LAI.LT.LMAX.AND.LAI.LT.3.1) THEN
          CLAI = 0.0
          IF(Z.LT.ROWSP) THEN
             CLAI = 3.1 * Z/ROWSP
             IF(LAI.LE.CLAI) INT = INT * LAI / CLAI
          ELSE
             INT=INT*LAI/3.1
          ENDIF
       ENDIF
       IF(INT.LT.0.) INT = 0.
       IF(INT.GE.0.95) INT = 0.95

       CALL TMPSOL

       IF(KDAY.EQ.1) AVTPFN(1) = TAVG
       RETURN
       END
```

```
      SUBROUTINE JULANTOCAL (JULIAN,IYEAR,MONTH,IDAY)
C   ************************************************************
C   *                                                          *
C   *   DATE SUBROUTINE.  CONVERTS JULIAN TO CALENDAR AND      *
C   *    ALLOWS FOR LEAP YEARS.                                *
C   *                                                          *
C   ************************************************************

      INTEGER DACNT(12)

      DACNT(1) = 31
      DACNT(2) = 28
      IF(IYEAR/4*4.EQ.IYEAR) DACNT(2) = 29
      DACNT(3) = 31
      DACNT(4) = 30
      DACNT(5) = 31
      DACNT(6) = 30
      DACNT(7) = 31
      DACNT(8) = 31
      DACNT(9) = 30
      DACNT(10) = 31
      DACNT(11) = 30
      DACNT(12) = 31

      MONTH = 1
      IDAY = JULIAN
      DO 100 I=1,12
        IF(IDAY.LE.DACNT(I)) GO TO 200
        MONTH = MONTH + 1
        IDAY = IDAY - DACNT(I)
  100 CONTINUE
  200 CONTINUE
      RETURN
      END
```

SUBROUTINE OUT

```
      SUBROUTINE OUT(ARRAY,TOTAL,IGO)
C   ***************************************************************
C   *                                                             *
C   *  THIS SUBROUTINE PLOTS THE SOIL SLAB AND THE DENSITIES     *
C   *  OF THE ARRAY ELEMENTS IN EACH CELL.                        *
C   *                                                             *
C   ***************************************************************

      DIMENSION CAPSCA(11), PSISCA(11), VNOSCA(11),
     .  ROOSCA(11), RANGE(11), ARRAY(40,20),TEMPSCA(11)

      CHARACTER TTL1*40, TTL3*40, TTL4*40, TTL5*40,TTL8*40,
     .  TTL7*40, UNITST*16, VNOUNI*24, VH2UNI*24, PSIUNI*24,
     .  NITUNT*16, TTL1R*40, TTL2R*40, UNITS*24, UNITSR*16,
     .  UNTS*24, UNTST*16, TL1*40, TL2*40, TEMPUNT*16,TTL9*40,
     .  TUNITS*24, UNITPS*16

      INCLUDE 'GOSCOM.FOR'

      DATA ROOSCA/1.0E-25,.0001,.0005,.005,.01,.015,.02,.025,.03,
     .           .035,.04/
      DATA PSISCA/-15.,-10.,-6.,-3.,-1.5,-1.,-.6,-.4,-.2,-.1,0./
      DATA VNOSCA/1.0E-25,.01,.02,.03,.04,.05,.06,.07,.08,.09,.1/
      DATA CAPSCA/0.0,.05,.1,.15,.2,.25,.3,.35,.4,.45,.5/
      DATA TEMPSCA/0.00,5.00,10.00,15.00,20.00,25.00,30.00,35.00,
     . 40.00,45.00,50.00/
      DATA TTL1R  /'ROOTS IN EACH CELL, TOTAL                '/
      DATA TTL2R  /'                                         '/
      DATA TTL1   /'VOLUMETRIC WATER CONTENT OF SOIL         '/
      DATA TTL3   /'                                         '/
      DATA TTL4   /'SOIL WATER POTENTIAL FOR            '/
      DATA TTL5   /'VOLUMETRIC NITRATE CONTENT OF SOIL       '/
      DATA TTL7   /'VOLUMETRIC AMMONIA CONTENT OF SOIL       '/
      DATA TTL8   /'AVERAGE SOIL TEMPERATURE                 '/
      DATA TTL9   /'AT THE END OF THE DAY                    '/
      DATA UNITS  /'G/CM**3 SOIL            '/
      DATA UNITPS /' BARS          '/
      DATA PSIUNI /' BARS IN ROOT ZONE       '/
      DATA VNOUNI /' MG/N PER CM**3          '/
      DATA VH2UNI /' CM**3/CM**3 SOIL         '/
      DATA TUNITS /' DEGREES CELSIUS         '/
      DATA UNITSR /' GM. DRY WEIGHT '/
      DATA UNITST /' MM WATER       '/
      DATA NITUNT /' LBS N PER ACRE '/
      DATA TEMPUNT/' DEGREES CELSIUS'/
C
C
 100  FORMAT(///6X,A40,15X,'DAY ',I2.2,'/',I2.2/6X,A40/
     . 6X,'UNITS - ',A24,23X,'LEGEND'//24X,'1 1 1 1 1 1 1 1 1 1 2'/
     . 6X,'1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0',18X,A1,
     . '  <= ',F8.4//53X,F8.4,' < ',A1,' <= ',F8.4)
 102  FORMAT(3(1X,I2,3X,20A2/),1X,I2,3X,20A2,7X,F8.4,' < ',
     . A1,' <= ',F8.4)
 104  FORMAT(3(1X,I2,3X,20A2/),1X,I2,3X,20A2,7X,F8.4,' < ',A1//
     . 6X,'TOTAL = ',F11.2,1X,A16)
 200  FORMAT(///6X,A40,15X,'DAY ',I2.2,'/',I2.2/6X,A40/
     . 6X,'UNITS - ',A24,23X,'LEGEND'//24X,'1 1 1 1 1 1 1 1 1 1 2'/
     . 6X,'1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0',18X,A1,
     . '  < ',E8.2//53X,E8.2,' < ',A1,' < ',E8.2)
 202  FORMAT(3(1X,I2,3X,20A2/),1X,I2,3X,20A2,7X,E8.2,' < ',
     .  A1,' < ',E8.2)
 204  FORMAT(3(1X,I2,3X,20A2/),1X,I2,3X,20A2,7X,E8.2,' < ',A1//
```

```
     . 6X,'TOTAL = ',F11.2,1X,A16)
C
C
      IF(IGO.EQ.1) THEN
        IF(TOTAL.EQ.TNNH4) THEN
           TL1=TTL7
        ELSE
           TL1=TTL5
        ENDIF
        TL2=TTL3
        UNTS=VNOUNI
        UNTST=NITUNT
        DO 20 I=1,11
          RANGE(I)=VNOSCA(I)
   20   CONTINUE
      ENDIF
      IF(IGO.EQ.2) THEN
        TL1=TTL1
        TL2=TTL3
        UNTS=VH2UNI
        UNTST=UNITST
        DO 40 I=1,11
          RANGE(I)=CAPSCA(I)
   40   CONTINUE
      ENDIF
      IF(IGO.EQ.3) THEN
        TL1=TTL1R
        TL2=TTL2R
        UNTS=UNITS
        UNTST=UNITSR
        DO 60 I=1,11
          RANGE(I)=ROOSCA(I)
   60   CONTINUE
      ENDIF
      IF(IGO.EQ.4) THEN
        TL1=TTL4
        TL2=TTL3
        UNTS=PSIUNI
        UNTST=UNITPS
        DO 80 I=1,11
          RANGE(I)=PSISCA(I)
   80   CONTINUE
      ENDIF
      IF (IGO.EQ.5) THEN
         TL1 = TTL8
         TL2 = TTL9
         UNTS = TUNITS
         UNTST=TEMPUNT
         DO 70 I=1,11
            RANGE(I)=TEMPSCA(I)
   70    CONTINUE
      ENDIF
      DO 1 K=1, 20
      DO 1 L=1, 40
      ARAYLK = ARRAY(L,K)
      DO 2 I=1, 11
      RANGE1 = RANGE(I)
      IF(ARAYLK.LE.RANGE1) GO TO 1
 2    CONTINUE
      I = 12
 1    KHAR(L,K) = KA(I)
      IF(IGO.EQ.2) GO TO 15
```

```
      RANGE1 = RANGE(1)
      WRITE(24,100) TL1,MO,DAZE,TL2,UNTS,KA(1),RANGE1,RANGE1,KA(2),
     .  RANGE(2)
      INDX=0
      DO 14 L=1, 33, 4
      INDX=INDX+1
      L1 = L+1
      L2=L+2
      L3=L+3
14    WRITE(24,102)L,(KHAR(L,K),K=1,20),L1,(KHAR(L+1,K),K=1,20),
     & L2,(KHAR(L+2,K),K=1,20),L3,(KHAR(L+3,K),K=1,20),
     .  RANGE(INDX+1),KA(INDX+2),RANGE(INDX+2)
      L37=37
      L38=38
      L39=39
      L40=40
      WRITE(24,104) L37,(KHAR(37,K),K=1,20),L38,(KHAR(38,K),K=1,20),
     & L39,(KHAR(39,K),K=1,20),L40,(KHAR(40,K),K=1,20),
     . RANGE(11),KA(12),TOTAL,UNTST
      RETURN
15    RANGE1 = RANGE(1)
      WRITE(24,200) TL1,MO,DAZE,TL2,UNTS,KA(1),RANGE1,RANGE1,KA(2),
     .  RANGE(2)
      INDX=0
      DO 16 L=1, 33, 4
      L1 = L+1
      L2=L+2
      L3=L+3
      INDX=INDX+1
16    WRITE(24,202)L,(KHAR(L,K),K=1,20),L1,(KHAR(L+1,K),K=1,20),
     & L2,(KHAR(L+2,K),K=1,20),L3,(KHAR(L+3,K),K=1,20),
     .  RANGE(INDX+1),KA(INDX+2),RANGE(INDX+2)
      L37 = 37
      L38=38
      L39=39
      L40=40
      WRITE(24,204) L37,(KHAR(L37,K),K=1,20),L38,(KHAR(L38,K),K=1,20),
     & L39,(KHAR(L39,K),K=1,20),L40,(KHAR(L40,K),K=1,20),
     .  RANGE(11),KA(12),TOTAL,UNTST
      RETURN
      END
```

APPENDIX B

SAMPLE OUTPUTS

This appendix comprises sample outputs for soil water potential (PSIS), root concentration (ROOTSV), volumetric water content of soil (VH2OC), volumetric nitrate content of soil (VNO3C), and volumetric ammonia content of soil (VNH4C).

Outputs are presented in the form of two-dimensional maps of single integers, with each integer indicating the level of a particular variable. Beside each map, a scale defines the possible range occupied by each representative integer.

SOIL WATER POTENTIAL DAY 06/25

UNITS: BARS IN ROOT ZONE

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
1	8	8	9	9													9	9	8	8
2	8	8	8	9	9	9	9	9	9	9	9	9	9	9	9	9	9	8	8	8
3	8	8	8	9	9	9	9	9	9	9	9	9	9	9	9	9	9	8	8	8
4	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
5	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
6	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
7	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
8	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
10	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
11	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
12	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
13	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
14	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
15	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
16	7	7	7	7	7	7	7	7	7	8	8	7	7	7	7	7	7	7	7	7
17	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
18	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
19	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
20	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
21	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
22	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
23	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
24	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
25	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
26	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
27	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
28	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
29	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
30	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
31	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
32	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
33	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
34	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
35	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
36	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
37	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
38	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
39	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7	7
40	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9

TOTAL = -0.04 BARS

LEGEND

<= -15.0000

-15.0000 < 0 <= -10.0000

-10.0000 < 1 <= -6.0000

-6.0000 < 2 <= 3.0000

-3.0000 < 3 <= 1.5000

-1.5000 < 4 <= 1.0000

-1.0000 < 5 <= 0.6000

-0.6000 < 6 <= 0.4000

-0.4000 < 7 <= 0.2000

-0.2000 < 8 <= 0.1000

-0.1000 < 9 <= 0.0000

0.0000 < *

ROOT CONCENTRATION IN EACH CELL DAY 06/25

```
UNITS: G/CM**3 SOIL                                       LEGEND

                       1 1 1 1 1 1 1 1 1 1 2
     1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0                     <=  0.0000

                                                  0.0000 < 0 <=  0.0001
 1   3 2 1
 2   4 3 2 2 1 0
 3   5 3 2 2 1 0
 4   6 4 3 2 2 1 0                                0.0001 < 1 <=  0.0005
 5   5 4 3 2 2 2 0
 6   4 4 3 2 2 2 0
 7   3 3 2 2 2 1
 8   3 3 2 2 2 1                                  0.0005 < 2 <=  0.0050
 9   2 2 2 2 2
10   2 2 2 2 0
11   2 2 2 0
12   2 1                                          0.0050 < 3 <=  0.0100
13   1
14
15
16                                                0.0100 < 4 <=  0.0150
17
18
19
20                                                0.0150 < 5 <=  0.0200
21
22
23
24                                                0.0200 < 6 <=  0.0250
25
26
27
28                                                0.0250 < 7 <=  0.0300
29
30
31
32                                                0.0300 < 8 <=  0.0350
33
34
35
36                                                0.0350 < 9 <=  0.0400
37
38
39
40                                                0.0400 < *

     TOTAL =    2.96 GM DRY WEIGHT
```

VOLUMETRIC WATER CONTENT OF SOIL DAY 06/25

UNITS: CM**3/CM**3 SOIL

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
1	7	8	8	8	2	2	2	2	2	2	2	2	2	2	2	2	8	8	8	7
2	7	7	7	8	8	8	8	8	8	8	8	8	8	8	8	8	8	7	7	7
3	7	7	7	8	8	8	8	8	8	8	8	8	8	8	8	8	8	7	7	7
4	7	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	7
5	7	7	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	7	7
6	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8
7	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8	8
8	8	8	8	9	9	9	9	9	9	9	9	9	9	9	9	9	9	8	8	8
9	8	8	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	8	8
10	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
11	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
12	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
13	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
14	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
15	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9
16	6	6	6	6	6	6	6	6	6	7	7	6	6	6	6	6	6	6	6	6
17	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
18	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
19	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
20	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
21	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
22	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
23	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
24	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
25	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
26	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
27	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
28	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
29	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
30	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
31	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
32	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
33	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
34	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
35	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
36	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
37	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
38	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
39	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6	6
40	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9

TOTAL = 747.07 MM WATER

LEGEND

< 0.00E+00

0.00E+00 < 0 < 0.50E-01

0.50E-01 < 1 < 0.10E+00

0.10E+00 < 2 < 0.15E+00

0.15E+00 < 3 <0.20E+00

0.20E+00 < 4 <0.25E+00

0.25E+00 < 5 < 0.30E+00

0.30E+00 < 6 < 0.35E+00

0.35E+00 < 7 < 0.40E+00

0.40E+00 < 8 < 0.45E+00

0.45E+00 < 9 < 0.50E+00

0.50E+00 < *

VOLUMETRIC NITRATE CONTENT OF SOIL DAY 06/25

UNITS: MG/N PER CM**3

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
2	0	0	0	0	1	1	1	1	1	1	1	1	1	1	1	1	0	0	0	0
3	0	0	1	1	2	2	2	2	2	2	2	2	2	2	2	2	1	1	0	0
4	0	0	2	3	3	3	4	4	4	4	4	4	4	4	3	3	3	2	0	0
5	0	0	2	3	3	4	4	4	4	4	4	4	4	4	4	3	3	2	0	0
6	0	0	2	3	3	4	4	4	4	4	4	4	4	4	4	3	3	2	0	0
7	0	0	1	3	3	4	4	4	4	4	4	4	4	4	4	3	3	1	0	0
8	0	0	1	2	3	3	3	3	3	3	3	3	3	3	3	3	2	1	0	0
9	0	0	1	2	2	2	2	2	2	2	2	2	2	2	2	2	2	1	0	0
10	0	0	1	2	2	2	2	2	2	2	2	2	2	2	2	2	2	1	0	0
11	1	1	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	1	1
12	1	1	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	1	1
13	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
14	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
15	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
16	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
17	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
18	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
19	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
20	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
21	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
22	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
23	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
24	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
25	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
26	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
27	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
28	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
29	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
30	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
31	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
32	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
33	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
34	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
35	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
36	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
37	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
38	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
39	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
40	0	0	0	0	0	0	0	0	0	0.	0.	0	0	0	0	0	0	0	0	0

LEGEND

			<=	0.0000
0.0000	<	0	<=	0.0100
0.0100	<	1	<=	0.0200
0.0200	<	2	<=	0.0300
0.0300	<	3	<=	0.0400
0.0400	<	4	<=	0.0500
0.0500	<	5	<=	0.0600
0.0600	<	6	<=	0.0700
0.0700	<	7	<=	0.0800
0.0800	<	8	<=	0.0900
0.0900	<	9	<=	0.1000
0.1000	<	*		

TOTAL = 263.24 LBS N PER ACRE

VOLUMETRIC AMMONIA CONTENT OF SOIL DAY 06/25

UNITS: MG/N PER CM**3

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
1			0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0		
2				0	0	0	0	0	0	0	0	0	0	0	0	0	0			
3	0			0	0	0	0	0	0	0	0	0	0	0	0	0	0			0
4	0	0	0	2	7	7	7	7	7	7	7	7	7	7	7	7	2	0	0	0
5	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
6	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
7	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
8	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
9	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
10	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
11	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
12	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
13	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
14	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
15	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
16	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
17	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
18	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
19																				
20																				
21	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
22	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
23	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
24	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
25	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
26	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
27	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
28	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
29	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
30	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
31	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
32	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
33	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
34	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
35	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
36	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
37	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
38	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
39	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
40	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

TOTAL = 50.50 LBS N PER ACRE

LEGEND

			<=	0.0000
0.0000	<	0	<=	0.0100
0.0100	<	1	<=	0.0200
0.0200	<	2	<=	0.0300
0.0300	<	3	<=	0.0400
0.0400	<	4	<=	0.0500
0.0500	<	5	<=	0.0600
0.0600	<	6	<=	0.0700
0.0700	<	7	<=	0.0800
0.0800	<	8	<=	0.0900
0.0900	<	9	<=	0.1000
0.1000	<	*		

REFERENCES

American Society of Civil Engineers. 1990. Evapotranspiration and irrigation water requirements. American Society of Civil Engineers Manuals and Reports on Engineering Practice No. 70. New York.

Baker, D.N., J.R Lambert, and J.M McKinion. 1983. GOSSYM: A simulator of cotton crop growth and yield. South Carolina Agricultural Experiment Station Technical Bulletin 1080. Clemson University, Clemson, SC.

Bar-Yosef, B., and J.R Lambert. 1977. Simulation of root growth and water and nitrate uptake and transport in the soil under field conditions. South Carolina Agricultural Experiment Station Agricultural Engineering Research Series 24. Clemson University, Clemson, SC.

Boyer, J.S. 1970. Leaf enlargement and metabolic rates in corn, soybean, and sunflower at various leaf water potentials. Plant Physiology 46:233–235.

Brady, N.C. 1984. The Nature and Properties of Soils. 9th ed. Macmillan Publishing Co., New York.

Briones, A.M. 1988. Parameter estimates of soil nitrogen mineralization for cotton crop simulation model GOSSYM. Terminal report as visiting scientist at the Department of Agronomy, Mississippi State University, October 1, 1987–July 31, 1988.

Campbell, R.B., D.C Reicosky, and C.W Doty. 1974. Physical properties and tillage of paleudults in the southeastern coastal plains. Journal of Soil and Water Conservation 29:220–224.

Gardner, W.R., and M.S Mayhugh. 1958. Solutions and tests of the diffusion equation to the drying of soils. Soil Science Society of America, Proceedings 22:197–201.

Graham, J., D.T. Clarkson, and J. Sanderson. 1973. Water uptake by the roots of marrow and barley plants. Annual Report of Letcombe Laboratory, Oxfordshire, England.

Hillel, D. 1980. Fundamentals of Soil Physics. Academic Press, New York.

Huck, M.G., F.W.T. Penning de Vries, and M.G. Keizer. 1975. A model for simulating root growth and water uptake. Proceedings of the International Botanical Congress. Union of Soviet Socialist Republics.

Kafkafi, U., B Bar-Yosef, and A. Hadas. 1977. Fertilization decision model—a synthesis of soil and plant parameters in a computerized program. Crop Science 125:261–268.

Lambert, J.R., and D.N. Baker. 1984. RHIZOS: A Simulator of Root and Soil Processes. South Carolina Agricultural Experiment Station. (Manuscript in preparation).

McWhorter, J.C., and B.P. Brooks, Jr. 1965. Climatological and solar radiation relationships. Mississippi Agricultural Experiment Station Bulletin 715. Starkville, MS.

Miller, R.D., and D.D. Johnson. 1964. The effect of soil moisture tension on carbon dioxide evolution, nitrification, and nitrogen mineralization. Soil Science Society of America, Proceedings 28:644–647.

Oosterhuis, D.M., S.D. Wullschleyer, R.L. Maples, and W.N. Miley. 1990. Better Crops with Plant Food. Potash and Phosphate Institute, Atlanta.

Pair, C.H., W.W. Hinz, C. Reid, and K.R. Frost. 1983. Irrigation. The Irrigation Association, Arlington, VA.

Prasad, S.N., and M.J.M. Romkens. 1982. An approximate integral solution of vertical infiltration under changing boundary conditions. Water Resources Research 18:1022–1028.

Ritchie, J.T. 1972. Model for predicting ET from a row crop with incomplete cover. Water Resources Research 8:1204–1213.

Sabey, R.B. 1969. Influence of soil moisture tension on nitrate accumulation in soils. Soil Science Society of America, Proceedings 33:263–266.

Schwab, G.O., R.K Frevert, T.W. Edminster, and K.K. Barnes. 1981. Soil and Water Conservation Engineering. 3d ed. John Wiley and Sons, New York.

Stanford, G., and S.J. Smith. 1972. Nitrogen mineralization potentials of soils. Soil Science Society of America, Proceedings 36:465–472.

Stapleton, H.N., D.R. Buxton, F.L Watson, D.J. Nolting, et al. No date. Cotton: a computer simulation of cotton growth. Arizona Agricultural Experiment Station Technical Bulletin 206. Tucson.

Taylor, H.M., and H.R. Gardner. 1963. Penetration of cotton seedling taproots as influenced by bulk density. Soil Science 96:153–156.

U.S Department of Agriculture, Soil Conservation Service. 1972. National Engineering Handbook, section 4.